Fortran スマートプログラミング

2024 年度版

摂南大学・日本原子力研究開発機構

田口 俊弘

2025 年 1 月 12 日

はじめに

パソコンが登場してから今日まで，計算機とそれを取り巻く環境の劇的な進歩には目を見張るばかりです．計算機の利用目的も多様になり，それぞれの目的に応じて様々なプログラミング言語が開発されました．しかし，数値計算やコンピュータシミュレーションにおいては，コンピュータ本来の能力である「高速計算」ができればいいので，それほど新しい言語は必要ありません．

コンピュータで数値計算をするために昔からよく使われているプログラミング言語には，Fortran と C 言語があります．最近の大学では C 言語で計算機の講義を行うところが増えてきて，Fortran は影が薄くなっているようですが，コンピュータシミュレーションの分野では Fortran の方がよく使われています．これは，Fortran が科学技術計算用言語として発展してきているので，数値計算に便利な書式が多数採用されているからです．たとえば，複素数計算や行列計算などは文法に組み込まれているので簡単に書くことができます．数値計算向けのプログラムを書くには Fortran の方が書きやすいし，完成したプログラムをそのまま大型計算機で高速に実行させることも可能です．

Fortran は，計算機の歴史の初期に開発された由緒ある言語で，“FORmula TRANslation”が語源です．当初から数値計算用言語として開発されており，筆者が計算機を利用し始めた数十年前には，数値計算をするといえば，ほとんどが Fortran でした．しかし，科学技術計算に特化するということはユーザーの専門分野が限られるということであり，その結果コンパイラの値段があまり下がらなかったのが Fortran をマイナーにした大きな要因だと思います．C 言語は，パソコンが誕生したときに，そのハードウェアを活用するためのプログラミング言語として比較的安価に供給され，これが広く使われている理由の一つです．C 言語はそもそも UNIX という OS を記述するために開発された言語なので，計算機のハードウェア寄りの書式が多く，必ずしも初心者向きの言語だとは思えないのですが，プログラミングの専門家にすれば Fortran も C 言語も大して変わりません．安い方にシフトしたのはある意味やむを得ないことだと思います．

さて，1990 年代に入って無料の OS である Linux や FreeBSD の開発が開始され，これが非常な勢いで普及しました．普及した最大の要因は，インターネットを通じて，世界中のアマチュアプログラマが OS やその上で動作するアプリケーションの開発に参加したことです．コンパイラも同様で，このフリー OS には，開発環境として有志が作成した無料の C 言語 (gcc) が付属し，さらにこれを利用した無料の Fortran(g77，gfortran など) も付属しています．このため，Linux などをインストールすれば，パソコンでも無料で Fortran が使えます．また，これらフリーの Fortran は Windows や MacOS 上で動作するものも作られていているので，価格で Fortran を敬遠する理由はなくなったと言えます．

本書は，数値計算やコンピュータシミュレーションを目的として，これから計算機プログラムの勉強を始めようとする人を対象にした Fortran プログラム作成法の解説書です．単に文法だけを説明するのではなく，コンピュータの構造の話なども交えながら，効率が良くてエラーを見つけやすい，“スマートな Fortran プログラム”の書き方を説明します．

長い歴史を持っている Fortran は，古いバージョンとの互換性を保つ必要性から，若干ゆるい文法体系になっています．たとえば，デフォルトでは変数宣言をしなくてもかまわないため，変数名を書き間違えたときに発見するのが難しく，思わぬエラーを引き起こす可能性があります．また，コンピュータシミュレーションのような長いプログラムを書くときには，プログラムのチェックや修正がしやすいように書いておかないと，メンテナンスに苦労します．このため，多少冗長になっても，読みやすく，後でデバッグしやすいように書くべきです．本書では，これを念頭に置いて説明をしています．

もちろん，数値計算やコンピュータシミュレーションにとっては計算速度が命ですから，できる限り計算が速くなるようなプログラムにすることも必要です．プログラムとはコンピュータでの計算手順を記述するものなので，コンピュータの構造を考慮してちょっと計算順序を変えるだけでも高速化できることがあります．本書では，このようなコンピュータの内部構造で知っておいた方が良い部分もいくつか説明しました．孫子曰く，「彼を知り己を知れば百戦殆うからず」と．

Fortran は古くからある言語なのですが，1991 年に策定された Fortran90 を境に大きく変貌しました [1]．配列演算などの新しい機能を導入すると同時に，現代的な書式で書けるようになったのです．それまで 1 行に 72 文字までしか書けなかったという制限は無くなったし，1 行に複数の文を書くこともできるようになりました．Fortran90 以降も何度か改版されて，2024 年現在の最新規格は Fortran2023 ですが，これまでの改版によりオブジェクト指向プログラミングや並列処理の記述も可能になっています．しかし，パソコンを利用した数値計算にはそのような新しい概念はあまり必要ないので，本書では Fortran90 の改訂版である Fortran95 レベルの文法で説明をしました．このレベルでプログラムを書いておけばフリーの Fortran でコンパイルできるというメリットもあります．

本書には 10 章ありますが，基本的なプログラミングに関する知識は第 5 章までで十分です．初心者の方は，まず第 5 章までの内容を使って色々なプログラムを書き，プログラミングというものを良く理解して下さい．各章に演習問題を付けていますので練習用に使っていただければと思います．ただし，第 4 章までの演習問題は各章の内容に合わせた問題になっていますが，第 5 章の演習問題はそれまでのプログラミング知識全般を応用して実用的なプログラムを書くための練習になるような問題になっています．また，第 4 章までの章末に「スマートテクニック」と称して，プログラムを書くときの心構え，高速化のこつ，知っておくと便利な文法について紹介しています．これらも合わせて利用すれば，より質の高いプログラムを書くことができると思います．

これに対し，第 6 章以降はある程度プログラミングに慣れてから必要に応じて読んで下さい．基本的なプログラミング技法を一通り習得し，本格的なプログラムを作成するようになると，要求された機能を単に実現するだけでなく，より使いやすく，よりわかりやすいプログラムに改良したくなってきます．第 6 章以降はこのような要求の手助けをする文法の紹介になっています．読んでみて，「これは便利そうだ」と思ったら使ってみて下さい．さらに，実際にパソコンで Fortran プログラムを動作させる方法や，その際の注意事項などをまとめて付録にしています．こちらも必要に応じて活用してもらえればと思います．

良い計算機プログラムを作成するには単に書式を勉強するだけではだめで，実際に作って，コンパイルして，実行させる，というところまで進んでから初めて問題に気がつくことも多々あります．本書では，筆者の経験を元に，どのような手順を選べばコンピュータが高速で動作するかとか，どのようにすればエラーを少なくするだけではなく，エラーがあってもすぐ見つけられるようにできるか，といった基本的な技法を交えながら説明をしています．Fortran は定期的に更新されていて，本書には書かれていない新しい文法や高速化手法を取り入れた書式なども追加されていますが，プログラムを書くときの心構えは変わらないと思います．本書でコンピュータの構造と Fortran プログラミングの基礎を身につけてもらったら，後は自分でより良い手法を見つけ出してプログラムをさらに発展させていって下さい．

目　　次

第 1 章　計算機のしくみと Fortran プログラムの基本的な書き方

コンピュータ (計算機) とは，プログラムに記述された指示に従って動作をする機械のことです．最近では，コンピュータの性能がアップして，“並列コンピュータ”という複数の計算を同時に実行できる計算機もあり，これらの性能を引き出すためのプログラムの書き方など，プログラミングも多様化していますが，命令を逐次実行するという計算機の基本的な動作原理が大きく変わったわけではありません．本章では，プログラムを書く際に知っておいた方が良いと思われる簡単な計算機のしくみの説明から始めて，電卓レベルの計算を Fortran で書いて実行できるようなプログラムの書式を説明します．

1.1　計算機のしくみとコンパイラ

計算機の大まかなしくみを説明しましょう．計算機とは，プログラムに書かれた命令に従って自動的に計算を実行してくれる機械のことですが，その中心部を大きく分ければ，図 1.1 に示すように，計算を実行する CPU(Central Processing Unit，中央処理装置) と，データを保存する主記憶装置 (メモリ) から構成されています．計算機の動作を指示するプログラムもメモリに保存されていますが，これらは“プログラム領域”と“データ領域”という別々の場所に保存されています．CPU は，基本的には 1 回あたり一つの動作を行うことしかできないので，その動作が終了するまで次の動作には移れません．この一つ一つの動作指令が「命令」です．命令には，四則演算を行う算術命令，メモリを指定してデータを読み込んだり書き込んだりするデータ転送，周辺機器の動作制御などがあります．



図 1.1　コンピュータのしくみ

計算機プログラムとはコンピュータへの命令を記述した文の集まりですが，CPU を動作させる本来の命令は，加算 1 回とかメモリ書き込み 1 回といった単純なものである上に，“機械語”と呼ばれる数値で表されているので，これで直接プログラムを書くことはまず不可能です．そこで，人間が理解しやすいように，単語や記号を使った数式を使って複数の計算機動作を 1 行で書けるようにしたものが，Fortran や C 言語のような“プログラミング言語”です．プログラミング言語を使ってプログラムを書いておけば，これを“コンパイラ”という機械語への翻訳ソフトを使って，コンピュータで実行可能な機械語 (実行形式) に変換することができます．しかし，コンパイラは翻訳機に過ぎません．プログラミング言語という，我々には理解しやすい書式で書けるとはいっても，あくまでもコンピュータを動作させる手順になるように書かねばならないということは忘れないで下さい．

さて，計算機の主記憶装置であるメモリは RAM(Random Access Memory) と呼ばれる半導体チップでできた記憶装置で，電源を切るとデータが消えてしまいます．このメモリ上のデータはプログラムでは“変数”という形で記述します．これに対し，一般的に RAM より記憶容量が大きくて，電源を切ってもデータを保持できる補助記憶装置としてハードディスクや SSD などがありますが，プログラム上ではこれらは「読む」「書く」という動作指定によりデータのやりとりを行います．また，パソコンなどを使

うときにはモニターを見ながらキーボードでプログラムを入力したり実行したりしますが，これらとのデータのやりとりも「読む」「書く」という動作としてプログラムに記述します．このため，プログラミングの基本は，“変数”という形で表現されているメモリ上のデータを使って様々な計算を行い，必要に応じて周辺機器との読み書きを通じてデータの入出力を行う，という形式になります．

このように，メモリ上のデータは変数で表現するため，ハードウェアを意識する必要はあまりないのですが，プログラミングをする上で知っておいたほうがいい概念に“メモリアドレス”があります．データを保存するメモリには「アドレス」または「番地」と呼ばれる通し番号が付いていて，CPU はこの番号を使って読み書きするメモリを指定します．Fortran のプログラムではメモリアドレスに相当する数値を直接操作することはありませんが，プログラムの書式の中にはメモリに保存されている数値ではなく，メモリアドレスで指定している場合がいくつかあります．たとえば次の文を考えてみましょう．

```
a = x
```

後で説明しますが，これを“代入文”といい，右辺の変数 x の値を左辺の変数 a に代入するという動作を表しています．計算機的には，x のメモリに保存されているデータを読み込んで，a のメモリに書き込むという動作を示しています．このとき注意して欲しいのは，右辺の変数と左辺の変数では意味が違うことです．右辺の x は，x という名で指定したメモリに格納されている数値を示しているのですが，左辺の a は保存先のメモリアドレスを指定しています．このため，

```
10 = 2
```

のような数値に代入する代入文や

```
x+b = 5
```

のような数式に代入する代入文はありえないわけです．もっとも，「代入文」という意味を考えれば，この程度のことは意識せずとも間違うことはないと思います．しかし，プログラムの書式の中には，変数を与えた場合に数値ではなくアドレスを指定していることを意識した方が良いものがいくつかあります．そのあたりを理解した上でプログラムを作成すれば，思わぬエラーが発生したときの原因を探る手がかりの一つになります．「プログラムエラー」というのは，単に計算式を書き間違えて実行結果が予想値と異なるという場合だけではありません．動作中に突然実行が停止するといった予想外の動作で現れるエラーもあるのです．

なお，プログラムもメモリに入っているので，やはりアドレスが存在し，CPU はプログラムの先頭アドレスから順に実行していきます．このとき，ジャンプ命令を使えば指定したアドレスに移動してそこからプログラムを実行することが可能なので，これを利用することで同じ計算をくり返し行ったり条件に応じて異なる計算を行うことができます．そもそも，「プログラムの実行開始」が他の計算機動作から実行指定したプログラムの先頭アドレスへジャンプして動作をそちらに移すことなのです．最近のプログラミング技法では，構造化プログラミングと言ってループや条件分岐などをジャンプ命令を使わずに書けるようになっているので，あまり意識する必要はないのですが，動作の基本であることは知っておいて下さい．

1.2 Fortran プログラムの書式

Fortran プログラムは，基本的にコンピュータに与える命令を次の形をした“文”で記述します．

```
動作指示語　動作制御パラメータ
```

すなわち，先頭に“動作指示語”と呼ばれる計算機の動作を指定する単語を書き，それに続けて，動作を制御するための数値や予約語を“動作制御パラメータ”として記述します．ただし動作制御パラメータを記述する必要のない，動作指示語だけの命令も存在します．

実行形式に変換されたプログラムを起動すると，コンピュータは，

実行開始処理　→　プログラムに記述された命令を順に実行　→　実行終了処理

という流れで動作します．実行を開始したときに最初に実行する部分を“メインプログラム”，または“メインルーチン”といいます．言わばプログラムの本体です．本章から第 3 章までは，メインプログラムだけを使った Fortran の基本的プログラムの書き方について説明します．メインプログラムと同レベルの完結したプログラムには，他のプログラムの中から実行開始を指示してその機能を利用する“サブルーチン”がありますが，これについては第 4 章で説明します．

1.3　メインプログラムの開始と終了

Fortran では特別な手続きをせずにプログラムを書くと，それがメインプログラムと仮定されます．しかし，それではサブルーチンと区別しにくいので，program 文を用いて最初にプログラムの名前を書きます．program 文は以下の形式です．

```
program プログラム名
```

これに対し，メインプログラムの終了は end program 文で指定します．

```
end program プログラム名
```

ここで，program 文と end program 文で指定する“プログラム名”は同じでなければなりません．このため，メインプログラムは次のような構造になります．

```
program code_name
  .......
  .......
end program code_name
```

プログラム名は，頭文字が a〜z のどれかであれば，後は a〜z，0〜9 をどの様な順序で並べたものでも使えます．また，この例のようにアンダースコア (_) も使えるので，これをスペースの代わりに使えば単語をつないだ長いプログラム名を付けることもできます．

program 文と end program 文の間に Fortran の文法に従った文を並べて，動作させたい計算手順を記述します．動作手順を記述する文を“実行文”といいます．しかし，プログラムに記述するのは実行文だけではありません．計算途中で必要となる変数 (メモリ) 領域を確保するための宣言文も書かねばなりません．このような計算動作に直接携わらない文を“非実行文”といいます．Fortran では，非実行文をプログラムの最初に集約して，実行文はその後に書きます．このため，動作の開始は program 文の次の文ではなく，最初の実行文になります．

完結したメインプログラムの一例を示します[1]．

```
program test_code
  implicit none
  real x,y,z
  x = 5
  y = 100
  z = x + y*100
  print *,x,y
  print *,'z =',z
end program test_code
```

[1] この例のように，プログラム内部の文は，先頭にスペースを数個入れて，program 文と end program 文よりも少し右にずらして書きます．そうすれば，メインプログラムの範囲が明確になります．

このプログラムにおける実行文・非実行文の区分と，動作開始から終了までの実行の流れを図 1.2 に示します．実行形式のプログラムを起動すると，最初の実行文から動作が開始し，上から順に 1 行ずつ実行して，end program 文に到達した段階で動作終了となります．2.2 節で述べる do 文を使ったくり返しや，3.1 節で述べる if 文による条件分岐など，動作指定によっては所定の位置にバックしたり，条件に応じて実行するかどうかの選択ができますが，それでもそれぞれの文が終了した後に次の文が実行されるという基本的動作は変わりません．

これは計算機が 1 回に一つの動作しかできないためです．プログラム全体を見渡して実行するのではなく，1 行ずつ順に実行していくので，バックするような動作指定をしない限り，下方に書いた実行文は上方に影響を及ぼしません．プログラムはこのことを常に念頭に置いて書かなければなりません．



図 1.2　プログラム開始から終了までの流れ

なお，if 文を使って，条件によって途中でプログラムを終了するときや，サブルーチン内でプログラムを終了するときには stop 文を用います．stop 文を実行すると，その時点でプログラムが終了します．たとえば，

```
program fluid_code
  implicit none
  real x,y
  integer m,n
   .......
  if (m < 0) stop
   .......
end program fluid_code
```

と書けば，m < 0 のときにプログラムは終了し，それ以降は実行しません．

1.4　代入文と演算の書式

多くの実行文では，いくつかの“数値”を伴って動作を指示します．プログラム中で“数値”を与える方法は 3 種類あります．−500 とか 3.14 のような数字を直接書く“定数”，x とか yrz のような単語で示した“変数”，およびそれらを使って x+3 や sin(y-5) のような計算手順を表した“計算式”です．計算式に書かれている手順も計算機の動作なのですが，プログラム中の計算式は，その計算結果を動作命令に与える“数値”として位置づけられています．このため，計算式を書いただけでは実行文になりません．

プログラムで最も良く使う実行文は“代入文”です．これは，

```
変数 = 計算式
```

という形をしています．計算式の代わりに，定数や変数を書くこともできます．1.1 節で説明したよう

に，プログラムにおける“変数”とはコンピュータのメモリ領域のことで，数値を入れる箱だと考えればいいでしょう．代入文は右辺の“数値”を左辺で示した変数メモリに格納する動作を表します．よって，

```
計算式 = 計算式        ! これはエラー
```

という形は使えません．Fortran において“＝”という記号は，「左辺と右辺が等しい」という意味ではないからです．たとえば，

```
x = 4 + 2
y = 9 - x
y = y + 1
```

のようなプログラムを考えてみましょう．プログラムは上から順に実行されるので，まず 4 + 2 の結果である 6 が変数 x に代入されます．次に，9 から変数 x に保存されている数値を引いた値が変数 y に代入されるので，y には 3 が代入されます．最後の文は数学の等式と見れば変ですが，イコールが“代入する動作”だと理解できれば特に不思議な文ではありません．代入という動作は，右辺の全ての演算が終了した後で実行されるため，最後の文では，まず y+1 を計算し，その結果が y に代入されます．すなわち y には 4 が代入されます．

また，変数に値を代入すると，それまでその変数に入っていた値が消えてしまうことも忘れてはいけません．たとえば，変数 x と変数 y に代入されている値を交換したいときに，

```
x = y
y = x
```

と書いてもうまくいきません．これでは，x と y が同じ値になってしまいます．交換したいときには，一度別の変数に保存しておく必要があります．たとえば，

```
s = x
x = y
y = s
```

のように書けば，x と y の値が入れ替わります．

Fortran における基本演算の書き方と使い方を表 1.1 に示します．

表 1.1　演算記号の書き方と使い方

演算記号	演算の意味	使用例	使用例の意味
+	足し算	x+y	$x + y$
-	引き算	x-y	$x - y$
*	掛け算	x*y	$x \times y$
/	割り算	x/y	$x \div y$
**	べき乗	x**y	x^y
-	マイナス	-x	$-x$

べき乗までの 2 項演算には以下のような優先順位があり，優先順位の高い方が低い方より先に計算されます．これは数学における演算順序と同じです．

```
べき乗 ＞ 掛け算または割り算 ＞ 足し算または引き算
```

掛け算と割り算のような同レベルの演算は左から順に計算されます．さらに，かっこを使えばかっこの中が優先的に計算されます．たとえば，

```
f = x + (y-3)/z**2
```

という文では，一番最初に y-3 を計算し，次に z の 2 乗を計算し，次に y-3 の結果を z の 2 乗の結果で割り，その結果を x に加えて，最後に f に代入する，という動作になります．

なお，ときどき見かける間違いに，

$$f = x + y/ab$$

という数式を次のようなプログラムにするものがあります．

```
f = x + y/a*b
```

数学では，y/ab の記述を $\dfrac{y}{ab}$ と解釈することが多いのですが，プログラムでは掛け算と割り算が同じレベルなので，y/a*b は “y を a で割って，その結果に b を掛ける ”です．よって，数学的解釈になるようにするには，

```
f = x + y/(a*b)
```

と書かなければなりません．

1.5 数値の型

コンピュータという機械が取り扱う数値は 2 進数で表されていて，基本的には整数です．たとえば，1byte の数は，8bit，つまり 0 か 1 のどちらかを選択できる数が 8 個並んだもので，10 進数では 0〜255 (=2^8-1) までの整数を表すことができます[2]．Fortran では小数点のない数字，56 とか −1112 などの数字を “整数型 ”といいます．Fortran の整数型数は 4byte (32bit) で表現されていて，-2^{31}〜$2^{31}-1$ の整数を扱うことができます．

しかし，数値計算やシミュレーションをする場合には，3.14 のように小数点以下を含んだり，3×10^{19} とか，1.6×10^{-27} とかいった様々なスケールの数値を取り扱うため，整数型だけでは不便です．特に，整数型数は小数点以下の表現ができないので，割り算をすると全て切り捨てになります．これを忘れてプログラムを書くと，よく間違いを起こします．たとえば，

```
x = (5/2)**2
y = x**(3/2)
t = 2/3*y*x
```

と書くと，x，y，z はいくつになると思いますか？ 答えは x = 4，y = 4，t = 0 です．これは整数の割り算が切り捨てになるため，5/2 → 2，3/2 → 1，2/3 → 0 になるからです．

そこで，整数型の他に，3.14 のような少数点以下を含んだり，1.6×10^{-27} のような，$A\times10^{B}$ という形式の数値を取り扱うことができます．これを “実数型 ”といい，A の部分を “仮数部 ”，B の部分を “指数部 ”といいます．通常，数値計算は実数型で計算しなければなりません．

実数型には有効数字の違う 2 種類が用意されています．単精度実数型と倍精度実数型です．単精度実数とは 4byte で表される実数のことで，仮数部の有効数字は 7 桁程度です．これに対し，倍精度実数とは 8byte (64bit) で表される実数のことで，仮数部の有効数字は 15 桁程度です．7 桁あれば問題ないと思われるかもしれませんが，何百万個もの数字の合計を計算したり，次数の高い多項式の計算をするときなどでは，計算回数の増加につれて有効桁数が減少することを考慮しなければなりません．このため，通常は倍精度実数で計算をします．

Fortran における基本的な実数型は単精度であり，倍精度実数型を使用するときにはそれを意識した書式で書かなければなりません．しかし，最近のコンパイラはオプションを指定すればデフォルトの実数型を倍精度にする “自動倍精度化機能 ”を持っているので，本書では，単精度実数型と倍精度実数型

[2]bit や byte の詳細と実数型の表現については 1.10.4 節参照.

を使い分けることはせず，単に“実数型”と表現します[3]．よって，特に単精度で計算する必要がなければ，コンパイル時に自動倍精度化オプションを付加して倍精度で計算して下さい．多くの計算機は倍精度実数計算を高速で行えるハードウェアを内蔵しているので，倍精度計算をしてもさほど実行速度は下がりません．

プログラム上で，整数型の定数と実数型の定数は小数点の有無で区別します．たとえば，`100` とか，`-12345` と書けば整数型ですが，`100.0` とか，`-12345.` とか，`-0.0314` とか書けば実数型になります．また，1.6×10^{23} を入力したいときには，`1.6e23` と書きます．すなわち，$A\times10^{B}$ は `A`e`B` と書きます．B が負の場合でも `1.6e-19` のように `e` の後に続けて書きます．また，`6e20` のように eB を付加した場合には小数点が無くても実数型になります．たとえば，

$$a = 3.141592\,r^2 + 3x^5 + 6.5\times10^{-5}\,x - 10^5$$

という式をプログラムで書けば以下のようになります．

```
a = 3.141592*r**2 + 3.0*x**5 + 6.5e-5*x - 1e5
```

この例のように，`r**2` とか `x**5` のように整数べき乗の指数 (2 とか 5) には整数型を使います．

先ほど整数型を使ったら期待どおりの結果が出ない例を挙げましたが，以下のように実数型を使えば正しい結果が得られます．

```
x = (5.0/2.0)**2
y = x**(3.0/2.0)
t = 2.0/3.0*y*x
```

Fortran の便利な機能の一つは，複素数が使えることです．複素数にも単精度複素数型と倍精度複素数型がありますが，自動倍精度化機能を使えばデフォルトが倍精度複素数型になるので，本書では単に“複素数型”と表現します．複素数型の定数は，

```
(0.0,1.0)
(1e-5,-5.2e3)
(-3200.0,0.005)
```

のように，2 個の実数をコンマでつないで，かっこで囲みます．前半が実部，後半が虚部です．つまりこの例は，それぞれ，i，$10^{-5} - 5.2\times10^{3}\,i$，$-3200 + 0.005\,i$ を表した複素数型定数です．

なお，ここまで読むと整数型は必要がないと思われるかもしれませんが，そうではありません．2.1 節で説明する配列の要素を指定する数や，2.2 節で説明する `do` 文のカウンタ変数は整数型でなければなりません．また，オン・オフを表すだけの変数を作るときにも整数型を使います．すなわち，整数型数は“順番”や“区別”を表すときに不可欠です．また実数の整数部を取り出したいときや，剰余 (あまり) を計算するときも整数型を利用します．数値計算プログラムの中でも整数型の使い道は多いのです．

計算式中に異なる数値型が混在するときは，情報量の多い方の型に合わせて計算し，その型の値が結果になります．たとえば，実数型と整数型の計算は整数型の数値を実数型に変換して実数型と実数型の計算を行い，実数型の結果になります．複素数型と実数型の計算の場合にはその実数型の数値を実部とした複素数型にして複素数計算をし，結果は複素数型になります．ただし，掛け算と割り算の計算は左から順に行うので注意が必要です．最初の整数型を使った計算例で，

```
t = 2/3*y*x
```

が 0 になるのは，最初に計算されるのが `2/3` という整数型÷整数型だからです．これを，

```
t = y*x*2/3
```

[3]自動倍精度化については付録 C 参照．精度の変更については 10.1 節で説明しています．

と書けば，切り捨ては起こりません．

1.6 変数の宣言

次に，数式の計算結果を保存する変数の使い方を説明します．変数の名前は，プログラム名と同様に頭文字が a〜z のどれかであれば，後は a〜z，0〜9 をどの様な順序で並べたものでも使えます．たとえば，abc とか k10xy 等です．Fortran では大文字と小文字を区別しないため，abc と ABC は同じ変数になります．変数にも型があり，計算結果に応じた型の変数を使わなければ正確に保存することができません．この変数の型を決める文を“宣言文”といいます．型を決めると同時に，変数のメモリ領域を確保します．このため，計算に用いる変数は全て宣言しなければなりません．宣言は一度しかできず，プログラムの実行時に変更することはできません．宣言文は非実行文です．

ところが Fortran には“暗黙の宣言”があり，通常は宣言しなくても文法的な間違いにならないので，タイプミスなどで思わぬエラーが発生する可能性があります．これを防ぐため，プログラムの 2 行目，すなわち program 文の次の行に必ず以下の implicit 文を書いておきます．

```
implicit none
```

この文を入れておけば，宣言せずに使用した変数があるとコンパイルエラーになり，タイプミスのチェックができます．

数値計算は基本的に実数で行いますが，実数型変数は次のように宣言します[4]．

```
real 変数1，変数2，...
```

これに対し，整数型の変数は，次のように宣言します．

```
integer 変数1，変数2，...
```

宣言文は非実行文なので，全ての実行文より前に書かなければなりません．たとえば，メインプログラムの始まりは以下のようになります．

```
program test1
  implicit none
  real x,y,z,omega,wave,area
  integer i,k,n,imin,imax,kmax
   .......
```

real や integer 等の型宣言文は何行書いても良いし，順番も無関係です．

Fortran の暗黙宣言では，名前の頭文字が a〜h か o〜z の変数は実数型で，i〜n の変数は整数型でした．このため，整数型変数の頭文字を i〜n にする習慣があります．全てを宣言する以上，基本的に変数名の付け方に制限はないのですが，整数型は用途が限られているので，頭文字を限定する方が良いと思います．実数型数の名前はあまり頭文字にこだわる必要はありませんが，原則として整数型をイメージする i,j,k,l,m,n の 1 文字変数は使わない方が無難です．

複素数型変数の宣言文は，

```
complex 変数1，変数2，...
```

です．複素数型も用途が限定されているので名前の付け方に規則をつける方が良いでしょう．複素数型変数名は，頭文字を c か z にすることが多いようです．

プログラムにおいて，数値を変数に保存する意義は大別して二つあります．一つはプログラム全域にわたって共通してその値を利用するためであり，もう一つは狭い範囲で計算結果を一時的に保存するた

[4] 1.5 節で述べたように，本書は自動倍精度化の利用を前提にして説明しているので，この宣言文で変数は倍精度実数型になりますが，自動倍精度化を利用しなければ単精度実数型になるので注意してください．複素数型変数の宣言も同様です．

めです．この二つは意識して使い分けるようにします．特に，前者の大域的に利用する変数には長めで意味のある名前を付けるべきです．これは，1 文字のような短い変数名を使うと，プログラムが長くなるにつれて，どこでその変数を使っているかを検索するのが難しくなるからです．

変数に数値を代入するときは，右辺の計算結果を左辺の変数の型に変換して代入します．このため，実数の計算結果を整数型の変数に代入すると，小数点以下は切り捨てられます．たとえば，

```
real x
integer n
x = 3.14
n = x + 6
```

の結果，n に代入される値は 9 です．

また，複素数型の計算結果を実数型の変数に代入すると，その複素数の実部が代入されます．たとえば，

```
real x
complex c
c=(1.0,-2.0)
x=c**2
```

とすると，x = −3.0 になります．逆に，複素数型の変数に実数型の数値を代入すると，実部に結果が代入され，虚部は 0 になります．たとえば，

```
real x
complex c
x=5
c=x**2
```

とすると，c = (25.0, 0.0) になります．

1.7 組み込み関数

数値計算上よく使う数学関数はあらかじめ用意されています．この用意された関数を “ 組み込み関数 ”といいます．表 1.2 に代表的な組み込み数学関数を示します．

表 1.2 数学関数の書き方と使い方

組み込み関数	名　称	数学的表現	必要条件	関数値 f の範囲
sqrt(x)	平方根*	$\sqrt{x}$	$x \geqq 0$	
sin(x)	正弦関数*	$\sin x$		
cos(x)	余弦関数*	$\cos x$		
tan(x)	正接関数	$\tan x$		
asin(x)	逆正弦関数	$\sin^{-1} x$	$-1 \leqq x \leqq 1$	$-\frac{\pi}{2} \leqq f \leqq \frac{\pi}{2}$
acos(x)	逆余弦関数	$\cos^{-1} x$	$-1 \leqq x \leqq 1$	$0 \leqq f \leqq \pi$
atan(x)	逆正接関数	$\tan^{-1} x$		$-\frac{\pi}{2} < f < \frac{\pi}{2}$
atan2(y,x)	逆正接関数[5]	$\tan^{-1}(y/x)$		$-\pi < f \leqq \pi$
exp(x)	指数関数*	e^x		
log(x)	自然対数*	$\log_e x$	$x > 0$	
log10(x)	常用対数	$\log_{10} x$	$x > 0$	
sinh(x)	双曲線正弦関数	$\sinh x$		
cosh(x)	双曲線余弦関数	$\cosh x$		
tanh(x)	双曲線正接関数	$\tanh x$		

関数名の後のかっこは必ず必要です．かっこの中の x や y を “引数 (ひきすう)”といいます．関数を計算式中に記述すると，引数に対する関数値が結果となってその計算式を実行します．たとえば，

```
c = exp(-x**2) + sin(10*x+3) - 2*tan(-2*log(x))**3
```

のように書くことができます．この例のように関数の引数に関数を使った計算式を与えることも可能です．

表 1.2 の数学関数は，引数に実数型数を与えると実数型の結果になる “実数型関数”ですが，名称に “*”の付いている関数は，引数が複素数型でも使えます．Fortran の組み込み関数には，引数の型や精度に応じて計算をする “総称名機能”があるので，複素数を引数に与えた場合の関数値は複素数型の結果になります[6]．

ただし，総称名機能では引数の数値型に対応した関数が用意されていないとコンパイルエラーになります．たとえば，$\sqrt{2}$ を計算したくて，sqrt(2) と書くとエラーです．“2”は整数型の定数であり，整数型数の平方根は用意されていないからです．sqrt(2.0) のように実数型を引数に書かなければなりません．

この他，絶対値や余りを計算するのにも組み込み関数を使うし，実数型を複素数型に変換するなどの型変換も組み込み関数を使って処理します．その中の代表的なものを表 1.3 に示します．

型変換関数は，数値型が限定されている場所に，その数値型以外の値を指定したいときなどに使います．たとえば，整数型の変数 n の平方根を計算したいときには，上記のように sqrt(n) と書くことはできません．このときは，sqrt(real(n)) と書きます．また，整数化は切り捨て演算を含んでいるので，それを積極的に利用するときにも使います．たとえば，正の実数 x の小数点以下を四捨五入した整数は，int(x+0.5) で計算することができます．

表 1.3　型変換関数などの組み込み関数

組み込み関数	名　称	引数の数値型	関数値の数値型	関数の意味
real(n)	実数化	整数	実数	実数型に変換
abs(n)	絶対値	整数	整数	n の絶対値
mod(m,n)	剰余[7]	2 個の整数	整数	m を n で割った余り
int(x)	整数化	実数	整数	整数型に変換 (切り捨て)
nint(x)	整数化	実数	整数	整数型に変換 (四捨五入)
sign(x,s)	符号の変更	実数	実数	s$\geqq 0$ なら $\lvert$x$\rvert$，さもなくば $-\lvert$x$\rvert$
abs(x)	絶対値	実数または複素数	実数	x の絶対値
mod(x,y)	剰余	2 個の実数	実数	x を y で割った余り[8]
real(z)	複素数の実部	複素数	実数	z の実部
imag(z)	複素数の虚部	複素数	実数	z の虚部
cmplx(x,y)	複素数化	2 個の実数	複素数	x $+ i$y
conjg(z)	共役複素数	複素数	複素数	z の共役複素数

複素数型の定数は，(1.0,-2.0) のような表現で指定することができますが，変数や計算式で実部と虚部を指定した複素数を作るときには関数 cmplx を使います．たとえば，

[5]逆正接関数 atan2(y,x) は，座標点 (x, y) の偏角を計算する関数です (x と y の順序に注意)．よって，表の数学的表現には “$\tan^{-1}(y/x)$”と記述してありますが，実際には x と y の値に応じて $-\pi$ から π の間の角度が結果になります．

[6]表 1.2 の必要条件と関数値の範囲は引数が実数型の場合です．この必要条件を満たさない実数型数を与えると実行時エラーになります．しかし複素数型を与えた場合には必ずしもエラーにはなりません．

[7]整数の剰余を計算する関数 mod の利用例としてくり返しの制御があります．3.1 節を参照して下さい．

[8]実数 x を実数 y で割ったときの “余り”とは，x/y の小数点以下を切り捨てた整数を n としたときの x$-n$y のことです．

```
complex z,z1
real x,y
x = 10.0
y = -3.5
z = cmplx(x,y)
z1 = cmplx(x**2,y+2.0)
```

のように書きます．この場合，z = x + iy であり，z1 = $x^2 + i(y + 2)$ になります．

1.8 print 文による簡易出力

数値計算を目的としたプログラムでは，得られた計算結果を表示したりファイルに保存する必要があります．入出力の詳細については第 5 章で説明しますが，最低限，print 文を使った標準形式による数値出力は覚えておいて下さい．print 文の一例を以下に示します．

```
integer n
n = 3
print *, 4+5, n, n*2, 2*n-11
```

このように，“print *,”に続いて変数や計算式をコンマで区切って並べると，それらの計算結果が横に並んで出力されます．上例の場合には，4+5 の結果である 9 から 2*n-11 の結果である −5 までが以下のように出力されます．

```
           9           3           6          -5
```

なお，“print *,”の“*”は，標準形式で出力することを意味しているのですが，とりあえずは形式的に書くものだと覚えて下さい．出力形式の指定方法の詳細は 5.3 節で説明します．

複数の数値を出力するときに数字だけを出力すると，どれがどの数値かわからなくなる可能性があります．このような場合は，文字列を併用して変数の意味を同時に出力します．Fortran における“文字列”とは，2 個のアポストロフィ「'」または 2 個のダブルクォーテーション「"」ではさんだ文字の並びのことで，print 文中で数値といっしょに並べると，その文字の並びがそのまま出力されます．たとえば，

```
real x
x = 3
print *,'x = ',x,'  x**3 = ',x**3
```

というプログラムの出力は，

```
x =    3.00000000000000        x**3 =    27.0000000000000
```

となります[9]．

1.9 コメント文，継続行，複文

Fortran では“!”の後に続く文字は全て無視されます．すなわち何を書いても実行とは無関係です．これを“コメント文”といいます．コメント文を機会あるごとにプログラム中に入れて，書かれている内容を表示するように心がけましょう．さもなくば，プログラムが長くなるにつれて，自分でも何を書いたのか忘れてしまいます．たとえば，

```
!  area of circle
   s = pi*r*r
```

のように，1 列目に“!”を書けば，その行はコメント行になります．また，

[9] ここで 15 桁の数値が出力されているのは実数が倍精度だからです．単精度実数を使った場合には 7 桁になります．

```
s = pi*r*r          !  円の面積
v = 4*pi*r*r*r/3    !  球の体積
```

のように，実行文の末尾に書き込むこともできるし，日本語を書くこともできます．ただし，日本語環境によっては正しく認識できずに文字化けすることがあります．可般性を考慮するならローマ字つづりでも良いから半角英数字だけでコメントを書いた方が無難です．

計算式が非常に長くて，作成に使っているエディタウィンドウの横幅を越えると読みづらくなります．また，コンパイラによっては 1 行に書ける文字数に制限があるので，そのままではエラーになることもあります．そこで，1 行の文を途中で改行して，複数行に分割して書くことができます．これを “継続行”といいます．分割したときは，次の行に継続するという意志を示す印として，行末に “&”の文字を書きます．たとえば，

```
print *,alpha,beta,gamma,delta,epsilon,zeta,eta,iota,kappa,lambda,mu
```

という 1 行は，

```
print *,alpha,beta,gamma,delta,epsilon &
       ,zeta,eta,iota,kappa,lambda,mu
```

と分割して書くことができます．継続行は何行にわたってもかまいません．

```
print *,alpha,beta,gamma &
       ,delta,epsilon &
       ,zeta,eta,iota &
       ,kappa,lambda,mu
```

と書いても同じです．ただし，最後の行に “&”を付加するとエラーになるので注意して下さい．

長い文字列を書くときは，文字列の最後に “&”を入れて継続させることができます．その際，次の行の開始文字の前にも “&”を書いておきます．たとえば，

```
print *,'ABCDEFG&
       &hijklmn'
```

と書けば，ABCDEFGhijklmn と出力されます．

逆に，1 行に複数の文を書くことも可能です．これを “複文”といいます．2 個以上の文を 1 行で書くには，文と文を分離する記号として “;”を挿入します．たとえば，

```
x = 1; y = 2; z = 3
```

のように書くことができます．ただし，複文の使用はこのような短い代入文を書く場合に限定して下さい．プログラムは 1 行で一つの完結した動作を表すのが基本です．1 行に複数の文を書くと，動作の流れが見えにくくなるので，多用しない方が良いと思います．

1.10 プログラミング スマートテクニック (その 1)

計算手順が複雑になったり大量のデータを処理するようになると，数式を単純にプログラムに置きかえるのではなく，計算機の特性を考慮したプログラムにすることで計算効率や精度を高めることができます．ここでは，計算速度を向上させるテクニックや，計算精度を高める書き方について説明します．また，計算機で利用できる数値の大きさについても説明します．

1.10.1 演算の速度を考える

コンピュータの計算動作は a+b のような加算が基本です．a-b のような減算は b を負数に変換して加算するだけなので加算とそれほど実行時間は変わりませんが，a*b のような乗算は加算のくり返し動作ですから，加減算よりかなり遅いです．a/b のような除算にいたっては，減算のくり返しを条件付きで行うので，さらに時間がかかります．この演算速度の比較を書けば次のようになります．

$$\text{加減算} \gg \text{乗算} \gg \text{除算}$$

このため，割り算ができるだけ少なくなるような計算手順にします．たとえば，

```
x = a/b/c
```

と書くより，

```
x = a/(b*c)
```

と書く方が速くなります．最近の計算機はかなり高速に計算することができるので，数回の計算だけならこのような書き換えはあまり意味がありませんが，大量にくり返し処理をするときには価値があります．

べき乗算はもっと遅いので，2 乗や 3 乗程度のときは掛け算にする方が速くなります．たとえば，

```
x = a**2 + b**3
```

と書くより，

```
x = a*a + b*b*b
```

と書く方が速くなります．ただし，指数が大きいときには意味がないし，プログラムもわかりにくくなるので 3 乗程度までで良いと思います．

実数のべき乗 (1.2 乗とか，3.7 乗など) は，対数関数と指数関数を使って計算するので時間がかかります．このため，$\frac{1}{2}$ 乗に関しては，組み込み関数 sqrt を利用する方が速いです．たとえば，

```
y = x**0.5
z = y**1.5
```

と書くより，

```
y = sqrt(x)
z = y*sqrt(y)
```

と書く方が速くなります．

1.10.2 桁落ちに気を付ける

実数型数の有効数字は倍精度でも 15 桁程度です．よって，値の接近した 2 個の実数の引き算をするときは気を付けなければなりません．たとえば，

$$2000.06 - 2000.00 = 0.06$$

ですが，計算結果 0.06 は元の 2000.06 と比べると有効数字が 5 桁も少なくなっています．これを “桁落ち” といいます．桁落ちするような引き算はできるだけ避けねばなりません．

たとえば，次のように変数 x1 に小さい正の数 h1 を加えて x2 を計算し，x2 に小さい正の数 h2 を加えて x3 を計算したとします．

```
real x1,x2,h1,h2
x1 = 1.0
h1 = 1e-7
h2 = 2e-7
x2 = x1 + h1
x3 = x2 + h2
print *,x3-x1,h1+h2
```

最後の print 文の出力を見ればわかりますが，x3-x1 の値が必要なときには，引き算で直接計算するよりも h1+h2 で計算する方が精度が上がります．

2 次方程式 $ax^2 + bx + c = 0$ の解を求めるときにも注意が必要です．この方程式の解の一つは，

$$x_1 = \frac{-b + \sqrt{b^2 - 4ac}}{2a}$$

ですが，$b > 0$ かつ $b^2 \gg |4ac|$ のときには，b と $\sqrt{b^2 - 4ac}$ がほぼ等しくなるので，$-b + \sqrt{b^2 - 4ac}$ の計算をすると桁落ちする可能性があります．そこでこの解を計算するときは，分子の有理化をした公式を使います．すなわち，分母分子に $-b - \sqrt{b^2 - 4ac}$ を掛けると，

$$x_1 = \frac{\left(-b + \sqrt{b^2 - 4ac}\right)\left(-b - \sqrt{b^2 - 4ac}\right)}{2a\left(-b - \sqrt{b^2 - 4ac}\right)} = \frac{2c}{-b - \sqrt{b^2 - 4ac}}$$

となりますが，最後の公式を使えば桁落ちする心配はありません．2 解とも計算するときには，2 解の積が c/a であることを利用して，まず桁落ちしない方の解 x_1 を計算した後で，もう一方の解を $c/(ax_1)$ で計算すると良いでしょう．

1.10.3 π の作り方

物理や数学の計算をするときには円周率 π を使うことが多いです．π の値を倍精度計算用に有効数字 16 桁で表せば，

```
real pi
pi = 3.141592653589793
```

となります．これをこのまま使えば良いのですが，計算機によっては有効数字が異なるし，数値を覚えるのも面倒です．そこで逆三角関数を使って計算で求める方法がよく使われています．たとえば，

```
real pi
pi = acos(-1.0)
pi = 2.0*asin(1.0)
pi = 4.0*atan(1.0)
```

などのように書くことができます．関数計算には時間がかかりますが，π の値は変化しないので計算するのは 1 回だけです．問題になるほどではありません．

1.10.4 コンピュータで表現可能な数値の大きさ

我々が日常的に使っている 10 進数は，10 のべき乗が基本です．すなわち，数字を 10 の多項式で表したときの係数を次数の大きい方から並べたものです．ただし，係数は 10 より小さい整数 (0〜9) でなければなりません．たとえば，1203 という数は，10 の多項式で表せば，

$$1203 = \underline{1} \times 10^3 + \underline{2} \times 10^2 + \underline{0} \times 10^1 + \underline{3} \times 10^0$$

となるので，表現が “1203”なのです．

これに対し，コンピュータ内部で使われている 2 進数は，2 のべき乗を基本として表された数です．この場合，多項式の係数は 2 より小さい整数でなければならないので，0 か 1 です．たとえば，10 進数で 123 と表示される数を 2 の多項式で表せば，

$$123 = \underline{1} \times 2^6 + \underline{1} \times 2^5 + \underline{1} \times 2^4 + \underline{1} \times 2^3 + \underline{0} \times 2^2 + \underline{1} \times 2^1 + \underline{1} \times 2^0$$

となるので，2 進数で表すと，1111011 となります．2 進数は 0 または 1 だけで表すことができるので，電子回路の ON/OFF 回路を使って実現することができます．これがコンピュータで 2 進数が使われている理由です．2 進数の 1 桁をビット (bit) といいます．現在のコンピュータは 2 進数 8 桁 (8bit) を基本に作られていて，これをバイト (byte) といいます．1byte は 8 桁の 2 進数ですから $2^8 = 256$ 個の数字を取り扱うことができます．これを 0 以上の整数と考える場合には，0〜255 となります．

さて，Fortran の整数型は 4byte なので，32 桁 (32bit) の 2 進数です．すなわち，2^{32} 個の数字が取り扱えます．しかし負の整数を含ませるために，整数型の範囲は -2^{31}〜$2^{31}-1$ になります[10]．2^{31} は 2,147,483,648 ですから，取り扱える数の上限はおよそ 21 億です．整数型の計算をするときは，絶対値がこれを超えないようにしなければなりません．

これに対し，実数型数は数値のビットを仮数部 f と指数部 e，および正負を決める符号ビットに分割して，$\pm f \times 2^e$ のような形式で表現します．指数部 e は負数も含んでいるので，10^{-19} のような絶対値の小さい数も表現できます．実数型におけるビット分割とそれによる数の表現には規格がいくつかありますが，現在最もよく使われているのは IEEE 754 と呼ばれている規格です．IEEE 754 における実数型のビット分割数と，それによる表現可能な絶対値の上限を表 1.4 に示します[11]．

表 1.4　実数型のビット分割数と表現可能な絶対値の上限 (IEEE 754)

数値型 (bit)	符号ビット	指数部ビット	指数部 e の範囲	仮数部ビット	絶対値の上限
単精度 (32)	1	8	−126〜127	23	3.4×10^{38} 程度
倍精度 (64)	1	11	−1022〜1023	52	1.7×10^{308} 程度
4 倍精度 (128)	1	15	−16382〜16383	112	1.1×10^{4932} 程度

たとえば，倍精度実数型は仮数部が 52bit なので，$2^{52} \fallingdotseq 4.5\times10^{15}$ 個の区別ができ，これが “有効数字 15 桁程度”の根拠になっています．また，指数部の最大が 1023 なので，最大 10^{308} 程度まで表現することができます．これに対し，単精度は指数部の最大が 127 なので，最大は 10^{38} 程度です．10^{38} のような大きな数字を扱うことはあまりないかもしれませんが，計算の途中で現れる可能性は考慮する必要があります．たとえば，プランク定数は $h \fallingdotseq 6.6\times10^{-34}$ Js なので，h の逆 2 乗が公式の中に入っていると，途中計算で 10^{38} を越えてしまいます．倍精度実数を使う意義はここにもあります．

なお，IEEE 754 での指数部には 2^e という以外にも意味があり，数字の組み合わせによっては，“無限大”や “非数”を表します．もし，`print` 文で数値を出力したときに，Infinity や INF という文字が出力された場合には無限大です．つまり，計算した結果が表現できる上限を超えたという意味です．これを “オーバーフロー”といいます．

これに対し，NaN と出力された場合は非数です．非数 (Not a Number) というのは，「数字じゃない」という意味ですが，具体的には，0 を 0 で割ったときや，−1 の平方根を計算したときの結果がこれに相当します．ただ，結果が非数になっても計算が継続することもあるので，どの時点で発生したのかを特定するのは結構面倒です．

[10] 負の整数のビット表現は 10.4 節参照.

[11] IEEE 754 に関しては「http://ja.wikipedia.org/wiki/浮動小数点数」を参考にしました．詳しくは，この Web ページを見て下さい．なお，表現できる正数の下限は，およそ「1/上限数」程度，という見当で良いと思いますが，IEEE 754 の規格ではもう少し小さい桁の数まで表せます．もっとも，実際にパソコンで上限と下限をチェックしてみたところ，下限は動作環境やコンパイラによって異なりました．また，4 倍精度に関しては上限が倍精度と同じ 10^{308} 程度しかないものもありました．すなわち，必ずしも IEEE 754 の規格通りに実装されているとは限らないようです．

演 習 問 題 1

(1–1)　2 次方程式の解

3 個の実変数 a，b，c に適当な値を代入し，それから，

$$ax^2 + bx + c = 0$$

の 2 解を計算して出力するプログラムを作成せよ．さらに，その解を $ax^2 + bx + c$ に代入して 0 に近い値になることもプログラムして確かめよ．また，a，b，c の値が不適切だと，エラーが出ることも確認せよ．(たとえば，$a = b = c = 1$ にしてみよ)

(1–2)　ヘロンの公式

三角形の 3 辺の長さを a，b，c とすると，その三角形の面積 S は次式 (ヘロンの公式) で表される．

$$S = \sqrt{p\,(p-a)(p-b)(p-c)}$$

ただし，$p = \dfrac{a+b+c}{2}$ である．

a，b，c に適当な数値を与え，この公式を使って三角形の面積を計算するプログラムを作成せよ．さらに，1 辺 a の正三角形の面積を別の方法で計算するプログラムを作成し，ヘロンの公式から求めた面積と一致するかどうか確かめよ．

(1–3)　面積，体積

1 個の実変数 a と整変数 n に適当な数値を与え，これから，半径 a の円の面積 S，半径 a の球の体積 V，1 辺 a の正 n 角形の面積 D を計算するプログラムを作成せよ．ここで，S，V，D は以下の公式で与えられる．

$$S = \pi a^2$$
$$V = \frac{4}{3}\pi a^3$$
$$D = \frac{na^2}{4\tan(\pi/n)}$$

(1–4)　相対論的エネルギー

相対性理論によれば, 質量 m [kg] の物質は, 真空中の光の速度を c (=2.99792458×10^8 [m/s]) とすると, mc^2 [J] の静止エネルギーを持っている．この公式を使って電子と陽子の静止エネルギーを計算するプログラムを作成せよ．ここで電子の質量は 9.10938356×10^{-31} [kg], 陽子の質量は $1.672621898\times10^{-27}$ [kg] である．なお, エネルギーの単位はエレクトロンボルト (eV) に換算して表示せよ．1eV は $1.6021766208\times10^{-19}$ [J] である．

また，物質が速度 v [m/s] で運動しているときの運動エネルギーは $m(\gamma-1)c^2$ である．ここで，$\gamma = \dfrac{1}{\sqrt{1-v^2/c^2}}$ である．電子と陽子が光速の 1%で動いているときの運動エネルギーと，90%で動いているときの運動エネルギーをそれぞれ計算するプログラムを追加せよ．こちらも単位はエレクトロンボルトとする．

(1–5)　複素数の n 乗根

複素数 z の n 乗根は，次のようにして計算することができる．まず，z の絶対値を r，偏角を θ とすると，

$$z = r(\cos\theta + i\sin\theta)$$

と表すことができるので，z の n 乗根の一つは，

$$z^{1/n} = r^{1/n}w_1, \qquad \text{ただし，} w_1 = \cos\frac{\theta}{n} + i\sin\frac{\theta}{n}$$

である．この w_1 を使って，$r^{1/n}w_1$，$r^{1/n}w_1^2$，$r^{1/n}w_1^3$，$\cdots$，$r^{1/n}w_1^n$，が z の n 乗根となる．

たとえば，2 の 5 乗根および 10 の 10 乗根をそれぞれ全て計算せよ．

(1–6)　複素インピーダンスと複素電流

交流回路理論によれば，複素インピーダンス Z [Ω] の回路に周波数 f [Hz] の交流電圧 E [V] を加えたとき，流れる複素電流 I [A] は，

$$I = \frac{E}{Z}$$

という複素数の演算を使って計算することができる．

たとえば，R [Ω] の抵抗，インダクタンス L [H] と抵抗 R_L [Ω] を持つコイル，キャパシタンス C [F] を持つコンデンサという 3 種類の回路素子の複素インピーダンスは，それぞれ，

$$Z_R = R, \qquad Z_L = R_L + i\omega L, \qquad Z_C = \frac{1}{i\omega C}$$

と計算することができる．ここで，$\omega = 2\pi f$ である．

この交流回路理論を用いて，R=100 [Ω] の抵抗，L=0.04 [mH] で R_L=10 [Ω] のコイル，C=100 [μF] のコンデンサという 3 個の回路素子を直列に接続した回路に，f=60 [Hz]，E=100 [V] の交流電圧を加えたときに流れる複素電流 I_s [A] を計算せよ．また，同じ 3 個の回路素子を並列に接続した回路に同じ交流電圧を加えたときに流れる複素電流 I_p [A] も計算せよ．なお，プログラムは複素数型を用いて作成すること．また，π の値はできるだけ正確な値を使うこと．

ここで，複素インピーダンス Z_1，Z_2，Z_3 を持つ 3 個の回路素子を直列につないだときの合成複素インピーダンスを Z_s とすれば，

$$Z_s = Z_1 + Z_2 + Z_3$$

である．また，同じ 3 個を並列につないだときの合成複素インピーダンスを Z_p とすれば，

$$\frac{1}{Z_p} = \frac{1}{Z_1} + \frac{1}{Z_2} + \frac{1}{Z_3}$$

である．

さらに，得られた複素電流を用いて直列接続と並列接続それぞれの回路における有効電力 P [W] と力率 $\cos\theta$ を計算するプログラムを追加せよ．ここで，有効電力 P は複素電力 EI の実数部であり，力率 $\cos\theta$ は次式で計算することができる．

$$\cos\theta = \frac{P}{|EI|}$$

第 2 章　配列と手順のくり返し

前章で説明した基本的プログラミングは，単純な計算動作を並べたものであり，言ってみれば関数電卓の操作をプログラムという書式で表しただけに過ぎません．コンピュータの能力を使いこなすには，同じような動作を何度もくり返したり，条件に応じて異なる動作をさせる手法の知識が必要です．本章では複数の数値をまとめて表現するための配列と，それを有効に利用するためのくり返し動作の書き方ついて説明します．



2.1　配列

これまで利用してきた“変数”は，型に応じた 1 個の数値を記憶することしかできませんでした．しかし，数値計算やコンピュータシミュレーションでは，数万個のデータを保存して，それぞれを条件に応じて変化させていく，なんていうのが当たり前のように行われます．そこで，数学で $a_1, a_2, \cdots, a_n$ のように変数に添字を付けて区別するように，数字で区別した変数を作ることができます．これを“配列”といいます．本節では配列の使い方について説明します．

2.1.1　配列宣言

配列は変数の一種なので，型宣言文を使って宣言しておかねばなりません．単一変数と異なるのは，宣言時に添字の範囲を示す整数値をかっこを使って付加することです．たとえば，次のように宣言します．

```
real a(10),b(20,30)
complex cint(10,10)
integer node(100)
```

ここで，a や node のように数字が 1 個の配列を 1 次元配列，b や cint のように数字が 2 個の配列を 2 次元配列といいます．3 次元以上の配列を作ることもできます．配列の名前の付け方，頭文字の選択などの規則は単一変数名の付け方と同じにします．宣言した配列の各部の名称を表 2.1 に示します．

表 2.1　配列の各部の名称

real a(10) と宣言した配列について	
記　号	名　称
a	配列名
a(3) など	配列要素
かっこ中の数字	要素番号，添字

配列宣言で指定した数値は要素番号の最大値を表し，要素番号の使用可能範囲は 1 から指定した最大値までです．たとえば a(10) の宣言では a(1) から a(10) までの 10 個の配列要素が使用可能になります．また，2 次元以上の配列の場合には各次元ごとの最大値を指定しているので，b(20,30) の宣言では全部で $20 \times 30 = 600$ 個の配列要素が使用可能になります．

問題によっては，要素番号として 0 や負数を使いたいときがあります．このようなときには“:”を間に入れて，使用可能な要素番号の最小値と最大値を同時に指定します．たとえば，

```
real ac(-3:5),bc(-20:20,0:100)
```

と宣言すると，1 次元配列 ac は，ac(-3) から ac(5) までの 9 個が使用可能であり，2 次元配列 bc は，bc(-20,0) から bc(20,100) までの $(20 \times 2 + 1) \times (100 + 1) = 4141$ 個が使用可能です．

2.1.2　メモリ上での配列の並び

配列はコンピュータ内部において連続したメモリ領域で実現されています．たとえば，real a(10) と宣言された配列は，図 2.1 左のように，a(1),a(2),...,a(10) の順で並んだ実数型のメモリです．



図 2.1　1 次元配列のメモリ並び

図からわかるように，a(3) とは a(1) から数えて 3 番目のメモリのことです．また，real a(10) の宣言では 10 個しかメモリを確保していないのですから，a(10000) のように範囲外の要素番号を指定すると問題が起こります．10000 番目の要素 a(10000) がどのメモリを示すのか不明だし，そもそも存在するという保証さえありません．

Fortran での配列名は配列を代表する名称であると同時に，配列の先頭要素を示します．たとえば，配列名 a は図 2.1 左のように a(1) を示します．また，ac(-3:5) のように最小値も指定して宣言した場合には，図 2.1 右のように並んでいて，配列名 ac は ac(-3) を示します．配列名が先頭要素を示すことは配列をサブルーチンの引数として与えるときに意識する必要があります．

2 次元以上の配列の場合は，左の方の要素番号から先に進むようにメモリ上で並んでいます．たとえば，

```
real b(3,2)
```

と宣言した場合，メモリ上での並びは図 2.2 のようになります．2 次元以上の配列も配列名は先頭要素を示しています．配列 b の場合，配列名 b は b(1,1) を示します．



図 2.2　2 次元配列のメモリ並び

2 次元と 3 次元の配列の並び方を式で表せば，

```
real b(m,n) の配列宣言で，b(i,j) は先頭から数えて i + m*(j-1) 番目
real b(ℓ,m,n) の配列宣言で，b(i,j,k) は先頭から数えて i + ℓ*(j-1 + m*(k-1)) 番目
```

となります．どちらの公式も一番右側の要素数 n に依存していません．一般的に，どんな次元の配列の順番を与える公式も一番右側の要素数には依存しません．

2.2 手順のくり返し — do 文

実行文は基本的に上から下へ順に実行されますが，それだけでは類似した手順をくり返すときに，必要な回数だけ同じような文を書かねばなりません．そこで，ある範囲の手順を必要な回数だけくり返し行わせる手段として do 文があります．配列を使って大量のデータを取り扱う場合，a(1)=...，a(2)=...，のように要素ごとの実行文を書くのは大変です．そこで，計算のパターンがわかっている場合には，do 文を使ってそのパターンをくり返す形で配列の計算を行うことができます．

do 文を使うときの基本形は，

```
do 整数型変数 = 初期値，終了値
  .......
  .......
enddo
```

です．最初の do の行が do 文で，do 文と最後の enddo 文で範囲を指定した一連の実行文がくり返し実行されます．この範囲を“do ブロック”といいます．また，プログラムの流れが循環するという意味で“do ループ”ともいいます．do 文の整数型変数を，本書では“カウンタ変数”と呼びます．do ブロックのくり返しは，カウンタ変数に“初期値”を代入することから開始し，do ブロック内の手順を 1 回実行するごとにカウンタ変数を 1 増加します．そして，カウンタ変数が“終了値”より大きくなった時点でくり返しは終了し，enddo 文の次の文に実行が移ります．たとえば，

```
do m = 1, 10
  a(m) = m
enddo
```

と書けば，a(1)=1，a(2)=2，...，a(10)=10 という 10 個の代入文を実行したことになります．do 文における，“カウンタ変数＞終了値”の判定は do ブロックの開始時にも行うため，初期値が終了値より大きい場合にはブロック内部が一度も実行されずに enddo 文の次に実行が移ります．

次の do 文のように，整数値をもう一つ追加することで増加量を指定することもできます．

```
do 整数型変数 = 初期値，終了値，増分値
  .......
  .......
enddo
```

このときカウンタ変数は初期値から開始して，“増分値”ずつ増加しながら do ブロック内の手順をくり返し，“カウンタ変数＞終了値”の時点で終了します．

たとえば，m が 10 以下の奇数のときのみ計算をしたいときは，

```
do m = 1, 10, 2
  .......
  .......
enddo
```

と書きます．このときの終了値は 10 ですが，10 は奇数ではないので計算しません．

増分値は負数を指定することもできます．負数のときにはカウンタ変数が減少していくので，“カウンタ変数＜終了値”になった時点で終了します．たとえば，100 から順に下って 1 までくり返すときには

以下のように書きます.

```
do m = 100, 1, -1
  .......
  .......
enddo
```

増分値が負数のときには，“初期値＜終了値”ならば do ブロック内部は一度も実行されません.

do 文のカウンタ変数に増分値を加えるタイミングは，enddo 文の実行時です．たとえば，

```
do m = 1, 3
  a(m) = m**2
enddo
```

という文は,

```
 m = 1
   [m＞3 の判定をする．満足しないので，do ブロックを実行]
 a(m) = m**2
 m = m + 1
   [m＞3 の判定をする．満足しないので，do ブロックを実行]
 a(m) = m**2
 m = m + 1
   [m＞3 の判定をする．満足しないので，do ブロックを実行]
 a(m) = m**2
 m = m + 1
   [m＞3 の判定をする．満足するので，do ブロックを終了]
```

という動作になります．この展開からわかるように，do ブロックを終了した時点で，カウンタ変数には終了値より大きい値が代入されています．上例では，do ブロック終了後，m＝4 です．do ブロック終了時の m の値を利用するときは，このことを考慮しなければなりません.

次のように，do ブロックの中に別の do ブロックを入れて多重にすることもできます．ただし，カウンタ変数は異なるものを使わなければなりません．これは，do ブロック内でカウンタ変数を変更することが禁止されているからです.

```
do k = 1, 100
  a(k) = k**2
  do m = 1, 10
    b(m,k) = m*a(k)**3
    c(m,k) = b(m,k) + m*k
  enddo
  d(k) = a(k) + c(10,k)
enddo
```

よく使うので覚えておくと便利なのが，合計を計算する do 文です．たとえば，n 要素の 1 次元配列，a(1), a(2), ..., a(n) の中に数値データが入っている場合，これらの合計 sum を求めるには do 文を使って以下のように書きます.

```
sum = 0
do m = 1, n
  sum = sum + a(m)
enddo
```

この文が合計を計算していることを理解するには, やはり以下のように手動展開してみると良いでしょう.

```
sum = 0
m = 1
sum = sum + a(m)
m = m + 1
sum = sum + a(m)
m = m + 1
sum = sum + a(m)
m = m + 1
 ......
```

ここで判定文は省略しました．このプログラムで重要なことは，do 文の前で変数 sum に 0 を代入していることです．これがないと正しい結果が得られないことがあります．このようなプログラムならではの書き方はパターンとして覚えておくと良いと思います．

なお，カウンタ変数の初期値，終了値，増分値には変数や計算式を記述することもできますが，それらの値は do ブロックの開始時に評価・決定されるので，do ブロック内部で修正があっても無関係です．たとえば，

```
j = 10
k = 1
do i = 1, j, k
   print *,i,j,k
   j = 20
   k = 2
enddo
```

のように書いてもエラーではありませんが，終了値 j や増分値 k は最初の do 文を実行した時点の数値でくり返し，終了します．

2.3 プログラミング スマートテクニック (その 2)

本節では do ループを使うときのテクニックや，プログラムを読みやすくするためのヒントなどを紹介します．配列を使った本格的なプログラムが書けるようになると，「文法的に間違いがなければ良い」という考えで書くのではなく，読みやすいプログラムになるよう心がける必要があります．ここで，“読みやすいプログラム”というのは，チェックがしやすいプログラムのことです．読みやすく書いておけば，書き間違いに気付きやすいし，実行時にエラーが出ても早期に誤りを発見することができます．以下のヒントは，プログラミングの初心者が読むと「細かいことだから無視してもいいや」と感じるかもしれませんが，慣れればそれほど手間ではありません．面倒がらずに心がけましょう．

2.3.1 do ループのテクニック

do ブロック内部に同じ計算をくり返す記述があるときには，あらかじめ計算をしておきます．たとえば，

```
do i = 1, 10000
  a(i) = b(i)*c*f**2
  b(i) = sin(x)*a(i)
enddo
```

と書くと，くり返しごとに c*f**2 や sin(x) を計算することになりますが，これらは do ブロック内部では変化しないのですから，あらかじめ計算しておくと速くなります．たとえば，

```
cf = c*f**2
sx = sin(x)
do i = 1, 10000
  a(i) = b(i)*cf
  b(i) = sx*a(i)
enddo
```

のように書き換えると速くなります．

また，do ブロック内で何度も同じ割り算をするときには，逆数を掛けるように書き換えると速くなります．これは，割り算が遅いからです．たとえば，

```
do i = 1, 10000
  a(i) = b(i)/c
  x(i) = a(i)/10.0
enddo
```

と書くより，

```
d = 1.0/c
do i = 1, 10000
  a(i) = b(i)*d
  x(i) = a(i)*0.1
enddo
```

と書く方が速くなります．もっとも，プログラムがわかりにくくなるという欠点もあるので，割り算の箇所が少なく，くり返し回数がさほど多くないときにはそれほど神経質に変形する必要はないと思います．

配列を使ったくり返し計算をするときは，メモリをできるだけ連続的に読み書きする方が速くなります．このため，多重 do ブロックを使って 2 次元以上の配列計算をするときは，左の要素から先に進めるようにします．たとえば，

```
real b(10,100)
integer m,n
do n = 1, 100
  do m = 1, 10              !  左の要素を先に進めると速い
    b(m,n) = m*n
  enddo
enddo
```

のように，2 次元配列 b(m,n) に計算結果を代入するときは，左側の要素 m に関する do ブロックを右側の要素 n の do ブロックの内側に持ってくる方が高速です．これは，内側の do ブロックのカウンタ変数が先に進むので，b(1,1),b(2,1),b(3,1)…のように，メモリの並んでいる順に格納していくからです．

これを，

```
real b(10,100)
integer m,n
do m = 1, 10
  do n = 1, 100             !  右の要素を先に進めると遅い
    b(m,n) = m*n
  enddo
enddo
```

のように書くと，b(1,1),b(1,2),b(1,3)…のように飛び飛びに格納していくので効率が悪くなり，スピードダウンします．

大きな数に小さい数を加えると，小さい数が桁落ちして情報が失われる可能性があります．たとえば，1 次元配列 a(i) の合計 sum を計算するプログラムは，

```
sum = 0
do i = 1, n
  sum = sum + a(i)
enddo
```

で良いのですが，a(i) が全て正数で，n が非常に大きい場合には，合計 sum がだんだん大きくなっていくので，後の方で加えた a(i) の情報が失われる可能性があります．倍精度実数を使うことを推奨している理由の一つがこれです．この桁落ちを防ぐには，たとえば，2 個ずつ加えてそれらを別の配列に入れ，その後それらを同様に 2 個ずつ加えていって，... とするのが良いのですが，プログラムが複雑になってしまいます．

ひとつの比較的簡単な解決法は，補助変数を使って，誤差を別に評価しておく方法です [2].

```
sum = 0
res = 0
do i = 1, n
  res = res + a(i)
  tmp = sum
  sum = sum + res
  tmp = sum - tmp
  res = res - tmp
enddo
```

この方法では，桁落ち分を補助変数 res に保存して別途加えるので，計算精度が上がります．倍精度計算をしていれば必ずしもこの方法にする必要はありませんが，精度の良い合計計算が必要になったときのために覚えておくと良い方法です．

2.3.2 多項式を計算する手法

多項式を計算するときは，掛け算の回数ができるだけ少なくなるように考えます．一般的に良い計算手法は horner 法です．たとえば，

$$y = a_0 + a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4$$

を，乗算を明示してプログラムにすると，

```
y = a0 + a1*x + a2*x*x + a3*x*x*x + a4*x*x*x*x
```

のようになり，10 回の乗算が必要です．ところが，これを変形して，

```
y = a0 + (a1 + (a2 + (a3 + a4*x)*x)*x)*x
```

と書けば，4 回の乗算で計算できます．これが horner 法です．

一般的に，n 次の多項式，

$$y = a_0 + a_1 x + a_2 x^2 + \cdots + a_n x^n$$

における係数 $a_0, a_1, a_2, ..., a_n$ が，下限が 0 の 1 次元配列，a(0),a(1),a(2),...,a(n) にそれぞれ代入されている場合には，次のような do 文を使って y を計算することができます．

```
y = a(n)
do i = n-1, 0, -1
  y = a(i) + x*y
enddo
```

なお，$x^8 + 1$ は $((x^2)^2)^2 + 1$ のように変形すれば 3 回の乗算で計算できます．すなわち，多項式によっては horner 法より速く計算できる手順が存在することもあります．

2.3.3 ブロックを明確にするために字下げする

これまでのプログラム例でも示していますが，do や 3.1 節で説明する if などのブロック中では字下げ (インデント) をしましょう．これは，ブロックの範囲が一目でわかるからです．たとえば，

```
do i = 1, n
  b(i) = a(i)
  c(i) = a(i)*sin(b(i))
enddo

do i = 1, n
  do k = 1, 10
    d(k,i) = k*b(i)
  enddo
enddo
```

のように，do ブロック内部の文をスペースを入れて右にずらせておくとブロックの範囲が一目で判断できます．通常，2〜4 個のスペース程度で良いと思います．

最近のエディタには，“オートインデント”といって，改行したときにその前の行と先頭がそろうようにカーソルが移動する機能があるので，インデントをするのはそれほどの手間ではありません．

2.3.4 文字間や行間は適当に空ける

文中の文字間は適当に空けたほうが読みやすくなります．筆者は “=”と “+”の両側に必ずスペースを入れることにしています．また，代入文が何行も続くときには “=”を上下にそろえると読みやすくなります．特に，同じパターンで少しずつ内容が違う文を並べるときには，パターンが並ぶようにスペースを入れることをお勧めします．以下は，筆者のシミュレーションプログラムの一部です．

```
wm( 1,m1) = (1-dx)*(1-dy)*(1-dz)
wm( 2,m1) =    dx *(1-dy)*(1-dz)
wm( 3,m1) = (1-dx)*   dy *(1-dz)
wm( 4,m1) =    dx *   dy *(1-dz)
wm(11,m1) = (1-dx)*(1-dy)*   dz
wm(12,m1) =    dx *(1-dy)*   dz
wm(13,m1) = (1-dx)*   dy *   dz
wm(14,m1) =    dx *   dy *   dz
```

これをスペース抜きで書くと，以下のようになります．

```
wm(1,m1)=(1-dx)*(1-dy)*(1-dz)
wm(2,m1)=dx*(1-dy)*(1-dz)
wm(3,m1)=(1-dx)*dy*(1-dz)
wm(4,m1)=dx*dy*(1-dz)
wm(11,m1)=(1-dx)*(1-dy)*dz
wm(12,m1)=dx*(1-dy)*dz
wm(13,m1)=(1-dx)*dy*dz
wm(14,m1)=dx*dy*dz
```

この二つは全く同じことが書いてあるのですから全く同じ結果を出します．しかし，パターンをそろえた前者は単なる美的趣味でスペースを入れたのではありません．この例では，どこか 1 カ所，たとえば dx を dy に書き間違えただけでも正しい結果が得られません．しかも，dx と dy を書き間違えても文法的なミスにはならず，出力結果を見ても間違いに気づかない可能性があります．よって，できるだけプログラムを書いている段階でミスを取り除いておかねばなりません．

このとき，前者のようにパターンをそろえておけば，横方向に 1 行ずつチェックするだけでなく縦方向にも比較しながらチェックすることができます．色々な角度からチェックすることで，ミスのないプログラムにすることができるのです．

この他，文と文の間に適当なコメント文や何も書いていない行を挿入すれば，プログラムの流れに区切りができて読みやすくなります．これは文章を段落に分けるように，プログラムを区分けすると考えればいいでしょう．

2.3.5 定数は意味のある名前をつけた変数に代入して使う

数学的に決まっている単純な定数 (2 倍するとか，1 を加えるとか，3 乗するとか) の場合以外は，できるだけ定数の使用量を減らすほうがいいと思います．プログラムを書いているときには理解していても時間が経ってから読んだときに意味がわからなくなる可能性があるからです．

たとえば，電子の質量 m_e ($= 9.10938356\times10^{-31}$ kg) と光の速度 c ($= 2.99792458\times10^{8}$ m/s) を使って計算すれば $m_ec^2 = 8.1871056497\times10^{-14}$ J ですが，これを単位とするエネルギー，$energy/m_ec^2$ を計算するプログラムを，

```
ene = energy/8.1871056497e-14
```

と書くと，後で $8.1871056497\times10^{-14}$ という数字がどこから来たのかわからなくなる可能性があります．こういうときには多少面倒でも，

```
me  = 9.10938356e-31              !  電子の質量
cl  = 2.99792458e8                !  真空中の光速
mc2 = me*cl**2
ene = energy/mc2
```

としておけば，わかりやすいしプログラム内容の記録にもなります．

また，do ブロックを 100 回くり返す，とか，10 回ごとに出力する，とかいう制御数も，変数にしておけば数値の意味が明示されるし，変更もしやすくなります．たとえば，

```
do i = 1, 100                     !  100 が 1 個
  a(i) = b(i)**2
enddo
sum = 0
do i = 1, 100                     !  100 が 1 個
  sum = sum + a(i)
enddo
ave = sum/100                     !  100 が 1 個
```

というプログラムにおいて，100 という数は配列要素数に依存する共通の数値ですから，以下のように変数にします．

```
imax = 100                        !  100 を変更するときはここだけで OK
do i = 1, imax
  a(i) = b(i)**2
enddo
sum = 0
do i = 1, imax
  sum = sum + a(i)
enddo
ave = sum/imax
```

imax は「i の最大値」という意味を込めた名前です．こうしておけば，100 を 200 に変更したいときには変更点が 1 カ所ですみます．このような変数化はプログラムが長くなるほど価値が増します．

演 習 問 題 2

(2–1)　くり返し出力

i=1,2,3,...,n の整数に対し，i，i^2，$1/i$，$\sqrt{i}$ の値を並べて出力するプログラムを作成せよ．なお，整数と実数の変換に注意すること．

(2–2)　3 次元ベクトルと電磁気学

3 次元ベクトル $\boldsymbol{A}=(A_1,A_2,A_3)$ を要素数 3 の 1 次元配列 `a(3)` で表すとする．まずベクトル $\boldsymbol{A}$ とベクトル $\boldsymbol{B}$ を配列で表してそれぞれに適当な数値を代入し，2 個の配列から，$\boldsymbol{A}$ と $\boldsymbol{B}$ の内積 $\boldsymbol{A}\bullet\boldsymbol{B}$，および外積ベクトル $\boldsymbol{A}\times\boldsymbol{B}$ を計算するプログラムを作成せよ．外積ベクトルも配列にすること．ただし，ベクトル $\boldsymbol{A}$ と $\boldsymbol{B}$ の内積，および外積は以下の公式で与えられる．

$$\boldsymbol{A}\bullet\boldsymbol{B}=A_1B_1+A_2B_2+A_3B_3$$
$$\boldsymbol{A}\times\boldsymbol{B}=(A_2B_3-A_3B_2,A_3B_1-A_1B_3,A_1B_2-A_2B_1)$$

(2–3)　2 行 2 列の行列計算

2 行 2 列の行列，$\mathbf{A}=\begin{pmatrix}a_{11}&a_{12}\\a_{21}&a_{22}\end{pmatrix}$ を，2 次元配列 `a(2,2)` で表すとする．2 個の 2 行 2 列の行列 $\mathbf{A}$ と $\mathbf{B}$ を表現する配列，`a(2,2)` と `b(2,2)` の各要素に適当な数値を代入して，$\mathbf{A}$ と $\mathbf{B}$ の足し算の行列 $\mathbf{C}=\mathbf{A}+\mathbf{B}$，掛け算の行列 $\mathbf{M}=\mathbf{AB}$，および，それぞれの行列式 D_A，D_B を計算するプログラムを作成せよ．ここで，足し算，掛け算の公式は以下で与えられる．

$$\begin{pmatrix}a_{11}&a_{12}\\a_{21}&a_{22}\end{pmatrix}+\begin{pmatrix}b_{11}&b_{12}\\b_{21}&b_{22}\end{pmatrix}=\begin{pmatrix}a_{11}+b_{11}&a_{12}+b_{12}\\a_{21}+b_{21}&a_{22}+b_{22}\end{pmatrix}$$
$$\begin{pmatrix}a_{11}&a_{12}\\a_{21}&a_{22}\end{pmatrix}\begin{pmatrix}b_{11}&b_{12}\\b_{21}&b_{22}\end{pmatrix}=\begin{pmatrix}a_{11}b_{11}+b_{12}b_{21}&a_{11}b_{12}+b_{12}b_{22}\\a_{21}b_{11}+b_{22}b_{21}&a_{21}b_{12}+b_{22}b_{22}\end{pmatrix}$$

また，行列 $\mathbf{A}$ の行列式は，$D_A=a_{11}a_{22}-a_{12}a_{21}$ で与えられる．

(2–4)　統計計算

1 次元配列，`a(n)` に，1,2,3,... のデータを順に n 個代入し，

$$\text{平　均 } \bar{A}=\frac{1}{n}\sum_{i=1}^{n}A_i$$
$$\text{標準偏差 } \sigma=\sqrt{\frac{1}{n}\sum_{i=1}^{n}(A_i-\bar{A})^2}$$

を計算するプログラムを作成せよ．

次に，`a(n)` に 1,3,5,... と奇数を順に n 個入れて，同様に平均と標準偏差を計算するプログラムを追加せよ．

(2–5)　漸化式

次の漸化式で計算される数値を，それぞれ，配列 `a(n)`，`b(n)` に代入し，それぞれの平均と標準偏差を計算せよ．

$$(1)\ a_{i+1} = 3a_i + 2, \qquad a_1 = -1$$
$$(2)\ b_{i+1} = \frac{2}{3}b_i - 4, \qquad b_1 = 3$$

(2–6)　常微分方程式 (初期値問題) の解法：サイクロトロン運動

一様な磁場中での荷電粒子の速度 $\boldsymbol{v}(t) = (u(t), v(t))$ は次の運動方程式を満足する．この運動方程式をオイラー法で解け．ただし，サイクロトロン周波数 ω_c は一定とする．

$$\frac{du}{dt} = \omega_c v$$
$$\frac{dv}{dt} = -\omega_c u$$

ここで，時間 t の関数 $f(t)$ に対して，1 回に進める時間ステップを Δt として $f_n = f(n\Delta t)$ としたとき，オイラー法とは n ステップ目の値 f_n と $n+1$ ステップ目の値 f_{n+1} を使って $\frac{df}{dt} \fallingdotseq \frac{f_{n+1} - f_n}{\Delta t}$ という近似で時間微分を評価する方法である．たとえば，x 方向は

$$\frac{u_{n+1} - u_n}{\Delta t} = \omega_c v_n$$

と近似できるから，これを変形して，$u_{n+1} = u_n + \omega_c v_n \Delta t$ となる．v_{n+1} も同様に変形すれば，オイラー法とは次の連立の漸化式を使って (u_n, v_n) から (u_{n+1}, v_{n+1}) を計算することである．

$$u_{n+1} = u_n + \omega_c v_n \Delta t$$
$$v_{n+1} = v_n - \omega_c u_n \Delta t$$

1 周期 $\frac{2\pi}{\omega_c}$ を適当な回数 N で分割して Δt を決定し，初期条件 $u_0 = 1$, $v_0 = 0$ から出発して，オイラー法で N 回計算するようなプログラムを作成せよ．そして，その結果得られる 1 周期後の速度，u_N，v_N と u_0，v_0 を比較せよ．また，10 周期後の速度，u_{10N}，v_{10N} とも比較せよ．

次に，オイラー法を少し修正したシンプレクティック法，(v_{n+1} の式に注意！)

$$u_{n+1} = u_n + \omega_c v_n \Delta t$$
$$v_{n+1} = v_n - \omega_c u_{n+1} \Delta t$$

のプログラムを作成し，オイラー法の結果と比較せよ．

なお，以上のプログラムは配列を使わないで作成すること．

(2–7)　常微分方程式 (境界値問題) の解法：1 次元ポワソン方程式

電荷密度 $\rho(x)$ が与えられているときに電位が満足する 1 次元ポワソン方程式

$$\frac{d^2V(x)}{dx^2} = -\rho(x)$$

を解くには，以下のようにする．

等間隔 h ごとに並んだ x 座標点 $x_i = hi, (i = 0 \sim N)$ に対して，$V_i = V(x_i)$，$\rho_i = \rho(x_i)$ として，上式の左辺を差分で近似すれば，

$$\frac{V_{i+1} - 2V_i + V_{i-1}}{h^2} = -\rho_i$$

となる．これを座標点に関して書き下せば，

$$\begin{aligned} V_0 - 2V_1 + V_2 &= -\rho_1 h^2 \\ V_1 - 2V_2 + V_3 &= -\rho_2 h^2 \\ V_2 - 2V_3 + V_4 &= -\rho_3 h^2 \\ &\vdots \\ V_{N-2} - 2V_{N-1} + V_N &= -\rho_{N-1} h^2 \end{aligned}$$

になるが，両端の電位 $V_0 = V(x_0)$，$V_N = V(x_N)$ が与えられていれば，これらは V_1，V_2，$\cdots$，V_{N-1} に対する連立 1 次方程式になる．

一般に，連立 1 次方程式，

$$a_i V_{i-1} + b_i V_i + c_i V_{i+1} = d_i \qquad (i = 1, 2, \ldots, N-1)$$

を V_0 と V_N を与えて解くときには，まず，$G_0 = 0$，$H_0 = V_0$ から出発して，

$$G_i = -\frac{c_i}{b_i + a_i G_{i-1}}, \qquad H_i = \frac{d_i - a_i H_{i-1}}{b_i + a_i G_{i-1}} \qquad (i = 1, 2, \ldots, N-1)$$

を計算し，次に，V_i を以下の式で計算する (計算の方向に注意！)．

$$V_i = G_i V_{i+1} + H_i \qquad (i = N-1, N-2, \ldots, 1)$$

以上の方法を用いて，次の 2 種類の電荷密度と境界条件における電位を計算するプログラムを作成せよ．

(1) $\rho(x) = 0$ 境界条件は，$V_0 = 0$，$V_N = 1$

(2) $\rho(x) = \sin\left(\dfrac{2\pi x}{Nh}\right)$ 境界条件は，$V_0 = 0$，$V_N = 0$

(2–8)　テーラー展開

テーラー展開の公式から指数関数や三角関数を計算するプログラムを作成せよ．ただし，第 0 項から第 n 項までの指数関数と三角関数のテーラー展開の係数は以下の通りである．

$$\begin{aligned} e^x &= 1 + \frac{x}{1!} + \frac{x^2}{2!} + \frac{x^3}{3!} \cdots + \frac{x^n}{n!} &&= \sum_{i=0}^{n} \frac{x^i}{i\,!} \\ \cos x &= 1 - \frac{x^2}{2!} + \frac{x^4}{4!} \cdots + (-1)^n \frac{x^{2n}}{(2n)!} &&= \sum_{i=0}^{n} (-1)^i \frac{x^{2i}}{(2i)!} \\ \sin x &= x - \frac{x^3}{3!} + \frac{x^5}{5!} \cdots + (-1)^n \frac{x^{2n+1}}{(2n+1)!} &&= \sum_{i=0}^{n} (-1)^i \frac{x^{2i+1}}{(2i+1)!} \end{aligned}$$

次数 n と変数値 x は複数個選んで計算し，組み込み関数で計算した値と比較して，n が大きくなるほど正確な関数値に近づくこと，x が大きくなるほど近似が悪くなることなどを確かめよ．

第 3 章　条件分岐とジャンプ



手順のくり返しを記述する do 文と並んで重要なのが，条件に応じて異なる動作をさせるための if 文です．do 文と if 文の使い方を習得すれば，どんなプログラムでも書けるといっても過言ではありません．本章では，if 文による条件分岐の書き方と，必要に応じて実行の流れを強制的に変えるジャンプについて説明します．

3.1　条件分岐 — if 文

条件に応じて異なる手順を行わせることを“条件分岐”といい，if 文を使って指定します．最も単純に，一つの条件に応じて一つの文を実行するかしないかを決めるだけのときは単純 if 文を使います．単純 if 文は以下の形式です．

```
if（条件）実行文
```

単純 if 文はかっこ内の“条件”が満足されれば，その右の“実行文”を実行し，条件が満足されなければ何もしないで次の文に実行が移ります．たとえば，

```
a = 5
if (i < 0) a = 10
b = a**2
```

と書くと，i < 0 の場合には a = 10 となり，それ以外は a = 5 のままなので，それに応じて b に代入される値が異なります．

しかし単純 if 文には実行文が一つしか書けないので，実行したい文が複数あるときには使えません．また，条件に合ったときの動作指定しかできないので，合わなかったときの動作を別に指定したいときには不便です．そこで，通常は単純 if 文ではなく，ブロック if 文を使います．ブロック if 文とは，if 文の実行文のところを then にした文のことで，以下のようにブロック if 文と endif 文で一連の実行文の範囲を指定します．

```
if（条件）then
  .......
  .......
endif
```

この指定された範囲を“if ブロック”といい，ブロック if 文に書かれた条件を満足したときのみ，if ブロック内の実行文が実行されます．たとえば，

```
a = 5
b = 2
if (i < 0) then
   a = 10
   b = 6
endif
c = a*b
```

と書くと，i < 0 の場合には a=10，b=6 の代入文を実行してから c=a*b を計算しますが，それ以外，すなわち i ≧ 0 の場合には if ブロック内を実行しないので，a も b も変化せず，a=5，b=2 のままで c=a*b を計算します．

さて，この例においては，a=5，b=2 という代入は i < 0 という条件を満足するときには不要で，満足しないときのみ実行すれば十分です．このように“条件を満足しない場合”に別の動作をさせたいときには if ブロック内に else 文を挿入します．else 文を挿入すると，ブロック if 文で指定した条件を満足しない場合に，else 文から endif 文までの実行文が実行されます．たとえば上記のプログラムは，

```
if (i < 0) then
   a = 10
   b = 6
 else
   a = 5
   b = 2
endif
c = a*b
```

と書くことができます．この if ブロックでは，i < 0 の場合には a=10，b=6 を実行，それ以外の場合には a=5，b=2 を実行します．

また，条件を満足しない場合に，さらに別の条件を指定したいときには else if 文を使います．else if 文も条件はかっこで指定し，その後に then を書きます．

```
if (i < 0) then
   a = 10
   b = 6
 else if (i < 5) then
   a = 4
   b = 7
 else
   a = 5
   b = 2
endif
c = a*b
```

この場合，i < 0 の場合には a=10，b=6 を実行，0 ≦ i < 5 の場合には a=4，b=7 を実行，それ以外 (i ≧ 5) の場合は a=5，b=2 を実行，となります．else if 文による新たな条件は if ブロック内にいくつ入れてもかまいません．その場合は，“その else if 文より以前の条件を全て満足しない場合に，その条件を満足すれば”という意味になります．これに対し，else 文は最後の 1 回しか使えません．

数値計算でよく使う代表的な比較条件の書き方を表 3.1 に示します．

表 3.1　比較条件の書き方

比較条件記号	記号の意味	使用例
==	左辺と右辺が等しいとき	x == 10
/=	左辺と右辺が等しくないとき	x+10 /= y-5
>	左辺が右辺より大きいとき	2*x > 1000
>=	左辺が右辺以上のとき	3*x+1 >= a(10)**2
<	左辺が右辺より小さいとき	sin(x+10) < 0.5
<=	左辺が右辺以下のとき	tan(x)+5 <= log(y)

使用例のように，比較条件を指定する場合には両辺どちらにも計算式を書くことが可能です．

さらに表 3.2 の論理演算記号を使えば，これらの条件を論理的につないだ条件や，否定した条件を与えることもできます．

表 3.2　論理式の書き方

論理演算記号	演算の意味	使用例
“条件 1”.or.“条件 2”	“条件 1”または“条件 2”のとき	x > 0 .or. y > 0
“条件 1”.and.“条件 2”	“条件 1”かつ“条件 2”のとき	x > 10 .and. 2*x <= 50
.not.“条件”	“条件”を満足しないとき	.not.(x < 0 .and. y > 0)

たとえば,

```
if (i > 0 .and. i <= 5) then
   a = 10.0
 else
   a = 0.0
endif
```

と書くと，i が 0 より大きく“かつ”5 以下のとき，すなわち，$0 < i \leqq 5$ のときに a = 10，それ以外は a = 0 となります．なお，横着してこのブロック if 文を,

```
if (0 < i <= 5) then      !  これはエラー
```

と書くことはできないので注意しましょう．

論理演算は，数値演算のように連ねて書くことも可能ですが，その際は表 3.2 の下に行くほど優先順位が高くなります．

```
      .not.  ＞  .and.  ＞  .or.
```

たとえば,

```
if (i > 100 .or. i > 0 .and. i <= 5) then
```

という if 文は,

```
if (i > 100 .or. (i > 0 .and. i <= 5)) then
```

と同等です．もっとも，論理演算の優先順位は数値演算の優先順位ほどなじみがないので，後者のようにかっこを利用して明示する方が良いと思います．

do 文でくり返し計算をする場合，“n 回ごとに出力する”というように，一定間隔で動作を変更したいことがあります．この場合には，表 1.3 に示した剰余を計算する組み込み関数 mod を利用して条件分岐をします．たとえば，10 回に 1 回出力するなら,

```
do m = 1, 10000
   ......
  if (mod(m,10) == 0) print *,m,x,y
   ......
enddo
```

のように書きます．プログラムというのは，基本的に全て計算であり，流れのコントロールも計算結果を使って行わなければなりません．このようなケースで mod を使うのは，パターンの一つとして覚えておくと良いでしょう．

3.2　無条件ジャンプ — goto 文，exit 文，cycle 文

1.1 節で説明したように，プログラムというのは基本的に上から順に実行していくものです．do 文を使えば，do ブロックで指定した範囲の実行文を所定の回数くり返すことができますが，くり返す範囲は固定されているし，くり返しを終了する条件はカウンタ変数の増加で決まります．

これに対し，より一般的にプログラムの流れを変えたいとき，たとえば途中で計算を中断してプログラムの最初からやり直す，とか，最後の文に一気に移動して終了する，とかいうときには goto 文を使います．goto 文を使えば，指定した行へ強制的に移動することができます．計算機的には，この移動のことを“無条件ジャンプ”，あるいは単に“ジャンプ”といいます．

goto 文とは，以下のように goto の後に“文番号”と呼ばれる整数を指定した文です．

```
   goto 文番号
```

これに対し，この goto 文でジャンプしたい先の行には，以下のように実行文の前にスペースを 1 個以上空けて文番号を書きます．

```
   文番号  実行文
```

goto 文を使ったプログラム例を以下に示します．

```
   cd = 10
   goto 11                ! 文番号 11 の行へジャンプ
   cd = 50
   ab = 20
   ij = 1
11  ab = 1000             ! この行へジャンプ
```

最後の行で ab=1000 の前に書かれた 11 が文番号です．この例では，最初の cd=10 の実行後，cd=50 から ij=1 までの文は実行されず，直ちに ab=1000 が実行されます．すなわち，cd=50，ab=20，ij=1 の文は書いていないのと等価です．

goto 文でバックすることも可能です．たとえば，次のように書けば，指定した文番号 22 の行と goto 文の間の動作をくり返し実行します．

```
   cd = 50
22  ab = 200              ! この行へジャンプ
   cd = cd + ab - ef
   ef = 10
   goto 22                ! 文番号 22 の行へジャンプ
   ij = 1
```

もっとも，この例ではいつまでたっても goto 文の次の ij=1 は実行されません．こういうのは“無限ループ”と呼ばれ，プログラムエラーの一つです．計算結果に応じて条件分岐し，goto 文より下の行へジャンプする別の goto 文や，プログラム自体を終了させる stop 文を挿入しなければ，プログラムは永遠に終了しません．

文番号は行の指定に使うだけなので，重複さえしなければどの行にどんな数値を付けても良いのですが，なるべく下に行くほど大きくなるように数値を選びます．また，文番号を特定の実行文に付けると，変更するときに付け替えが面倒だと思うときには continue 文を挿入します．continue 文に動作は無く，文番号で位置を指定するためだけに用います．たとえば，上記のプログラムは，

```
   cd = 50
22  continue              ! この行は動作がない
   ab = 200
   cd = cd + ab - ef
   ef = 10
   goto 22
   ij = 1
```

と等価です．

goto 文は，多用するとプログラムの流れがわかりにくくなるし，無限ループに陥る可能性もあるので

使用はできるだけ避けるべきです．基本的なくり返しや，条件に応じたジャンプは do ブロックと if ブロックでほとんどすべて書くことができます．

do ブロックを使ってくり返し計算をするとき，条件によって途中でくり返しを終了したい場合があります．この場合，原理的には goto 文を使わなければなりませんが，goto 文の使用を極力避けるという方針から，exit 文が用意されています．do ブロック中で exit 文を実行すると，その do ブロックの外に飛び出して enddo 文の直後から実行を継続します．

たとえば，要素数 n の配列 a(n) に代入されている値を条件付きで平均するため，

```
  sum = 0
  do m = 1, n
    sum = sum + a(m)
    if (sum > 100) exit
  enddo
  ave = sum/n
```

のように書いたプログラムは，次の goto 文を使ったプログラムと等価です[12]．

```
  sum = 0
  do m = 1, n
    sum = sum + a(m)
    if (sum > 100) goto 10
  enddo
10 ave = sum/n
```

なお，do ブロックの中から外へのジャンプはできますが，外から中に入るジャンプは禁止されています．

do ブロックの途中で実行を中断し，残りの部分をスキップしてカウンタ変数を進めるときには cycle 文を使います．たとえば，

```
  do m = 1, n
    sum = sum + a(m)
    if (sum > 100) cycle
    sum = sum*2
  enddo
```

のように書いたプログラムは，次の goto 文を使ったものと同じです．

```
  do m = 1, n
    sum = sum + a(m)
    if (sum > 100) goto 10
    sum = sum*2
10 enddo
```

すなわち，cycle 文の実行は enddo 文にジャンプするのと等価です．enddo 文にジャンプすれば，カウンタ変数 m を増加して，m > n かどうかをチェックした後，条件を満足しなければ再び do ブロックを最初から実行することになります．ただし，cycle 文は do ブロック内部の制御なので次のように if ブロックを使って同じ動作をさせることができます．

```
  do m = 1, n
    sum = sum + a(m)
    if (sum <= 100) then
      sum = sum*2
    endif
  enddo
```

この方が，条件に応じた動作を明示している点で良いと思います．goto 文を多用しない方がよい，とい

[12] このプログラムを修正して，合計した要素 a(1)〜a(m) の平均値を計算したい場合には，単に n の代わりに m で割るように書き換えるだけでは不完全です．なぜだか考えてみましょう．

う意味合いと同じで，cycle 文もあまり使わない方が良いと思います．

なお，do ブロックが多重の場合，exit 文や cycle 文はその文を含む最も内側の do ブロックに対しての動作になります．より外側の do ブロックの外に抜け出たいなどの場合には，do 文にラベルを付けて exit 文や cycle 文のジャンプ先を指定します．

ラベルを付けるには，do 文と enddo 文の 2 箇所にラベル名を付加します．たとえば，

```
   sum = 0
out: do m = 1, n
     do l = 1, n
       sum = sum + a(l,m)
       if (sum > 100) exit out
     enddo
   enddo out
   ave = sum/n
```

のように書くことができます．“out”がラベル名です．do 文に付加するラベルは，ラベル名の最後にコロン (:) を付けて，do の前に書きます．これに対し，enddo 文に付加するラベルは，対応する do 文と同じラベル名を enddo の後ろにスペースを入れて書きます．コロンは不要です．このプログラムは，

```
   sum = 0
   do m = 1, n
     do l = 1, n
       sum = sum + a(l,m)
       if (sum > 100) goto 10
     enddo
   enddo
 10 ave = sum/n
```

と等価です．なお，複数の do 文にラベルを付けるときは，ラベル名が重複しないようにしなければなりません．

3.3 プログラミング スマートテクニック (その 3)

本節では条件判断によりくり返しを制御する do while 文や，if 文を使うときの注意点などについて説明します．

3.3.1 do while 文と無条件 do 文

do 文には，2.2 節で述べた基本形の他に，条件でループの動作を続けるか終了するかを決める形式も用意されています．これが do while 文です．do while 文は以下のような書式です．

```
   do while（条件）
     .......
     .......
   enddo
```

この場合，do while 文の条件を満足しなくなるまで，その do ブロック内部をくり返し実行します．もし条件を満足しなければ，do ブロックは終了して enddo 文の次に実行が移ります．たとえば，

```
   integer n
   n = 100
   do while (n > 0)
     n = n/2
     print *,n
   enddo
```

と書けば，n を 2 で割っていって，0 になった時点でループが終了します．ここで，n を整数型にしているところがポイントです．実数型だとループがいつ終了するかわかりません．

また，“do のみ”の do 文，“無条件 do 文”もあります．この場合，do ループを終了させる条件がないので，do 文と enddo 文で指定した範囲をいつまでもくり返します．たとえば，

```
do
  sum = sum + x**2
  if (sum > 100) exit
  x = 1.2*x + 0.5
enddo
```

と書くと，sum が 100 を超えるまで計算を続けます．この無条件 do 文を使うときは，ブロック内部に適切な条件分岐で外に出る記述を入れておかないと，無限ループになって永遠に終了しないので注意して下さい．

3.3.2 select case による選択肢に応じた分岐

if 文や if ブロックを使えば，基本的にはどんな条件分岐に対応するプログラムでも書くことができます．しかし，いくつかの数字の中から一致するものを探して，それに応じた動作をさせるような，“選択”に特化した条件分岐を行うときには，select case を使うと便利です．select case は，以下のような select case 文と end select 文で指定したブロックで，その中に case 文を使って必要な選択肢を指定します．この選択肢を指定する記述を“選択子”といいます．

```
select case (式)
 case (選択子 1)
    選択処理 1
    .......
 case (選択子 2)
    選択処理 2
    .......
       :
 case default
    選択処理 0
    .......
 end select
```

select case の動作は，次のようになります．まず最初の select case 文のかっこで指定した“式”を計算し，ブロックの中からその計算結果と一致する選択子を持つ case 文を探します．その結果，一致する case 文があれば，その後に書かれたブロックの処理を実行した後に end select 文の次の行に進みます．たとえば，結果が選択子 1 に一致したときには選択処理 1 のブロックを実行し，選択子 2 に一致したときには選択処理 2 のブロックを実行します．もし，どの選択子とも一致しない場合には case default 文の後の選択処理 0 のブロックを実行します．case default のブロックは省略可能で，その場合にどの選択子とも一致しなければ，何もせずに end select 文の次の行に進みます[13]．なお，“式”の結果は，“整数型”，“文字列”，および 3.3.3 節で説明する“論理型”のいずれかでなければなりません．また，case 文の選択子は“式”のデータ型と一致している必要があります．

たとえば，次のような文を書くことができます．

```
integer n
 .......
select case (n)
 case (1)
   print *,'Single'
 case (2)
   print *,'Double'
 case default
   print *,'Others'
 end select
```

[13] ここでは case default のブロックを一番最後に書いていますが，順番は任意です．

このプログラムの動作は，整数変数 n が 1 ならば'Single' という文字を出力し，2 ならば'Double' という文字を出力し，それ以外ならば'Others' という文字を出力します．

選択子は，1 個の数値や文字の指定だけではなく“:”を使った範囲指定も可能で，“a:b”と書けば，「a 以上かつ b 以下の場合」という意味になります．また，“a:”と書けば，「a 以上の場合」，“:b”と書けば，「b 以下の場合」という意味になります．

さらに，複数の条件を“,”でつなげば，「その中のどれかに一致した場合」という意味になります．たとえば，“a,b,c”と書けば，「a か b か c のどれかに一致する場合」という意味です．この複数選択には範囲指定を入れることもできるので，たとえば以下のように書くことができます．

```
select case (n)
 case (1:9)
   print *,'Single'
 case (10:30,100)
   print *,'Double'
 case (1000,2000,3000:)
   print *,'Triple'
 end select
```

この例では，n が 1〜9 のときは，'Single' という文字を出力し，10〜30 または 100 のときには'Double' という文字を出力し，1000 か 2000 または 3000 以上のときには'Triple' という文字を出力します．case default 文のブロックがないので，それ以外の場合には何もしません．なお，ある case 文で指定した選択子が他の case 文の選択子と重複するとエラーになるので注意して下さい．

select case は，定数と一致する条件分岐にのみ使える書式なので，用途は限られていますが，分岐条件が多くて if 文では長くなるときに使えばプログラムがわかりやすくなります．

3.3.3 論理型データ

Fortran で取り扱える数値データには，整数型，実数型，複素数型があります．この他，print 文で出力するときに使用する文字列もデータ型の一種で，これについては第 5 章で詳しく説明しますが，Fortran には，もう一つ，“論理型”と呼ばれるデータ型が用意されています．論理型は，“真”と“偽”という 2 種類の値しか取らないデータ型のことで，大小関係の判定結果を変数に保存したり，if 文に記述して条件分岐に使うことができます．また，組み込み関数の中には論理型の結果を返すものがあるので，使い方を知っておくと便利です．

論理型の定数は 2 つしかありません．“.true.”と“.false.”です．これらは以下の意味を持ちます．

```
.true.      真
.false.     偽
```

論理型の変数は logical を使って宣言します．

```
logical 変数 1, 変数 2, ...
```

3.1 節で説明したように，if 文には表 3.1 の比較条件や表 3.2 の論理式を用いて判定条件を記述することができますが，論理型変数にはこの判定条件の「真偽」を代入することができます．

たとえば，次のように変数を宣言して利用することができます．

```
logical l1,l2,l3
integer m
m  = 10
l1 = .true.                !  定数の代入
l2 = m == 5                !  比較条件の判定結果の代入
l3 = l1 .and. l2           !  論理演算結果の代入
print *,l1,l2,l3           !  論理型も出力可能
```

ここで，l1 には定数 “.true.”が代入され，l2 には “m==5”の判定結果，すなわち「m は 5 に等しいか」という判定の真偽が代入されます．m は 10 なので，l2 は “.false.”です．l3 には「l1 かつ l2」の真偽が代入されるので “.false.”になります．この結果，最後の print 文の出力は，

```
T F F
```

のようになります[14]．

論理型変数は，if 文や do while 文の条件の中に記述して判定に使うことができます．たとえば，

```
logical l1
integer m
do m = 1, 10
   l1 = m < 5
   if (.not.l1) exit
enddo
```

のように書くことができます．

論理型変数は，if 文を使わなくても大小関係の判定結果を保存することができるという点で，使い方によってはスマートなプログラムを作るのに役立ちます．ただし，Fortran の論理型変数は整数型と同じ 4byte の領域を利用するため，2 値しか表現できないのにもかかわらずメモリの節約にはなりません．

3.3.4 0.1 を 10 回足しても 1 に等しくならない

実数型を使って条件分岐をするときに注意しなければならないことがあります．次のプログラムを考えてみましょう．

```
real x
integer m
x = 0.0
do m = 1, 20
  x = x + 0.1                !  0.1 をくり返し加える
  if (x == 1.0) exit
enddo
print *,'x = ',x
```

この do ループは，実数型変数 x に 0.1 をくり返し加えていきますが，ループの 10 回目で if 文の条件「x が 1.0 に等しい」を満足し，exit 文によりループの外に出て，print 文の出力は 1.0 になると予想されます．しかし，実際にやってみるとわかりますが，print 文の出力は 2.0 です．つまり，if 文の条件は満足しない，すなわち，x が 1.0 に等しくなることはないのです．なぜでしょうか．

これは，コンピュータが 2 進数で計算しているためです．1.10.4 節で述べたように，10 進数で 123 と表現される数を 2 の多項式で表せば，

$$123 = 1 \times 2^6 + 1 \times 2^5 + 1 \times 2^4 + 1 \times 2^3 + 0 \times 2^2 + 1 \times 2^1 + 1 \times 2^0$$

となり，これが 2 進数で表したときに 1111011 となる理由でした．コンピュータの数値は全て 2 進数なのですから，小数点以下の数値も 2 進数で表さねばなりません．たとえば，10 進数の 0.123 は

$$0.123 = 1 \times 10^{-1} + 2 \times 10^{-2} + 3 \times 10^{-2}$$

のように，10 の負べき乗を基本として表された数のことですが，これと同様に，2 進数で小数点以下の数 x を表すには，

$$x = b_1 \times 2^{-1} + b_2 \times 2^{-2} + b_3 \times 2^{-2} + \cdots$$

[14] ここで，T が “真”，F が “偽”であることを示しています．この出力表示はコンパイラによって異なることがあります．

となるような 0 または 1 の係数 $b_1, b_2, b_3, \cdots$ を選んで,

$$0.b_1b_2b_3\cdots$$

と並べます．コンピュータはこの係数 $b_1, b_2, b_3, \cdots$ で小数点以下を表現しています．

さて，コンピュータのメモリで表現できる 2 進数の桁 (ビット数) には限りがあるので，小数はどこかで打ち切る必要があります．ところが，2 進数の場合，有限小数で表せるのは，2^{-n} の和だけです．たとえば，0.5 ($=2^{-1}$) や 0.125 ($=2^{-3}$) やその和である 0.625 は有限の 2 進小数で表すことができますが，2^{-n} の和で表せない数値は無限小数になってしまいます．たとえば，10 進数の 0.1 は,

$$0.1 = 0\times2^{-1} + 0\times2^{-2} + 0\times2^{-3} + 1\times2^{-4} + 1\times2^{-5} + 0\times2^{-6} + 0\times2^{-7} + 1\times2^{-8} + 1\times2^{-9}\cdots$$

となり，2 進数で表せば，0.0001100110011... という循環小数になります．よって，コンピュータにおける 0.1 は近似値でしかなく，10 回加えても 1 に等しくはなりません[15]．

このように，実数計算をするときは 10 進数と 2 進数の違いによって予想外の結果になることがあるので注意が必要です．たとえば，先ほどの 0.1 を加えていくプログラムを期待どおりに終了させるには，“1.0 に等しい”ではなく，“1.0 に充分近い”という判定にする必要があります．たとえば,

```
x = 0.0
do m = 1, 20
  x = x + 0.1
  if (abs(x-1.0) < 1.0e-3) exit    !  誤差が小さくなったら終了するように変更
enddo
print *,'x = ',x
```

のように書けば，print 文では 1.0 が出力されるはずです．なお，このプログラムでは誤差を 10^{-3} にしていますが，この判定条件は問題に依存します．

また，同じ少数点以下の数をくり返し加えるときには，回数を掛け算した方が精度が上がります．たとえば,

```
x = 0.0
do m = 1, 100
  x = x + 0.1
   .......
enddo
```

と書くよりも,

```
do m = 1, 100
  x = 0.1*m
   .......
enddo
```

と書く方が，x の精度は上がります．

[15]ちなみに，この理屈で言えば，0.1×10 という掛け算も 1.0 と等しくならないはずですが，手元のパソコンでやってみると if 文を満足しました．これは，四捨五入のように計算結果を丸めるためです．しかし，丸め方はハードウェアやコンパイラの仕様に依存するので，それを期待してプログラムを書くのは避けた方が無難だと思います．

演 習 問 題 3

(3–1)　2 次方程式の解 (解の判別付き)

3 個の実変数 a，b，c に適当な値を代入し，それから，

$$ax^2 + bx + c = 0$$

の解を計算して出力するプログラムを作成せよ．まず，解の判別式，$D = b^2 - 4ac$，を計算して，2 実解になったとき，重解になったとき，虚解になったときでそれぞれ正しい結果を出力するようにせよ．さらに，その解を $ax^2 + bx + c$ に代入して 0 に近い値になることを，判別式の値と数値型に応じて計算手順を変えて確かめよ．

(3–2)　非線形方程式の解法：逐次代入法

$x_0 = 0$ として，次のように次々に代入していくと，x_n は徐々に $\cos x = x$ の解に近づくことがわかっている．

$$\begin{aligned} x_1 &= \cos x_0 \\ x_2 &= \cos x_1 \\ x_3 &= \cos x_2 \\ &\vdots \\ x_{n+1} &= \cos x_n \end{aligned}$$

最初に，この代入を 100 回くり返し，その後で 101 回目の値 x_{101} と 100 回目の値 x_{100} の差を計算するプログラムを作成せよ．

それができたら，収束条件を $|x_{n+1} - x_n| < \varepsilon$ とし，ε を適当に小さい値に定めて (たとえば，10^{-7})，収束したら x_n を出力して終了するプログラムにせよ．このとき，収束するまでの回数も出力せよ．

なお，この問題では配列を使わないこと．

(3–3)　非線形方程式の解法：Newton 法

次の公式をくり返し用いて，方程式 $f(x) = 0$ の実数解の一つを近似的に求めるアルゴリズムを Newton 法という．

$$x_{n+1} = x_n - \frac{f(x_n)}{f'(x_n)}$$

以下の関数 $f(x)$ に対し，適当な初期値 x_0 を与えて Newton 法で解を求めるプログラムを作成せよ．また収束条件を $|x_{n+1} - x_n| < \varepsilon$ とし，ε は適当に定めよ．このとき，収束するまでの回数も出力せよ．

(1)　$f(x) = x - \cos(x)$

(2)　$f(x) = x^3 - 2$

それぞれの問題に対し，得られた結果を $f(x)$ に代入して，答えがどの程度 0 に近いかを確認せよ．

第 4 章　サブルーチン



ここまで，Fortran の基本的な文法と高速化のテクニック，エラーの出にくいプログラムの書き方などを説明しました．ここまでの知識を使うだけで，原理的にはどんなプログラムを作ることも可能です．プログラムとは計算機に仕事をさせるための手順ですから，それほどバラエティはありません．動作のくり返しを書くための `do` 文や，条件分岐の `if` 文が使いこなせれば，計算機の能力のほとんどを引き出すことができます．

しかし長いプログラムを書くときは，メンテナンスのしやすさを考慮して書かなければなりません．プログラムが短ければ全体を見ながらチェックや修正ができますが，長くなるとチェックに時間がかかり，見落としによる修正ミスが起こる可能性も高くなります．見落としを防ぐには何度も見直す必要があり，チェック時間はさらに増大します．

このメンテナンスを容易にする手法の一つが“サブルーチン”です．サブルーチンとはメインプログラムと同じレベルの閉じたプログラムのことで，必要に応じて他のプログラムから呼び出してその機能を使います．長いプログラムをサブルーチンに分割し，それらをビルディングブロックとして全体を構成すれば，プログラム作成や修正の作業が楽になります．本章ではこのサブルーチンの作り方と使い方を説明します．

4.1　サブルーチンの利用目的

“ルーチン”とは，動作開始点と終了点が定義されている閉じたプログラム手順を示す用語です．プログラムを起動したときに最初に動作するのはメインプログラムですが，これを“メインルーチン”ともいいます．“サブルーチン”は，メインプログラムと同様に動作開始点と終了点を持っていますが，メインプログラムと異なり，単独で動作することはできません．必ず，他のルーチンに動作開始命令を記述して，必要に応じて実行させる必要があります．一つのプログラムにメインプログラムは一つですが，サブルーチンは名前が異なれば複数存在してもかまいません．また，いくつかのルーチンごとにファイルを作成して別々にコンパイルし，必要に応じて結合 (リンク) させることも可能です．

サブルーチンを利用する目的は主として 3 つあります．

(1) 複数の場所で同じ計算手順を使用するため
(2) 方程式の解法や行列式の計算などのような定型処理をするため
(3) 長いプログラムを分割してメンテナンスを容易にするため

(1) がサブルーチン本来の使い方です．ある一連の手順を複数の場所で使うとき，それぞれの場所に同じプログラムを書くと，プログラムが長くなるし，修正の際に何カ所も同じ修正をしなければなりません．このとき，その手順をサブルーチンに置き換えれば，プログラムは短くなるし，チェック作業も短縮されます．

(2) は (1) を発展させたものです．複雑な数学公式や汎用性のある数値計算アルゴリズムをプログラム中に直接記述するときは，プログラム中で宣言された変数や配列を使って書きます．このため，同じ計算手順を別のプログラムで使いたくてもそのまま使えるとは限りません．そこで計算手順をサブルーチンにして，計算に必要な数値を与える部分と計算結果の数値を受け取る部分だけが外部から見えるようにしておきます．そうすれば，受け渡し部を書き換えるだけで，複雑な計算手順を色々なプログラムから利用することができます．このようにブラックボックス化したサブルーチンは，他の人に提供したり，他の人からもらって利用することも可能で，そのような汎用性のあるサブルーチンを集めたものが，“ライブラリ”です．

さて，計算機シミュレーションのように多数の解析手順を複合したプログラムを書く場合には，(3) の目的が重要になります．様々な物理過程の計算や，入出力に関する作業など，ある程度まとまった内容ごとにサブルーチンにするのです．文章でいえば，章や節に分割するようなものです．このため，メインプログラムには，それぞれのサブルーチンを呼び出す文と，必要ならそれらをくり返すための do 文だけを書いておきます．シミュレーションプログラムは，サブルーチンという部品を使って組み立てるものだと考えればいいでしょう．

サブルーチンに分割しておけば，プログラムのメンテナンスが楽になります．これは，それぞれのルーチンが独立しているので，プログラムを修正したときの影響や，エラーが発生する原因がルーチン内に限定されるからです．

4.2 サブルーチンの宣言と呼び出し

サブルーチンは subroutine 文で開始を宣言し，end subroutine 文で終了します．すなわち，次のような構造にします．

```
subroutine subr1
  implicit none
  real a,b
  integer i
   .......
   .......
end subroutine subr1
```

先頭の subroutine 文で，subroutine の後に指定した文字の並び (この例では subr1) を “サブルーチン名”といいます．また，最後の end subroutine 文には subroutine 文と同じサブルーチン名を指定します．サブルーチン名の付け方は，program 文におけるプログラム名の付け方と同じです．見てわかるように，メインプログラムと同じ構造です．サブルーチン内部のプログラムの書き方も基本的にメインプログラムと同じで，subroutine 文の次に implicit none を書き，非実行文を上方に集約して，その後に実行文を書きます．サブルーチンはそれ自体で閉じているので，実行文中で使う変数や配列は，基本的にその内部で宣言しなければなりません．ただし，異なるルーチン間で共有可能な変数を別途用意することは可能です．これについては，4.7 節で説明します．

上の例では，subroutine 文にサブルーチン名しかありませんが，これを “引数なしサブルーチン”といいます．これに対して，サブルーチン名の後に，かっこで囲んだ変数リストを付加することができ，これを “引数ありサブルーチン”といいます．以下に引数ありサブルーチンの一例を示します．

```
subroutine subr2(x,m,y,n)
  implicit none
  real x,y,a,b,z(10)
  integer m,n,i,k
   .......
   .......
end subroutine subr2
```

リスト中の変数を “引数”といいます．この例では，x,m,y,n が引数です．引数は，サブルーチンとそのサブルーチンを使うルーチンの間で数値を受け渡すために使います．引数もサブルーチン内部の変数なので，必要な型に応じた宣言をしなければなりません．

サブルーチンを動作させてその機能を使うときは，call 文を使ってそのサブルーチンを指名します．このため，サブルーチンの動作を指定することを，“サブルーチンを呼び出す”とか “コールする”といいます．call 文は以下のような形式です．

```
call サブルーチン名                                  ! 引数なしサブルーチン用
call サブルーチン名（数値または変数のリスト）        ! 引数ありサブルーチン用
```

たとえば，先ほど出てきた，引数なしと引数ありの二つのサブルーチンを使うときは，それぞれ，次の (1) と (2) のようにコールします．

```
real z
integer m
call subr1          !..................... (1)
m = 21
call subr2(10.0,100,z,m*5+1)          !.....  (2)
```

1.4 節で説明したように，計算式は計算結果の数値を動作命令に与えるので，call 文の引数には，(2) の一番右のように計算式を与えることもできます．

サブルーチンは単独では動作できないので，メインプログラムが別途必要です．たとえば，次のようなセットを構成して初めて一つのプログラムが完成します．

```
program stest1
   implicit none
   real x,y
   x = 5.0
   y = 100.0
   call subr(x,y,10)              !  サブルーチンの呼び出し
   print *,x,y
end program stest1

subroutine subr(x,y,n)
   implicit none
   real x,y
   integer n
   x = n
   y = y*x
end subroutine subr
```

ここでは，メインプログラムを先に，サブルーチンを後に書きましたが，逆でもかまいません．サブルーチンが複数存在する場合も，ルーチンを記述する順番は実行結果とは無関係です．サブルーチンの中から別のサブルーチンをコールすることも可能です．

上記のプログラム実行の流れを図 4.1 に示します．



図 4.1　サブルーチンの呼び出しと実行の流れ

図のように，メインプログラムの call 文でサブルーチンを指名すると，サブルーチンの一番最初の実行文に動作が移り，サブルーチン内部の動作が完了すると，コールしたメインプログラムに戻って，その call 文の次の文から実行を継続します．

条件に応じて，途中でサブルーチンの処理を打ち切って戻るときには return 文を用います．

```
subroutine subr(x,y,m,n)
  implicit none
  real x,y
  integer m,n
   .......
  if (m < 0) return
   .......
end subroutine subr
```

この例では, m < 0 のときにサブルーチンの処理が終了し, コールしたルーチンに戻ります．ここで return 文の代わりに stop 文を用いれば，プログラム自体が終了します．

4.3 ローカル変数と引数

サブルーチン内部で宣言した変数や配列は，メインプログラムや他のサブルーチンから独立しています．すなわち，同じ名前を使っても全く別の変数です．たとえば，

```
program stest1
  implicit none
  real x,y
  x = 10.0
  y = 30.0
  call subr1
end program stest1

subroutine subr1
  implicit none
  real x,y
  print *,x,y
end subroutine subr1
```

と書いても，サブルーチン subr1 中の print 文によって出力される x と y はメインプログラムで代入した 10.0 と 30.0 ではなく，全く無関係な数字です．逆に言えば，他のルーチンでの宣言を気にせずに変数や配列の名前を決めることができます．このルーチン内部でのみ有効な変数を “ローカル変数”といいます．

上記のプログラムを期待通り働かせるために，コール側ルーチンの数値をサブルーチンに伝えるのが引数です．たとえば，上記のプログラムを次の subr2 のように書き直せば，サブルーチン中の print 文で出力される x と y は，メインプログラムと同じ 10.0 と 30.0 になります．

```
program stest2
  implicit none
  real x,y
  x = 10.0
  y = 30.0
  call subr2(x,y)
end program stest2

subroutine subr2(x,y)
  implicit none
  real x,y
  print *,x,y
end subroutine subr2
```

引数は，数値の受け渡しをするための窓口に過ぎないので，コール側の引数とサブルーチン側の引数の変数名を同じにする必要はありません．次のサブルーチンに置きかえても全く同じ動作をします．

```
subroutine subr2(a,b)
  implicit none
  real a,b
  print *,a,b
end subroutine subr2
```

外部からの影響を受けたり，外部に影響を与える，という性質を持つことを除けば，引数もローカル変数です．すなわち，コールしたルーチンからその詳細は見えません．見えるのは，引数という窓口の並びだけです．このため，引数ありサブルーチンを使うときに重要なポイントは，

(1) 引数の数
(2) 対応する引数の型

が call 文と subroutine 文とで一致していなければならないことです．たとえば，

```
program stest3
  implicit none
  real z
  integer n
  z = 200.0
  n = 21
  call subr3(10.0,z**2,100,n*5+1)
end program stest3

subroutine subr3(x,y,m,n)
  implicit none
  real x,y
  integer m,n
  print *,x,y,m,n
end subroutine subr3
```

というプログラムでは，表 4.1 のような対応になっています．

表 4.1　サブルーチンの引数対応

call 文	subroutine 文	数値型
10.0	x	実数
z**2	y	実数
100	m	整数
n*5+1	n	整数

ここで注意すべきなのは，第 1 引数の x は実数型なので実定数 10.0 を与え，第 3 引数の m は整数型なので整定数 100 を与えていることです．もし，第 1 引数に同じ意味だろうと思って 10 という整定数を与えると，実行時エラーになる可能性があります．

サブルーチン内部の実行文で，引数に数値を代入すると，call 文の対応する引数に与えた変数にその数値が代入されます．たとえば，

```
program stest4
  implicit none
  real x,y,p
  x = 10.0
  y = 30.0
  call subr4(x+y,20.0,p)
  print *,x,y,p
end program stest4
```

```
subroutine subr4(x,y,z)
  implicit none
  real x,y,z
  z = x*y
end subroutine subr4
```

と書くと，サブルーチン subr4 中の x はコール側の x+y，すなわち 40.0 であり，サブルーチン中の y はコール側の 20.0 なので，サブルーチン中の z には x*y の計算結果である 800.0 が代入されます．このとき，z が引数なので，call 文の対応する位置にある変数 p に 800.0 が代入され，call 文の次の print 文では x，y，p として，10.0，30.0，800.0 が出力されます．

このようにサブルーチンの引数に数値を代入することで，call 文の引数変数に代入される数値を “戻り値”といいます．戻り値を使えば，サブルーチンの動作で得られた結果をコール側で受け取ることができます．ただし，call 文の引数には定数を与えたり計算式を書いてもよい，と述べましたが，戻り値を指定する引数は別で，必ず変数か配列にしなければなりません．理由を次節で説明します．

4.4 間接アドレスを用いたルーチン間におけるデータの受け渡し

ルーチン間のつながりをもう少し詳しく考えてみましょう．各ルーチンは基本的にプログラム領域とデータ領域から構成されています．プログラム領域とは文字通りプログラム命令が記録されている領域のことであり，データ領域とは変数や配列のために用意されたメモリ領域のことです．プログラム的には，変数や配列の宣言文がデータ領域の指定を意味しています．



図 4.2　プログラム領域とデータ領域

変数や配列が各ルーチンごとに宣言されなければならないのは，データ領域が各ルーチンそれぞれに付属しているからです．このため，図 4.2 のように異なるルーチンで同じ名前の変数を宣言しても，それらは別のデータ領域に所属しています．これがローカル変数です．ローカル変数は，そのルーチン内部からしか参照することができません．

このため，あるルーチンで計算した数値を別のルーチンで使うには特別な手続きが必要になります．これが引数です．引数を使えば，コール側のルーチンとサブルーチンの間でデータをやりとりすることができます．では，具体的にどうやってデータの受け渡しをしているのでしょう．

データをサブルーチンに渡す手段として，最も単純に考えられるのは，引数の数値を直接サブルーチンの変数に代入することです．図 4.3 を見て下さい．図のように，コール側の引数 x と y の内容 (この例では 5.0 と 100.0) が，サブルーチンの引数として宣言された変数，a と b に代入されています．この方式

は，引数の値を直接渡して呼び出す，という意味で“call by value”と呼ばれています．“call by value”は，コール側からサブルーチン側への値の引渡しについては問題ありません．C 言語はこの方式を採用しています．



図 4.3　C 言語の“call by value”を使ったデータ渡しでは値を送り返せない

しかし，この方式では，サブルーチン側で作成したデータをコール側に戻すことはできません．サブルーチンは，コール側から送られてきた“値”はわかるのですが，それ以上の情報がないので，コール側のどこにデータを代入すればいいかわからないからです．図 4.3 のように，サブルーチン中の代入文，`b=a+25.0`，でサブルーチンの変数 `b` に 30.0 という値を代入しても，サブルーチンのデータ領域の `b` に代入されるだけで，メインプログラムのデータ領域には影響がありません．

この問題を解決するために，Fortran は“call by value”ではなく，“call by reference”という方式を採用しています．“call by reference”とは，変数の内容ではなく，変数の存在場所 (メモリアドレス) をサブルーチンに伝えて呼び出す方式です．1.1 節で述べたように，CPU が変数 (メモリ) とデータのやりとりをするときにはメモリアドレスを利用しています．たとえば，簡単な代入文，

```
a = 1500.0
b = a
```

を考えてみましょう．この結果，`b` のメモリに 1500.0 という値が書き込まれますが，代入文 `b=a` において，`a`(アドレスを `[103]` とする) という変数からデータを取って来て `b` に代入するという流れは，

```
a のアドレス指定　→　[103] 番地のメモリ内のデータ (1500.0) の呼び出し
　　→　b に代入
```

となります．流れを図に描けば図 4.4 のようになります．このような指定したアドレスのメモリに入っているデータを読み書きする方式を“直接アドレス”といいます．



図 4.4　直接アドレスによるメモリ参照

これに対し，あるアドレス (AD_1) のメモリに入っているのが別のメモリのアドレス (AD_2) で，アドレス AD_1 を指定して，アドレス AD_2 のメモリに入っているデータを読み書きする方式を“間接アドレス”といいます．Fortran では，“call by reference”でサブルーチンを呼び出すので，引数にはコール側ルーチンのメモリアドレスが入っています．このため，サブルーチン内で引数の値を読み書きするときは間接アドレスを使います[16]．たとえば，

```
program stest3
  implicit none
  real a
  a = 1500.0
  call subr(a)
end program stest3

subroutine subr(s)
  implicit none
  real b,s
  b = s
end subroutine subr
```

というプログラムを考えてみましょう．サブルーチン subr の中の代入文 b=s によって，変数 b のメモリの値は 1500.0 になりますが，引数変数 s には a の内容である 1500.0 ではなく，メインプログラムの変数 a のアドレス [103] が入っています．このため， b=s という代入文で，変数 s (メモリアドレスを [106] とする) からデータを取ってきて変数 b に代入する流れは，

```
s のアドレス指定  →  [106] 番地にあるメモリアドレスデータ (103) の呼び出し
                 →  [103] 番地にあるデータ (1500.0) の呼び出し
                 →   b に代入
```

となります．図示すれば図 4.5 のようになります[17]．



図 4.5　間接アドレスによるメモリ参照

間接アドレスによる読み書きは直接アドレスよりも時間がかかります．よって，b=s の代入文において，代入する値を保持している右辺の変数 s が間接アドレス指定の場合にはメリットがありません．しかし，代入される左辺の変数 b が間接アドレス指定の場合には，このしくみを利用することで，コール側ルーチンの変数にサブルーチン側のデータを代入することができます．

たとえば，図 4.6 のようなプログラムを考えます．サブルーチンの引数変数 b に a+25.0 の結果である 30.0 を代入すると，b にはメインプログラムの変数 y のアドレスが入っているので，間接アドレス指定により y に 30.0 が代入されます．

[16]最近の Fortran では，引数を“call by value”で呼び出すことも可能ですが，手続きが少し必要です．“call by value”の引数利用方法については 8.4 節で説明します．

[17]配列を実現するのにも間接アドレスが使われています．たとえば，a(10) は a(1) から数えて 10 番目のメモリです．よって，配列の先頭要素 a(1) のアドレスをメモリに記録し，それに 9 (=10 − 1) を加えて間接アドレスを用いれば，a(10) のメモリを読み書きすることになります．



図 4.6　Fortran では “call by reference” によりデータを送り返すことができる

戻り値を利用するときは，対応する引数にコール側変数のアドレスを与えなければなりません．これが変数または配列を指定しなければならない理由です．戻り値指定の引数に定数や計算式を与えると実行時エラーになります[18]．

4.5　配列を引数にする場合

配列を call 文の引数にするときは，配列名を引数にすることも，配列要素を引数にすることも可能です．2.1.2 節で述べたように，配列名は配列の先頭要素アドレスを示すので，配列名を引数にすることと，その配列の第 1 要素を引数にすることは同じ意味になります．たとえば，real a(10) と宣言した 1 次元配列に対して，

```
call sub(a)   と   call sub(a(1))
```

は同じ動作をします．

これを受けるサブルーチンでの宣言は，サブルーチン内部でそのデータをどう使うかで決めます．一例を示します．

```
subroutine sub(x)
  implicit none
  real x                    !  単一変数宣言
  x = 10.0
end subroutine sub
```

この例では引数 x が単一変数として扱われています．このため，call 文で指定した a(1) だけがサブルーチン内部で利用でき，他の要素は使えません．これは，call sub(a) のように，配列名を与えた場合でも同じです[19]．

[18]ではなぜコール側の引数に定数や計算式を与えることができるのかというと，定数や計算式の結果の数値を一度別のメモリ領域に保存して，そのメモリアドレスを引数に与えているからです．この特別なメモリはユーザーが書き換えることができないようになっているので，サブルーチン側で代入をするとエラーになるわけです．

[19]ただし，このサブルーチンを利用するときに call sub(a) のように配列名のみを与えたプログラムを書いて gfortran でコンパイルするとエラーになりました．これは，Fortran の規格の中に，subroutine 文の引数が単一の数値 (スカラー) の場合には，call 文の対応する引数はスカラーでなければならないという規定があるので，それに忠実に従っているためです．関接アドレスを使ってサブルーチンに値を引き渡している以上は，エラーにするのは少しやり過ぎだと思うんですけどね．

一方，call sub(a(1)) のように配列要素を与えた場合にはエラーになりません．規格でも「配列要素名以外のスカラー」のように例外として許されています．call 文の引数が配列要素の場合は subroutine 文の引数はスカラーでも配列でもよい，ということになります．

このエラーを理解した上で回避する方法は付録 A で紹介します．

これに対し，

```
subroutine sub(x)
  implicit none
  real x(10)                !  配列宣言
  integer i
  do i = 1, 10
    x(i) = i
  enddo
end subroutine sub
```

のように引数 x を配列宣言すれば，あたかもコール側ルーチンの a という配列がサブルーチン内部の配列 x になったかのように取り扱うことができます．ここで “あたかも”という言葉を使ったのは，サブルーチンでの宣言文の書き方によってはそうならないことがあるからです．たとえば，サブルーチンでの配列宣言 x(10) の要素数 10 はコール側と同じ数にする必要はありません．x(100) のように大きくても，x(1) のように小さくてもかまわないのです．これは，引数 x が配列の先頭要素アドレスを保持している 1 個の変数にすぎないからです．サブルーチンでの引数配列の宣言は，配列の形状 (次元や下限値) を明示するためのダミーにすぎません．

このため，上記のサブルーチン sub を，

```
 call sub(a(3))
```

のようにコールすると，a(3) を先頭とする 1 次元配列として処理します．すなわち，サブルーチン内部の配列 x(1)，x(2)，...，x(10) は，コール側配列の a(3)，a(4)，...，a(12) に対応します．

また，b(10,10) と宣言した 2 次元配列の j 列目の 10 個の要素，b(1,j)，b(2,j)，...，b(10,j) を処理したいときには，

```
call sub(b(1,j))
```

と書けばいいことになります．これは，b(1,j)〜b(10,j) がメモリ上で並んでいるためです[20]．

このように，サブルーチンの引数が受けるのは，配列の先頭アドレスの情報だけなので，コール側ルーチンで宣言した要素数とサブルーチンで宣言した要素数を一致させる必要はありません[21]．与えられた配列をどう宣言するかは，サブルーチン内部の計算の都合に合わせて，サブルーチン側で決めることができます．よって，汎用性のあるサブルーチンにするには，配列だけではなく，配列の要素数も引数に加えてサブルーチンに伝える必要があります．

配列の形状さえわかればいいのですから，サブルーチンで引数配列を宣言するときは，数字の代わりに “*”を書くことができます．例として，要素数 n の 1 次元配列 a のデータを，同じ長さの 1 次元配列 b にコピーするプログラムを考えてみましょう．最も単純な答えは，以下のようなものです．

```
subroutine copy(a,b,n)
  implicit none
  real a(*),b(*)
  integer n,i
  do i = 1, n
    b(i) = a(i)
  enddo
end subroutine copy
```

[20] これに対し，i 行目の要素，b(i,1)，b(i,2)，...，b(i,10) はメモリ上で並んでいないため，この方法で処理することはできません．この場合には部分配列を利用します．詳しくは 7.2 節で説明します．

[21] 要素数だけではなく，次元さえ一致させる必要はありません．コール側が 2 次元配列で，サブルーチンでの宣言が 1 次元配列でも動作は可能です．たとえば，上記のサブルーチン copy は，2 次元配列でも使用できます．これは，2.1.2 節で述べたように 2 次元配列もメモリ上では 1 次元的に並んでいるからです．ただし，要素数を与える引数 n には全要素数を指定する必要があります．たとえば，a(10,10) と宣言された配列ならば，n に 100 (=10×10) を与えます．

このように，サブルーチンの引数配列 a や b の宣言を “*”にすることで，要素数が不定の 1 次元配列であることを明示しています[22].

“*”による宣言は 2 次元以上の配列でも使うことができますが，制限があります．たとえば，a(*,*) という形の宣言はできません．なぜなら，2 番目の要素を進めるのは 1 番目の要素が最大値，すなわち宣言した上限数に達したときなので，1 番目の要素が不定では情報不足で進められないからです．このため，“*”が使えるのは，一番右側の要素数のみです[23]．たとえば，real a(3,*) と宣言すると，第 1 要素数が 3 の 2 次元配列として計算されます．

しかし，これではどんな 2 次元配列でも使えるサブルーチンを作ることはできません．そこで，“整合配列”と呼ばれる機能が用意されています．整合配列とは subroutine 文の引数の中にある整数型変数を利用して宣言した形の引数配列のことです．たとえば，以下のように宣言することができます．

```
subroutine copy2d(a,b,m,n)
  implicit none
  real a(m,n),b(m,n)
  integer m,n,i,j
  do j = 1, n
    do i = 1, m
      b(i,j) = a(i,j)
    enddo
  enddo
end subroutine copy2d
```

この例では，m と n が引数の中にあるので，これらを使って引数の配列 a と b を宣言することができるのです．a(m,n) のような単純な宣言だけではなく，a(0:2*m,n-1) のような，下限指定や計算式を使った宣言も可能です．

このサブルーチンの使用例を次に示します．サブルーチンの引数 a，b にどんな要素数の 2 次元配列を与えても，コール側ルーチンにおける配列宣言と同じ要素数を m，n に与えれば，正しく動作します．

```
program stest1
  implicit none
  real a(10,20),b(10,20),c(100,200),d(100,200)
   .......
  call copy2d(a,b,10,20)
  call copy2d(c,d,100,200)
end program stest1
```

なお，配列宣言の際には，コロンを付加して下限を指定することができますが，これをサブルーチンに引き継ぐにはサブルーチン側でも同じ下限指定をする必要があります．たとえば，

```
program stest2
  implicit none
  real a(-1:100)
  call sub(a,a,100)
end program stest2

subroutine sub(x,y,n)
  implicit none
  real x(-1:n),y(*)
  integer n
  x(10) = 10.0
  y(10) = 10.0
end subroutine sub
```

[22]なお，“*”だけ書くと配列の下限は 1 です．もし下限を変更したいときには下限を別途指定します．たとえば，配列 a の下限を 0 にする場合は，real a(0:*) と宣言します．

[23]2.1.2 節の配列の順番を与える公式が一番右側の要素数に依存しないことを思い出して下さい．

と書いた場合，メインプログラムと同じ下限指定をした配列 x では x(10) がメインプログラムの a(10) に対応するので，a(10) に 10.0 が代入されます．しかし，下限指定をしていない配列 y では下限が 1 ですから，a(-1) から数えて 10 番目の a(8) に 10.0 が代入されます．任意の下限を持つ配列に対して動作するプログラムにするには，下限の数値も引数変数にした整合配列を使います．

4.6 関数副プログラム

sin(x) のように，引数を使って内部で計算した結果を計算式中で使うことができる“関数”を自作することもできます．これを“関数副プログラム”といいます．関数副プログラムは，サブルーチンとほとんど同じ構造をしていますが，以下のような違いがあります．

(1) subroutine の代わりに function を書く
(2) 関数名を変数として宣言し，それに計算結果 (関数値) を代入しなければならない

これ以外は，引数ありサブルーチンと同じです．引数に関する注意もサブルーチンと同じで，戻り値を与える引数を含めることもできます．関数副プログラムとは，引数以外に特別な戻り値を 1 個持つサブルーチンの変種だといえます．この戻り値が関数値になります．

関数副プログラムの一例を示します．

```
function square(x)
  implicit none
  real square,x                  !  関数名の型宣言
  square = x*x                   !  関数名に値を代入する
end function square
```

このように，function 文で開始し，end function 文で終了します．また，関数名 square を実数型宣言し，引数 x の 2 乗を代入して終了しています．すなわち，この例は引数 x に実数型の数値を与えると，x^2 を関数値として与える関数になります．

関数副プログラムを使うことは，“関数名”という変数を使うことであると考えます．このため，作成した関数を使用するルーチンでも関数名を型宣言する必要があります．たとえば，上例の square を使うには，

```
program ftest1
  implicit none
  real x,y,square                !  使用する関数名を型宣言する
  x = 5.2
  y = 3.0*square(x+1.0) + 50.5
  print *,x,y
end program ftest1
```

のように，square という関数名を関数副プログラムでの宣言と同じ型で宣言しておかなければなりません．関数副プログラム中で宣言した関数名の型と，その関数を使用するルーチンで宣言した関数名の型が異なる場合には実行時エラーになります．

関数名に値を代入するという形式は，Fortran の古くからある文法の一つですが，関数名が長いときなど，計算式の中に関数名を入れたくないときのために，function 文に付加する result 句が用意されています．result 句を使えば，関数値を代入する変数名を任意に指定することができます．たとえば，上記の関数副プログラム square は，以下のように書くことができます．

```
function square(x) result(y)  !  result 句付きの function 文
  implicit none
  real x,y                !  result 句で指定した変数の型宣言
  y = x*x                 !  result 句で指定した変数に値を代入すると関数値になる
end function square
```

このように，function 文の末尾に “result(変数名)”を加えるのが result 句です．result 句で指定した変数に値を代入すれば，その値が関数値になります．このため，result 句の変数は関数値の型と同じ型宣言をしておく必要があります．

4.7 モジュールを使ったグローバル変数の利用

ローカル変数のおかげで各ルーチンの独立性は保たれますが，引数でしかルーチン間のデータ受け渡しができないのは不便です．引数は，かっこを使った並びで指定するので，受け渡す変数が多いと書くのが大変です．しかも並びが重要なので，順番を 1 カ所間違えただけでもエラーが発生します．そもそも引数による受け渡しはアドレス情報を経由したものなので，コール側とサブルーチン側で完全に同じ変数であるという保証はありません．

4.1 節で，計算機シミュレーションでは計算内容に応じてサブルーチンを作成すると述べましたが，この用途においては，ルーチン間で共有する変数が多数必要です．しかし，共有変数を全て引数にするのは効率が悪く，エラーも発生しやすくなります．

そこで，ルーチン内部でのみ意味を持つ “ローカル変数”に対して，“グローバル変数”が用意されています．グローバル変数とは，どのルーチンから参照しても共通した値を保持している変数のことで，Fortran では，モジュールと use 文の組み合わせで利用可能です[24]．

モジュールは，次のようにモジュール名を指定した module 文で開始して end module 文で終了します．モジュールの中で変数を宣言すれば，宣言された変数がグローバル変数として利用できます[25]．

```
module モジュール名
  宣言文 1
  宣言文 2
   .......
end module モジュール名
```

サブルーチンと同じ構造をしていることからわかるように，モジュールの記述は，メインプログラムやサブルーチンと同レベルです．たとえば，

```
module data1
  integer nmin,nmax
  real tinitial,amatrix(20,30)
end module data1
```

のように書きます．

モジュールはサブルーチンや関数副プログラムと異なり，プログラムに記述しただけではその内容を利用することができません．利用するルーチンの先頭に，use 文を使ってモジュール名を指定してその利用を宣言する必要があります．use 文は以下のような形式です．

```
use モジュール名
```

use 文は implicit 文よりも前に書かなければなりません．モジュールを使ったプログラムの一例を以下に示します．

[24] モジュールや use 文は Fortran90 から採用された比較的新しい仕様で，それ以前の Fortran では common 文でグローバル変数を実現していました．しかし，common 文には不便な点が多いので本書では説明を省略しています．

[25] モジュールはグローバル変数を用意するためだけに使うものではなく，サブルーチンや関数副プログラムを含めてプログラムパッケージにすることができます．詳しくは第 8 章で説明します．

```
module global
  real xaxis,yaxis
end module global

program stest4
  use global
  implicit none
  xaxis = 5.0
  yaxis = 100.0
  call subr4
  print *,xaxis,yaxis
end program stest4

subroutine subr4
  use global
  implicit none
  print *, xaxis,yaxis
  yaxis = 25.0
end subroutine subr4
```

モジュールの中で宣言された変数や配列は，メインプログラムやサブルーチンとは独立して存在したデータ領域にあり，use 文で利用を宣言すれば，どのルーチンからでも参照することができます．このプログラムのメモリイメージを図示すれば，図 4.7 のようになります．



図 4.7　モジュールで宣言されているグローバル変数の参照

モジュールは，名前が異なれば，一つのプログラムに複数作成することも可能であり，一つのルーチンが複数のモジュールを利用することも可能です．また，モジュールの中に use 文を書いて別のモジュールの変数を利用することもできます[26]．ただし，モジュールは use 文で利用を宣言するより前で (プログラムの上方で) 定義されている必要があります．このため，通常はこの例のように全てのルーチンより前に記述しておきます．

モジュールで宣言されている変数は，use 文を書くだけで利用できるという利便性がありますが，ルーチン内で明示的に宣言されていないので，不用意に使ってしまう可能性があります．そこで，次のような only 句を使って，ルーチン内で必要な変数だけに宣言を限定することができます．

[26] 一つの応用例として parameter 変数の利用があります．4.8.2 節を参照して下さい．

```
use モジュール名，only : 変数1，変数2，...
```

配列の場合には配列名だけを変数の位置に記述します．

たとえば，前の例で，

```
use global，only : yaxsis
```

のように書くと，yaxsis 以外の変数 (xaxsis) は，このルーチンでは宣言されていないのと同等です．宣言されていない変数は，ローカル変数として別途宣言することも可能です．

グローバル変数は，多用するとルーチンの独立性が失われてしまいます．基本的にはローカル変数でプログラムを作り，ルーチン間で共有する必要がある変数のみグローバルにしてください．また，グローバル変数は広い範囲で存在する可能性があるので，意味のある長めの名前を付けます．これは，1.6 節で述べた「大域的に有効な変数名の付け方」です．

4.8 プログラミング スマートテクニック (その 4)

サブルーチンを利用するときは，変数の使い分けが一つのポイントです．これまでローカル変数とグローバル変数の使い分けについて説明しましたが，ローカル変数にも 2 種類あり，これを使い分ければサブルーチンの動作をより細かく制御することができます．ここでは，ローカル変数の使い分け方と，プログラムのメンテナンスをさらに容易にしたり汎用性を持たせるための文法などを紹介します．

4.8.1 拡張宣言文とそれを用いたローカル変数の使い分け

これまで変数や配列の宣言に使っていた宣言文は，

```
型指定 変数1，変数2，...
```

という形式でした．ここで，“型指定”には“integer”とか，“real”のような数値型を記述し，“変数”には変数名か，配列名とその要素数を記述します．この宣言文は，データ領域におけるメモリの確保という意味もあります．

さて，Fortran90 から宣言文の記述形式が拡張されました[27]．拡張宣言文を使うと，メモリを確保すると同時に初期値を代入したり，メモリの特性を指定することができます．拡張宣言文は以下の形式です．

```
型指定［, 属性1，属性2，...］:: 変数1［=数値1］, 変数2［=数値2］，...
```

ここで，角かっこ，“［”と“］”は，この中が省略可能であるという意味で使っているだけなので，角かっこ自体は書かないで下さい．拡張宣言文を使用するときは，型指定部と宣言する変数や配列の間にコロン 2 個“::”を書く必要があります．また，属性には宣言する変数の特性を指定するための予約語を記述します．属性も数値も省略して，単に型指定と変数の間に“::”を書いただけの宣言文は，書かずに宣言した旧来の書式と同じ意味になります．

属性を書かずに，単に数値を代入した形で宣言すると，その変数に指定した数値を代入して，プログラムの動作が開始することを意味します．たとえば，

```
integer :: imax=10，jmax=100
real :: xx=1.0，yy=2.0
```

のように宣言すると，それぞれの変数に指定された値を代入した状態で動作が開始します．ただし，この宣言文中での数値代入は一度だけです．サブルーチン中で数値代入した変数を宣言しても，コールす

[27] 本書では，拡張された宣言文のことを“拡張宣言文”と呼んで，旧来の型指定だけの宣言文と区別します．

るたびに数値が代入されるのではありません．このため，その変数を実行文で変更すると，その変更した結果が残ります．たとえば，

```
program stest1
  implicit none
  call subr1
  call subr1
  call subr1
end program stest1

subroutine subr1
  implicit none
  integer :: n=1
  print *,n              ! コールするたびに n は増加する
  n = n + 1
end subroutine subr1
```

というプログラムでは，メインプログラムでサブルーチン subr1 が 3 回コールされていますが，サブルーチン中の print 文の出力は，1 回目が 1，2 回目が 2，3 回目が 3，となります．宣言文における代入は，プログラム開始時の初期値だと考えて下さい．

さて，上記のサブルーチンの動作を考えるとき，一つ注意しなければならないことがあります．4.4 節で，ローカル変数は各ルーチンのデータ領域に所属しているという説明をしましたが，もう少し詳しく言うと，サブルーチンのデータ領域には 2 種類あります．一つは，サブルーチンがコールされた時点で一時的に生成されるメモリ領域で，もう一つは，サブルーチンの呼び出しに関係なく常に確保されたメモリ領域です．本書では，前者を “一時メモリ領域”と呼び，後者を “固定メモリ領域”と呼ぶことにします．旧来の宣言文で宣言したローカル変数は，一時メモリ領域に所属します[28]．一時メモリ領域は，サブルーチンを呼び出すたびに場所や内容が変わる可能性があるので，同じサブルーチンを 2 回コールした場合，1 回目のコールでローカル変数に代入した値が，2 回目にコールした時点でそのまま残っているとは限りません．

たとえば，

```
program stest2
  implicit none
  call subr2(1)
  call subr2(2)
end program stest2

subroutine subr2(n)
  implicit none
  integer n
  real x,y
  if (n == 2) print *,x,y          ! この出力値は不定
  x = 10.0
  y = 100.0
end subroutine subr2
```

のようなプログラムを書いたとします．1 回目の call 文，call subr2(1) で，ローカル変数 x と y に，それぞれ 10.0 と 100.0 という値が代入されますが，2 回目の call 文，call subr2(2) を実行したときに，print 文で出力した x と y が 10.0 と 100.0 になる保証はないのです[29]．ローカル変数に代入した値を保証するには，固定メモリ領域に所属させなければなりません．拡張宣言文を使って初期値を代入したローカル変数は，一時メモリ領域ではなく，固定メモリ領域に所属します．最初のサブルーチン例，

[28] これに対し，メインプログラムの変数とモジュール内のグローバル変数は，宣言文の記述にかかわらず，全て固定メモリ領域に所属します．

[29] コンパイラによっては，全てのローカル変数を固定メモリ領域に所属させるものがあり，このプログラムでもうまくいく場合があります．しかし，文法的には保証されていないのですから，当てにするべきではありません．

subr1 の変数 n が，コールするごとに 1 ずつ増加した値を出力するのは，初期値 1 を代入して宣言することで固定メモリ領域に所属させたためです．

数値を代入せずに，ローカル変数を固定メモリ領域に所属させるには，変数に save 属性を指定します．たとえば，上記のサブルーチン subr2 を以下のように書き換えれば，2 回目のコールで 10.0 と 100.0 が出力されます．

```
subroutine subr2(n)
  implicit none
  integer n
  real, save :: x,y                ! save 属性を指定
  if (n == 2) print *,x,y
  x = 10.0
  y = 100.0
end subroutine subr2
```

変数の初期値が未定のときや，ローカル配列を固定メモリ領域に所属させたいときには save 属性を使って下さい．また，初期値を代入する場合でも，固定メモリ領域に所属させることを積極的に利用するのであれば，次のように save 属性を付加して，固定メモリ領域にあることを明示した方が良いと思います．

```
integer, save :: n=1
```

拡張宣言文の属性には save の他にも色々ありますが，ここでは配列を宣言するときに便利な dimension 属性を紹介しましょう．たとえば，10×10 の 2 次元実数型配列を 5 個用意するとき，通常の宣言では，

```
real a(10,10), b(10,10), c(10,10), d(10,10), e(10,10)
```

のように書きますが，dimension 属性を使えば，これを次のように書くことができます．

```
real, dimension(10,10) :: a, b, c, d, e
```

すなわち，dimension 属性に配列の形状 (次元や要素数) を書いておけば，宣言する変数の位置には，配列名を記述するだけになります．属性は複数付加することができるので，dimension 属性に save 属性を加えて宣言することも可能です．

4.8.2 parameter 変数の利用

配列を宣言するときに便利なのが，parameter 属性を指定した変数，parameter 変数です．parameter 変数には必ず値を代入しなければなりません．たとえば，

```
integer, parameter :: imax=10, jmax=200
```

のように書くと，imax と jmax が parameter 変数になります．

さて，配列を宣言するとき，その要素数は整数定数で指定しなければなりません．このため，配列の長さを変更するときには，宣言した要素数に関連した数値を全て変更する必要があり，手間がかかるだけでなく，見落としたり，書き間違える可能性があります．たとえば，

```
real a(100),b(100)               ! 100 が 2 カ所
integer i
do i = 1, 100                    ! 100 が 1 カ所
  a(i) = i*i
  b(i) = a(i)**2
enddo
```

というプログラムにおいて，配列要素数の 100 は，宣言文だけでなく，do 文の終了値にも入っています．このため，配列要素数を変更するときは 3 カ所変更しなければなりません．このとき，配列宣言の要素数を parameter 変数に代入する形で用意しておくと，配列要素数をその parameter 変数で置きかえることができます．たとえば，上のプログラムは次のように書き換えることができます．

```
integer, parameter :: imax=100
real a(imax),b(imax)
integer i
do i = 1, imax
  a(i) = i*i
  b(i) = a(i)**2
enddo
```

このプログラムならば，imax に代入する数値を変更するだけで，配列要素数の変更は完了です．

parameter 変数を含んだ計算式を別の parameter 変数への代入値にしたり，配列宣言の要素数に使うこともできます．これらは，その計算式の結果で宣言したことに相当します．たとえば，

```
integer, parameter :: imax=100, imax2=imax**2
real a(imax-1),b(0:imax*2),c(-imax2:imax2)
```

という宣言は，以下の宣言文と同等です．

```
integer, parameter :: imax=100, imax2=10000
real a(99),b(0:200),c(-10000:10000)
```

ただし，他の parameter 変数に代入するときは，代入される変数 (この例では imax2) が，代入する計算式に使う変数 (この例では imax) より後で宣言されなければなりません．

実は，parameter 属性を指定した変数は，“**変数＝数値**”の形式で変数名と数値の対応関係を示しているだけで，メモリとしての実体はありません．この対応関係を使った“変数”から“数値”への置き換えは，コンパイルの段階で行われるので，非実行文である宣言文でも使えるのです．逆に言えば，parameter 変数に実行文を使って数値を代入することはできません．たとえば，

```
integer, parameter :: imax=100
real a(imax),b(imax)
imax = 5                ! これはエラー
```

と書いた場合，“imax=5”はコンパイルエラーです．これは，プログラムに“100=5”と書いたことに等しいのですから当然だと言えます．

このように，parameter 変数を使えば配列要素数の変更が楽になりますが，それでもルーチンごとのローカルな宣言では，変更が必要になったときに使用している全てのルーチンで parameter 変数の設定値を変更しなければなりません．そこで，

```
module global_param
  integer, parameter :: imax=300, jmax=200
end module global_param
```

のように，parameter 変数を記述したモジュールを用意して，共有するルーチンで use 宣言をするのが良いと思います．use 宣言はモジュールの中でも使うことができるので，以下のように記述することが可能です．

```
module global_param
  integer, parameter :: imax=300, jmax=200
end module global_param

module mod_array
  use global_param                          ! モジュール内でも use 宣言可能
  real abc(imax,jmax),cd(jmax*2-1)          ! グローバル配列宣言
end module mod_array

program mtest1
  use mod_array                             ! use global_param は不要
  implicit none
  integer km(imax)                          ! ローカル配列宣言
   .......
```

ここで，あるモジュールの中で別のモジュールを use 宣言した場合，前者のモジュールを use 宣言したルーチンでは，後者のモジュールも合わせて use 宣言したことになります．上記のメインプログラム mtest1 ではモジュール mod_array しか use 宣言をしていませんが，mod_array の中で use 宣言をしているモジュール global_param も同時に宣言したことになり，その中にある変数 imax もメインプログラムで利用可能になります．

なお，次のようにメインプログラムに global_param の use 宣言を加えてその使用を明示してもエラーではありません．

```
program mtest1
  use global_param                          ! 書いてもエラーではない
  use mod_array
  implicit none
  integer km(imax)
   .......
```

これは，同じモジュールの use 宣言を重複して書いてもエラーにはならないからです．

parameter 変数は，数学定数や物理定数を用意するのにも便利です．筆者は次のような定数の入ったモジュールを作成して，必要なサブルーチンから利用するようにしています．

```
module constants
  real, parameter :: pi=3.141592653589793, pi2=pi*2
  real, parameter :: clight=2.99792458e8, hplanck=6.626070040e-34
  real, parameter :: emass=9.10938356e-31, pmass=1.672621898e-27
  real, parameter :: echarge=1.6021766208e-19, emc2ev=emass*clight**2/echarge
end module constants
```

parameter 変数は不用意に書き換えられることがないので，このような定数の定義にはうってつけです．また，コンパイル時に数値の置き換えをするのですから，代入という実行動作が不要になり，わずかですが計算時間の短縮になります．

4.8.3 include 文

プログラムが完成して，後は問題に応じて parameter 変数を変更するだけになると，parameter 変数の変更のためだけにプログラムファイルを修正するのは煩わしいし，変更するときに誤ってプログラムの内容を書き換えてしまうリスクもあります．プログラムを変更すると，プログラムファイルの日付が変更されるので，いつの時点で完成したプログラムかもわからなくなります．これらは，特に他人にプログラムを提供するときに問題になります．

そこで，parameter 変数の部分だけをプログラム本体から分離して保存し，必要に応じて結合すると便利です．結合する一つの手段に include 文の利用があります．include 文とは，以下のように include の後に，ファイル名を文字列で記述した文です．

```
include ’ファイル名’
```

include 文を書くと，コンパイラはその位置に“ファイル名”で指定したファイルの内容を挿入したプログラムを生成して，それをコンパイルします．たとえば，4.8.2 節のモジュール global_param を“global.inc”という名前のファイルに書き込んで保存しておけば，次のように記述することができます．

```
include ’global.inc’                    ! ここに global.inc の内容が挿入される

module mod_array
  use global_param
  real abc(imax,jmax),cd(jmax*2-1)
end module mod_array

program mtest1
  use mod_array
  implicit none
  integer km(imax)
   .......
```

include 文が指定するファイルの内容は，それを include 文の位置に挿入したときに，プログラム全体が正しく動作すればいいので，モジュールやサブルーチンのような完結したプログラムである必要はありません．また，同じファイルを一つのプログラムの複数の場所で include 指定することもできます．たとえば，

```
integer, parameter :: imax=300, jmax=200
```

だけをファイルに書いて，サブルーチンごとに include 指定することも可能です．

4.8.4 サブルーチンや関数副プログラムを引数にする方法

本章で説明したサブルーチンや関数副プログラムは，変数や配列などの数値を引数に持つものでした．これに対し，サブルーチンや関数副プログラムといった“外部副プログラム”を引数に持つサブルーチンや関数副プログラムを作ることも可能です．たとえば，方程式 $f(x) = 0$ を解くサブルーチンにおいて，関数 $f(x)$ を計算する関数副プログラムの名前も引数に加えておけば，コールするときにユーザーが関数を指定することができる，より汎用的なサブルーチンになります．

サブルーチン名を引数にしたサブルーチンを作るのは簡単で，引数並びの中にサブルーチン名を入れるだけです．たとえば，

```
subroutine subrout(subr,xmin,xmax,n)
   implicit none
   real xmin,xmax,dx,y
   integer n,i
   dx = (xmax-xmin)/n
   do i = 0, n
      call subr(dx*i+xmin,y)          !  引数 subr を使った call 文
      print *,i,y
   enddo
end subroutine subrout
```

のように書くことができます．この例では，subr が引数としてのサブルーチン名で，これをコールする文を書くことができます．関数副プログラムを引数にする場合も同様ですが，関数名の型宣言は必要です．たとえば，次のように書きます．

```
subroutine funcout(func,xmin,xmax,n)
   implicit none
   real func,xmin,xmax,x,dx,y          !  引数 func が関数なので型宣言をする
   integer n,i
   dx = (xmax-xmin)/n
   do i = 0, n
      x = dx*i + xmin
      y = func(x)**3                   !  引数 func を関数として使う
      print *,x,y
   enddo
end subroutine funcout
```

このように，外部副プログラムを引数に持つサブルーチンを作るのは簡単ですが，このサブルーチンを使用するときは，「引数が変数ではなく，外部副プログラムである」という宣言が必要です．この宣言は external 文で行います．たとえば，

```
subroutine sub(x,y)
   implicit none
   real y,x
   y = 2*sin(x) + cos(x**2)
end subroutine sub

function fun(x)
   implicit none
   real fun,x
   fun = sin(x)**3
end function fun
```

というサブルーチンや関数副プログラムを引数にして，上記のサブルーチン (subrout や funcout) をコールするときは，

```
program test_func
   implicit none
   external fun,sub                     !  external 文を使って宣言する
   call subrout(sub,0.0,3.0,10)
   call funcout(fun,0.0,3.0,10)
end program test_func
```

のように書きます．external 文による宣言がないと，sub や fun という単語が “変数”と見なされて，型宣言されていないというエラーになります[30]．external 文は非実行文なので，変数の型宣言と同様に，全ての実行文より前に書く必要があります．

さて，なぜこの程度で外部副プログラムを引数にできるのか，しくみを簡単に説明しましょう．4.4 節で述べたように，Fortran では引数に変数を与えるとき，変数の内容を与えるのではなく，そのアドレスを与えます．一方，変数にアドレスがあるのと同様に，プログラムにもアドレスが付いています．これは，プログラムが保存されているプログラム領域でのメモリ番地のことです．goto 文での無条件ジャンプや if 文での分岐，サブルーチンコールなど，必要に応じて実行の流れを変えるときには，プログラムのアドレスを指定してジャンプします．プログラムも変数も，場所は違いますが，コンピュータのメモリ上にあるという意味では同じです．よって，引数に変数を与えるのも，サブルーチン名を与えるのも，メモリアドレスを与えるという意味では同じであり，特に書式を変更する必要はありません．

しかし，プログラム名が指定するアドレスと変数や配列名が指定するアドレスはメモリ領域が異なるので区別しなければなりません．外部副プログラムを引数として持つサブルーチンの内部では，その副プログラムを使用している記述があるはずなので，それを判定材料として引数がプログラムアドレスか

[30] ちなみに，サブルーチン funcout に与えている関数名 fun は型宣言をしていませんが，これはエラーになりません．ただし，このルーチン内で fun を関数として計算に使う場合には型宣言をする必要があります．

変数アドレスかを区別しています．一方，このサブルーチンを使う call 文側では，外部副プログラムの名前しか引数に書けないので，それだけで変数か外部副プログラム名かを判定することはできません．この区別のために用意されているのが external 文です．external 文を使って宣言された名前は，サブルーチン名または関数副プログラム名と判定されてコンパイルされます．なお，external 文による宣言を忘れてコンパイルエラーになったとき，エラーを防ぐために sub や fun を real や integer などで型宣言すると，エラーは出ず，コンパイルもリンクも成功します．しかし，実行しても正常な動作はしないので注意が必要です．

なお，関数を引数に持つサブルーチンに sin や exp のような組み込み関数名を与えてコールするときは，external 文の代わりに intrinsic 文で宣言します．たとえば，

```
program test_sin
   implicit none
   intrinsic sin
   call funcout(sin,0.0,3.0,10)
end program test_sin
```

のように書きます．これは，Fortran における組み込み関数が宣言をしなくても使える特別なものだからです．しかし，それなら宣言せずとも使えるようにしてくれればいいと思うのですが，intrinsic 宣言をしておかないと「sin という変数が宣言されていない」というエラーになります．このあたりは Fortran コンパイラの仕様の問題だと思います．

4.8.5 時間計測用サブルーチン

プログラムによっては，日付を取得してデータ保存の際に記録したり，計算時間を測定してアルゴリズムの性能を調べたいときがあります．時間計測は計算機のハードウェアに依存する作業なのですが，現在では計算機が持つ時刻情報を取得するサブルーチンが標準で用意されているので，Fortran にも標準の時間計測ルーチンが付属しています．

まず，日付と時刻を取得する場合には，サブルーチン date_and_time を使います．この関数の呼び出し方は以下の通りです．

```
   call date_and_time(d,t,z,v)
```

ここで，引数 d，t，z，v に表 4.2 に示す型の変数を与えると，コールした時点での日時の情報をそれぞれに代入して戻ります．なお，型に付随している数値は，文字列の場合は文字数，整数の場合は整数配列の要素数の必要最小値です．

表 4.2 date_and_time の引数

引数	引数名	型 (要素数)	意味	出力例	出力の意味
d	date	文字列 (8)	年月日	20241009	2024 年 10 月 09 日
t	time	文字列 (10)	時刻 (時分秒)	183541.308	18 時 35 分 41.308 秒
z	zone	文字列 (5)	標準時との差	+0900	+09 時間 00 分
v	values	整数型 (8)	整数化した時刻データ	【本文参照】	

ここで，整数型配列 v(10) には，以下の整数値が代入されます．

v	年	月	日	時差	時	分	秒	秒/1000

たとえば，表 4.2 の出力例の場合には，v(1)〜v(10) の値が以下のようになります．

```
    2024,   10,    9,    540,    18,    35,    41,    308
```

ここで，時差 (標準時との差) は分単位の数値が代入されるので，日本時間である +9 時間の場合には 540 になります．

このサブルーチンでは，引数の中で不要なものを省略することができます．省略するときは，引数名に用意した変数を代入する形でコールします．たとえば，values の整数データのみを利用したい場合には，

```
integer val(8)
call date_and_time(values=val)
print *,val
```

のように書くことができます．

date_and_time は日時を調べるときには便利ですが，計算時間を計りたいときには少し不便です．この場合には，サブルーチン cpu_time を使う方が良いでしょう．cpu_time は実数型の変数を 1 個与えてコールします．

```
call cpu_time(sec)
```

戻り値 sec には秒単位の時刻が代入されます．よって，以下のようにすれば計算に要した時間を計測することができます．

```
real sec1,sec2
call cpu_time(sec1)
  計測したい処理
     :
     :
call cpu_time(sec2)
print *,sec2-sec1
```

すなわち，計測したい処理の前後で cpu_time をコールして，2 個の戻り値の差を計算すれば，秒単位の処理時間を得ることができます．

ただし，cpu_time は Fortran95 から用意されているサブルーチンです．それ以前のバージョンを使っている場合には，サブルーチン system_clock を使って下さい．system_clock の呼び出し方は以下の通りです．

```
call system_clock(tc,trate,tcmax)
```

ここで，引数の意味は表 4.3 の通りです．

表 4.3　system_clock の引数

引数	引数名	型	引数の意味
tc	count	整数型	時刻カウント数
trate	count_rate	整数型	1 秒間のカウント数
tcmax	count_max	整数型	カウント数の最大値 時刻カウント数が最大値を超えると 0 に戻る

これらの引数は全て戻り値なので，変数で与える必要があります．また，一部を省略することも可能です．たとえば，時間を秒単位で計測したいときには次のように書くことができます．

```
integer tc1,tc2,trate
call system_clock(tc1,trate)
  計測したい処理
     :
     :
     :
call system_clock(tc2)
print *,real(tc2-tc1)/trate
```

すなわち，system_clock を 2 回コールして，そのカウント数の差を計算し，その差を 1 秒間のカウント数で割れば，秒単位の経過時間が得られるというわけです．なお，この例のように後の引数を省略する場合には引数変数の指定だけでいいのですが，前の引数を省略したいときには，date_and_time のように，引数名で指定します．たとえば，trate のみ必要な場合には，

```
call system_clock(count_rate=trate)
```

のように書きます．

なお，cpu_time で測った経過時間と system_clock で測った経過時間とは必ずしも一致しません．cpu_time はその名の通り，CPU が働いた時間を測るものなので，計算に関係のない時間は含まれません．これに対し，system_clock は内部時計による経過時間を測るものなので，実際の経過時間になります．このため，一般的には cpu_time で測った時間の方が短くなります．たとえば，read 文 (5.4.1 節) を用いてデータ入力をした場合，system_clock で測ると入力に要した時間も含めた経過時間になりますが，cpu_time の場合には，待ち時間が含まれないので，経過時間はかなり短くなります．

4.8.6 乱数発生用サブルーチン

世の中の現象は全てが定まった法則によって動いているとは限りません．知らない道を歩いていて分岐点に出くわしたとき，どっちに進むかをコイントスで決めることもあれば，すごろくのようにサイコロを振って次の一手を決めることもあります．

このような，事象の発生が確率的にしかわからない現象を計算機で調べるとき，さいころの代わりをするのが “乱数”です．Fortran には，乱数を発生するサブルーチン random_number が用意されているので，これを用いれば乱数を使った計算ができます．たとえば，1 個の乱数を使うときには

```
call random_number(x)
```

と書けば，変数 x に $0 \leqq x < 1$ の範囲の実数が代入されます．このサブルーチンの戻り値 x には規則性が無く，同じ呼び出しをくり返しても，それ以前に発生させた値との相関はありません．

乱数を 1 回で複数用意したいときには配列を使います．たとえば，実数配列に乱数を代入するときは次のように書きます．

```
real a(20)
call random_number(a)          ! 配列 a の全ての要素に乱数を代入
call random_number(a(1:10))    ! a(1)〜a(10) に乱数を代入
```

random_number は，実際に雑音のような規則性のないもので値を決めているのではなく，ある一定の公式を使ってできるだけ乱数に近い数字列を計算しているものです．これを “擬似乱数”といいます．擬似乱数には初期値が必要で，初期値が同じ場合には同じ乱数列を発生します．このため，プログラムを実行するごとに異なる乱数列になるとは限りません．そこで，乱数の初期値を調べたり，自分で決めるためのサブルーチン random_seed が用意されています．random_seed の使い方は次の 4 通りです．

```
call random_seed              ! 初期値をシステムが設定する
call random_seed(size=n)      ! 初期値のサイズを変数 n に代入する
call random_seed(put=seed)    ! 初期値を整数配列 seed(n) に格納した値に設定する
call random_seed(get=seed)    ! 初期値を整数配列 seed(n) に代入する
```

random_seed は，引数を同時に 2 個以上指定することができないので注意して下さい．

一般に，初期値は乱数発生のアルゴリズムに依存するので 1 個とは限りません．そこで，初期値を設定するときには，まず二番目の size を使った call 文により個数を取得します．その後，その個数を要素数に持つ整数配列を用意して，get を使えば現在の初期値が得られ，put を使えば初期値を任意に設定して，異なる乱数列を発生させることができます．

たとえば，4.8.5 節で紹介した時刻計測ルーチンを使って，次のようなプログラムにすれば，実行の度に異なる乱数を発生させることができます．

```
real a(10)
integer n,val(8),seed(10)
call date_and_time(values=val)
call random_seed(size=n)
call random_seed(get=seed(1:n))
seed(1) = ((val(8)*100 + val(7))*100 + val(6))*100 + val(5)
call random_seed(put=seed(1:n))
call random_number(a)
```

このプログラムでは，date_and_time で得られた時分秒の数値から 1 個の整数を作り，これを get で得られた初期値配列の一番目に格納して与えています．こうすることで，毎回異なる初期値を使うので，異なる乱数が発生するというわけです．

なお，このプログラムでは配列 seed の要素数を 10 個にしていますが，初期値の数はコンパイラに依存するので 10 個では足らない場合があります．より可般性を持たせたいときには，配列の動的割り付けなどを利用すればいいでしょう (7.6 節参照)．

演 習 問 題 4

(4–1)　2 次方程式の解 (サブルーチン利用)

3 個の実変数 a，b，c を引数とし，それから，

$$ax^2 + bx + c = 0$$

の 2 解 x_1，x_2，および判別式 $D = b^2 - 4ac$ を計算して戻り値にするサブルーチンを作成せよ．ただし，2 実解になる場合は，x_1，x_2 をそれぞれ戻り値変数 `x1` と `x2` に代入し，2 虚解になる場合は，実部を `x1` に，虚部を `x2` に代入するようにせよ．

次にそれを使って問題 (3–1) の解答をサブルーチンを使う形に修正せよ．

(4–2)　ヘロンの公式 (関数副プログラム利用)

三角形の三辺の長さを a，b，c とすると，その三角形の面積 S は次式（ヘロンの公式）で表される．

$$S = \sqrt{p\,(p-a)(p-b)(p-c)}$$

ただし，$p = \dfrac{a+b+c}{2}$ である．

a，b，c の数値を引数とし，この公式を使って三角形の面積を計算して関数値にする関数副プログラムを作成せよ．

次にそれを使って問題 (1–2) の解答を関数副プログラムを使う形に修正せよ．

(4–3)　3 次元ベクトルと電磁気学 (サブルーチン利用)

3 次元ベクトル $\boldsymbol{A} = (A_1, A_2, A_3)$ を要素数 3 の 1 次元配列 `a(3)` で表すとする．電場ベクトル $\boldsymbol{E}$，磁場ベクトル $\boldsymbol{B}$ および荷電粒子の速度ベクトル $\boldsymbol{v}$ を配列で表して，これらの配列を引数として与えると内積 $\boldsymbol{v} \bullet \boldsymbol{E}$，および外積ベクトル $\boldsymbol{v} \times \boldsymbol{B}$ を計算して戻り値とするサブルーチンを作成せよ．なお戻り値の外積ベクトルも配列とする．これを使って問題 (2–2) の解答をサブルーチンを使う形に修正せよ．

(4–4)　統計計算 (サブルーチン利用)

要素数 n の 1 次元配列，`a(n)` を引数とし，

$$\text{平　均}\ \ \bar{A} = \frac{1}{n}\sum_{i=1}^{n} A_i$$

$$\text{標準偏差}\ \ \sigma = \sqrt{\frac{1}{n}\sum_{i=1}^{n}(A_i - \bar{A})^2}$$

を計算して戻り値とするサブルーチンを作成し，問題 (2–4) の解答をサブルーチンを使う形に修正せよ．

(4–5)　非線形方程式の解法：Newton 法 (サブルーチン利用)

問題 (3–3) の Newton 法を用いて方程式の解を計算するプログラムをサブルーチンを使う形に修正せよ．まず，変数値 x を与えると関数値 $f(x)$ とその微分値 $f'(x)$ を返すようなサブルーチンを作成する．それを使って Newton 法で解を求めるプログラムにすればよい．

(4–6)　行列計算

n 行 n 列の行列 a_{ij} を 2 次元配列 `a(n,n)` の要素 `a(i,j)` で表すとする．2 個の行列 a_{ij} と b_{ij} から行列の和，c_{ij}，および行列の積，d_{ij} を計算するサブルーチンを作成せよ．このとき，整合配列を用いて与える配列の次元が自由に変えられるようにすること．

ここで，行列の和と積の定義は以下の通りである．

$$c_{ij} = a_{ij} + b_{ij}$$
$$d_{ij} = \sum_{k=1}^{n} a_{ik} b_{kj}$$

(4–7)　行列式

`u(3)`，`v(3)`，`w(3)` という 3 つの 1 次元配列データから行列式

$$Det = \begin{vmatrix} u(1) & v(1) & w(1) \\ u(2) & v(2) & w(2) \\ u(3) & v(3) & w(3) \end{vmatrix}$$

を計算して関数値とする関数副プログラムを作成せよ．ここで，行列式 Det を展開すると，以下のようになる．

$$\begin{aligned} Det = {} & u(1)v(2)w(3) + u(2)v(3)w(1) + u(3)v(1)w(2) \\ & - u(3)v(2)w(1) - u(2)v(1)w(3) - u(1)v(3)w(2) \end{aligned}$$

(4–8)　3 元連立 1 次方程式の解法

3 元 1 次の連立方程式,

$$\begin{aligned} a_{11}x_1 + a_{12}x_2 + a_{13}x_3 &= b_1 \\ a_{21}x_1 + a_{22}x_2 + a_{23}x_3 &= b_2 \\ a_{31}x_1 + a_{32}x_2 + a_{33}x_3 &= b_3 \end{aligned}$$

を解くプログラムのサブルーチンを作成せよ．

クラメールの公式によれば，答えは以下のようになる．

$$x_1 = \frac{\begin{vmatrix} b_1 & a_{12} & a_{13} \\ b_2 & a_{22} & a_{23} \\ b_3 & a_{32} & a_{33} \end{vmatrix}}{D}, \qquad x_2 = \frac{\begin{vmatrix} a_{11} & b_1 & a_{13} \\ a_{21} & b_2 & a_{23} \\ a_{31} & b_3 & a_{33} \end{vmatrix}}{D}, \qquad x_3 = \frac{\begin{vmatrix} a_{11} & a_{12} & b_1 \\ a_{21} & a_{22} & b_2 \\ a_{31} & a_{32} & b_3 \end{vmatrix}}{D}$$

ただし,

$$D = \begin{vmatrix} a_{11} & a_{12} & a_{13} \\ a_{21} & a_{22} & a_{23} \\ a_{31} & a_{32} & a_{33} \end{vmatrix}$$

である．

プログラムは，3 行 3 列の行列 a_{ij} を 2 次元配列 `a(3,3)` の要素 `a(i,j)` で表し，b_i を `b(i)` という 1 次元配列で表し，x_i を `x(i)` という 1 次元配列で表して，サブルーチンの引数は，`a` と `b` と `x` とする．行列式の計算には，問題 (4–7) で作成した関数副プログラムを利用すればよい．

(4–9)　モンテカルロ法

世の中の動きは，運動方程式のような微分方程式を用いて確実に予測できるとは限らない．たとえば，さいころを投げて次にどの数字が出るかで進み方を決めるすごろくのように，進む先が確率的にしか決められない事象の積み重ねを計算することで未来を予測する方法をモンテカルロ法という．モンテカルロ法により円周率 π を計算するプログラムを作成してみよう．

計算機でさいころの代わりをするものは 4.8.6 節で説明した乱数である．まず，乱数発生用サブルーチン `random_number` を使って 2 個の乱数 x と y を発生させて 2 次元座標点 (x, y) を作る．この乱数の座標点を N 個作り，その中で原点から半径 1 の円より内側にある点の数 N_1 を数えて $p = \dfrac{4N_1}{N}$ を計算する．座標点数 N が増えると，この p が π に近づくことを確かめよ．

第 5 章　データ出力の詳細とデータ入力

プログラムは計算をさせるだけでは意味がありません．結果を出力して初めて完結します．これまでは，最も単純な print 文でとりあえず画面に表示する方法を使ってきましたが，本章では好みに応じて数値の出力形式を変える方法や，ファイルに保存する方法について説明します．さらにデータを入力することでプログラムを変更せずに動作を変える方法についても説明します．

5.1　データ出力先の指定

これまでもたびたび登場しましたが，最も単純な出力命令は，print 文です．

```
print form, データ 1, データ 2,  ...
```

print 文で出力すると，画面 (正確には標準出力) にデータが表示されます．*form* は出力形式を指定するためのものですが，とりあえず “*” を書いておけば，データの数値型に応じた標準形式で表示します．

たとえば，

```
print *, x, y(i), x**2+5, '   abc = ', 10
```

のように，データの位置には，変数や配列を与えても良いし，計算式や定数を与えることも可能です．また，文字列を適当に入れて，データの意味を並記することもできます．

画面ではなく，ファイルに出力するときは write 文を用います．print 文との違いは，出力先を指定する整数値 *nd* と出力形式指定の *form* をかっこで記述することです．

```
write(nd, form) データ 1, データ 2,  ...
```

nd を “装置番号” といいます．装置番号は，出力ファイルを識別する数字で，任意に選ぶことができます．また，変数や整数値を結果とする計算式を与えることもできるので，条件に応じて出力先を変更することも可能です．*nd* に適当な整数を与えて write 文を実行すると，“fort.*nd*” という名のファイルに出力されます[31]．たとえば，

```
write(20,*) x,y,z
```

と書けば，x,y,z の値が “fort.20” という名前のファイルに出力されます．この例のように，*form* に “*” を与えれば，print 文と同じく，データの数値型に応じた標準形式で出力します．

なお，原則として $nd \geqq 10$ にして下さい．これは，Fortran コンパイラの仕様によって，1 桁の数値は予約されている可能性があるからです．たとえば，$nd = 6$ にすれば print 文と同じ標準出力の指定になるので，write(6,*) と書けば，ファイルへの出力はなく，画面に表示されます[32]．

5.2　配列の出力，do 型出力

出力文において，データの位置に配列名を書けば，全要素が並んで出力されます．たとえば，

[31] “fort” の部分はコンパイラに依存しますが，多くは fort のようです．ファイル名は 5.6 節で説明する open 文を使えば変更することができます．また，コンパイラによっては，open 文を使わなくても装置番号に対応する環境変数の指定で，任意の名前を持つファイルに変更することが可能です．

[32] *nd* として，“*” を記述することもでき，これも標準出力指定です．すなわち，Fortran での標準出力には “print *form*,”，“write(6,*form*)”，“write(*,*form*)” の 3 通りがあります．

```
real a(4)
do i = 1, 4
  a(i) = i
enddo
print *,a
```

というプログラムを実行すると，

```
  1.00000000000000   2.00000000000000   3.00000000000000   4.00000000000000
```

のように，配列要素が先頭から順に並んで出力されます．このとき，配列要素が多いと適当に改行を入れて出力されます．よって，この単純な出力は配列要素が少ないとき以外はあまりお勧めできません．

たとえば，2 次元配列を，

```
real a(3,3)
do j = 1, 3
  do i = 1, 3
    a(i,j) = i*j
  enddo
enddo
print *,a
```

のように出力すると，a(1,1),a(2,1),a(3,1),a(1,2),...,a(3,3) という並びで 9 個の要素が出力されます．そこで，次のように do 文を使って出力する方がいいでしょう．

```
do j = 1, 3
  do i = 1, 3
    print *,a(i,j)
  enddo
enddo
```

しかし，print 文や write 文は，1 文ごとに改行をするので，複数の print 文や write 文を使って，横に続けて書くことはできません[33]．よって，この例の出力では全部で 9 行必要です．行数を減らすために，1 行に行列の 1 行を出力するには，次のように書く必要があります．

```
do i = 1, 3
  print *,a(i,1),a(i,2),a(i,3)
enddo
```

しかし，この例のように列の数がわかっているときは良いのですが，列数が多いときや変数で指定したいときには不便です．そこで do 型出力が用意されています．do 型出力とは，

```
（データ 1, データ 2,..., 整数型変数 = 初期値，終了値）
```

のような，かっこで囲んだ形式です．これをデータの位置に記述すれば，do 文のように，整数型変数が初期値から終了値まで 1 ずつ増えていき，その変数で指定されるデータが横に並んで出力されます．do 型出力を複数並べたり，配列以外のデータと並べて出力することもできます．たとえば，

```
print *,(a(i,j),j=1,3)
```

は，さきほどの “print *,a(i,1),a(i,2),a(i,3)” と書くことと同等です．また，

```
print *,5*x,(i,a(i),b(i),i=1,3)
```

は，“print *,5*x,1,a(1),b(1),2,a(2),b(2),3,a(3),b(3)” と書くことと同等です[34]．

[33] format を利用すれば続けて書くことは可能です．5.3 節を参照して下さい．

[34] なお，終了値＜初期値となる値を与えた場合には，数値は一つも出力されません．ただし，改行は行います．

do 文と同様に，3 番目の数として増分値を付加することもできます．

```
（データ 1, データ 2, . . . , 整数型変数 = 初期値，終了値，増分値）
```

たとえば，

```
   print *,(a(i),i=10,1,-2)
```

は，“print *,a(10),a(8),a(6),a(4),a(2)”と書くことと同等です．

do 型出力は，“データ”の位置に do 型出力を入れて多重にすることもできます．たとえば，

```
   print *,((a(i,j),j=1,3),i=1,3)
```

は，“print *,a(1,1),a(1,2),a(1,3),a(2,1),a(2,2),a(2,3),a(3,1),a(3,2),a(3,3)”と書くことと同等です．この例のように，多重のときには内側の整数型変数 (この例では j) が先に進みます．

なお，do 型出力の整数型変数は，do 文のカウンタ変数に相当します．このため，do 型出力の入っている出力文を do ブロックの中に入れるときには，do 文のカウンタ変数とも重複しないようにしなければなりません．

5.3 format による出力形式の指定

標準出力形式 (“*”指定) で実数を画面に出力すると，有効数字 15 桁の数字を使って表示されます．このため，あまり多くのデータを横に並べることはできないし，結果の確認だけなら，それほど有効数字は必要ありません．また，標準形式の出力には 1 行の出力文字数に制限があるため，出力数が制限を超えると自動的に改行されてしまいます．このため，同じ形式の出力をくり返しても，行ごとに小数点の位置が違うこともあり，大量に出力するのには向きません．

これらの問題は，出力形式 (format) を指定することで解決することができます．これまで“*”を書いていた *form* の位置に format を指定すれば，小数点以下の桁数を小さくしたり，必要に応じて数字と数字の間にスペースを空けたり，改行を入れたりすることができます．また，自動改行されることがないので，幅広く出力することも可能です．

5.3.1 format の指定方法

print 文や write 文における format の指定方法は 2 通りあります．まず，format 文による指定です．一例を示します．

```
   real x,y
   integer n
   x = 1.5
   y = 0.03
   n = 100
   print 600,x,y,n                ！  600 は format 文の文番号
600 format(’  x = ’,f10.5,’   y = ’,es12.5,’   n = ’,i10)
```

最後の文が format 文です．format 文は，サブルーチンの引数のように，出力形式の指定をリストにしてかっこで囲み，先頭に文番号を付けます．この文番号を print 文や write 文の *form* に指定すれば，その format にしたがってデータが整形されて出力されます．この例では，600 が *form* 指定です．文番号は重複できないので，ルーチン内では format 文ごとに異なる数字をつけなければなりません[35]．

一方，複数の print 文や write 文が同じ format 文を指定するのは可能です．たとえば，次のように共用することができます．

[35] 文番号の付け方に決まったルールはありませんが，“format 用である”という意味をどこかに入れておいた方が良いでしょう．この例で 600 を使っているのは，“6”が昔の標準出力装置番号だったころの慣習です．

```
  real x,y,u,v
  integer n,k
   .......
  print 600,x,y,n
  write(20,600) u,v,k
 600 format('  x = ',f10.5,'   y = ',es12.5,'   n = ',i10)
```

format 文は，書式を示すだけで，コンピュータの動作はありません．このため，記述の位置には無関係で，非実行文より後でさえあれば，指定する print 文や write 文より前に書くことも可能です．

format を指定するもう一つの方法は，文字列を使って *form* の位置に直接 format の内容を記述することです．たとえば，上記の format を print 文や write 文に埋め込んで，

```
  print "(' x = ',f10.5,'   y = ',es12.5,'   n = ',i10)",x,y,n
  write(20,"(' x = ',f10.5,'   y = ',es12.5,'   n = ',i10)") u,v,k
```

のように書くことができます．このとき，“format”の文字は不要ですが，両端のかっこは必要です．なお，Fortran での文字列は「'」で囲むのが基本ですが，「"」も使えるので，format 内部に文字列が入っているときには，この例のように format の両端を「"」で囲みます．

format を出力文中に文字列で書き込む方法は，文番号を必要としない点は良いのですが，出力文が長くなるし，同じ format を何度も使うときには不便です．そこで，文字列変数を利用して書く方法があります．文字列変数を利用する方法については 6.3 節で説明します．

5.3.2 format の書き方と編集記述子

それでは format の書き方を説明しましょう．まず，format 中の文字列はそのまま出力されるので，必要に応じて適宜挿入します．たとえば，前節の format では，' x = 'や，' y = 'は，スペースも含めてそのまま出力されます．

次に，出力文中のデータ値の並びに対し，それぞれの出力形式を指定する“編集記述子”を選んで，前から順に記述します．前節の format では，f10.5，es12.5，i10，が編集記述子で，print 文の並びに対し，

```
  print 600,          x        ,      y       ,       n
                      ↓                ↓               ↓
600 format('  x = ',f10.5,'   y = ',es12.5,'   n = ',i10)
```

という対応で出力形式を指定しています．文字列と編集記述子で指定したデータ値は，その並び順に出力されます．よって，この print 文を実行したときの出力は以下のようになります．

```
     x =    1.50000   y =  3.00000e-02   n =        100
```

出力形式を指定する編集記述子の主要なものを表 5.1 に示します．データの数値型に応じた編集記述子を使用しないと正しい値が出力されないので注意して下さい．なお，表 5.1 で斜体文字 (w, m, d) は整定数で指定します．また，編集指定の文字 (F や ES など) を指定数 (w など) と区別するために大文字で書きましたが，小文字でも同じ意味です．たとえば，i10 は整数型値を幅 10 文字で出力することを意味し，f10.5 は実数型値を幅 10 文字，小数点以下 5 桁で出力することを意味しています．このため，

```
  real x,y
  integer m,n
  x = 1.5
  y = 0.03
  m = 100
  n = 10
  print "(f10.5,f10.5,i10,i10.5)",x,y,m,n
```

というプログラムの出力は,

```
     1.50000    0.03000        100      00010
+----+----+----+----+----+----+----+----+
```

となります．2 行目の目盛りは位置を確認するために書いたものですが，10 文字の中に右寄せで出力されているのがわかります[36]．なお，出力文字数が指定の幅 w を越えると，“*****”のように“*”が w 個出力されます.

表 5.1　主要な編集記述子

編集指定	数値の型	編集の意味
Iw	整数	幅 w で整数を出力する 注
I$w.m$	整数	幅 w で整数を出力する 注 出力整数の桁が m より小さいときには, 先頭に 0 を補う ($w \geqq m$)
F$w.d$	実数	幅 w で実数を固定小数点形式で出力する d は小数点以下の桁数 ($w \geqq d+3$)
E$w.d$	実数	幅 w で実数を浮動小数点形式で出力する d は小数点以下の桁数 ($w \geqq d+8$) 仮数部の 1 桁目は 0 になる
G$w.d$	実数	幅 w で実数を固定小数点形式または浮動小数点形式で出力する どちらになるかは，実数の指数部の大きさで決まる d は小数点以下の桁数
ES$w.d$	実数	幅 w で実数を浮動小数点形式で出力する (d は E 編集と同じ) 0 以外の数値を出力すると，仮数部の 1 桁目は 1 から 9 になる
EN$w.d$	実数	幅 w で実数を浮動小数点形式で出力する (d は E 編集と同じ) 0 以外の数値を出力すると，仮数部の整数は 1 以上 1000 未満となり，指数部は 3 で割り切れる数になる
A	文字列	文字列をそれ自身の長さの幅で出力する
Aw	文字列	幅 w で文字列を出力する
Lw	論理型	幅 w で論理型値を出力する (T または F)

注 I 編集の I の代わりに B を使えば 2 進数表示になり，Z を使えば 16 進数表示になります.

E 編集を使って実数を浮動小数点形式で出力すると，小数点の前が 0 になります．たとえば,

```
real x,y
x = 1.5
y = 3.14e10
print "(e15.5,e15.5)",x,y
```

の出力結果は,

```
    0.15000e+01    0.31400e+11
```

となります．これでは感覚的にわかりにくいし，表示字数が 1 個無駄になります．そこで，ES 編集や EN 編集を使う方が良いでしょう．たとえば,

[36] A 編集による文字列の出力の場合は左寄せになります.

```
real x
x = 3.14e10
print "(es15.5,en15.5)",x,x
```

の出力結果は,

```
   3.14000e+10   31.40000e+09
```

となります.

各編集記述子と出力の数値は 1 対 1 対応にしなければならないので，配列を出力するときには出力要素数と同数の編集記述子を書かなければなりません．このとき，同じ編集記述子をくり返すならば，編集記述子の前に整数 r を付加して，“r 回反復する”という指定ができます．たとえば，“3f10.5”は“f10.5,f10.5,f10.5”と書くことと同等です.

さらに，実数，整数，実数，整数のようなくり返しのときには，かっこで囲んで反復指定をすることができます．たとえば，次のように書くことができます.

```
real x,y
integer m,n
x = 1.5
y = 0.03
m = 5
n = 100
print "(2(f10.5,'  ',i7))",x,m,y,n
```

この print 文の format は，“f10.5,' ',i7,f10.5,' ',i7”と書くことと同等です.

複素数は，“実部，虚部”という実数のペアであり，計算機内部的には要素数 2 の 1 次元配列と同型です．このため，複素数を出力するときは，複素数 1 個あたり，実数の編集記述子を 2 個並べる必要があります．たとえば，以下のように書きます.

```
complex c
c = (1.0,-2.0)
print "(4f8.3)",c,c**2
```

この出力結果は,

```
   3.000  -2.000   5.000 -12.000
```

となります．しかし，これでは複素数という感じが出ないので，少し工夫して,

```
complex c
c = (1.0,-2.0)
print "(2(f8.3,' + (',f8.3,')i   '))",c,c**2
```

などとしてみれば良いでしょう．この出力結果は,

```
   3.000 + (  -2.000)i      5.000 + ( -12.000)i
```

となります.

なお，format 中の編集記述子の数よりも出力文のデータ値の方が多い場合には，編集記述子の数だけ出力した後で改行し，同じ format を再度使って残りのデータ値を出力します．文字列が入っていれば，文字列も再度出力されます.

逆に，format 中の編集記述子の数よりも出力文のデータ値の方が少ない場合には，対応するデータ値がないと検出された時点で出力を終了し，それ以降の編集記述子は無視されます．無視された記述子以降は文字列等が入っていても全て無視されます．そこで，配列を出力するときなどは，反復指定に大きめの数値を与えておくことができます．たとえば,

```
real a(4)
integer m
a(1) = 2.25
a(2) = 30.2
a(3) = 400.7
a(4) = 5000.6
print "(10(f10.2,'cm  '))",(a(m),m=1,4)          !  反復指定は 10
```

のように，反復指定を 10 回にしておいても，出力結果は，

```
    2.25cm      30.20cm      400.70cm     5000.60cm
```

となります．

文字列のように出力文中のデータ値との対応がない編集記述子もあります．代表的なものを表 5.2 に示します．ここで，r は整定数で指定します．たとえば，2 次元配列 a(3,3) の配列要素を 3 行 3 列の行列のように 1 行あたり 3 個ずつ出力するときは，スラッシュ (/) 編集を使って，

```
real a(3,3)
 .......
print "(3(3f12.5/))",((a(i,j),j=1,3),i=1,3)
```

と書くことができます．

表 5.2　出力文中のデータ値との対応がない編集記述子

編集指定	編集の意味
/	改行する
r/	r 回改行する
rX	r 個スペースを挿入する
$	print 文や write 文終了時の改行を抑制する
:	出力文中の数値の出力が終わった時点で format 中の以後の出力を打ち切る

X 編集は出力中に適当な数のスペースを入れるときに使います．これは「'　　'」のように，スペースの文字列を与えるのと同等です．

ドル ($) 編集は，改行コードを出力しない指定です．通常，print 文や write 文で出力すると，出力後に改行をするため，複数の print 文や write 文を使って 1 行に出力することはできません．これは，指定した出力文字に加えて最後に改行コードを出力しているためです．ドル ($) 編集を使えば改行コードを出力しないので，複数の print 文や write 文を使って 1 行に続けて文字を出力することができます．たとえば，do ループ中に入れた print 文で，計算結果を横に並べて出力することもできます．

最後のコロン (:) 編集はわかりにくいので，例を使って説明します．先ほど反復指定に大きめの数値を与えておくことができる例を示しましたが，このとき数字の前に文字列を付けようとすると問題が起こります．たとえば数値の前に $ 記号を付けたくて，

```
real a(4)
integer m
a(1) = 2.25
a(2) = 30.2
a(3) = 400.7
a(4) = 5000.6
print "(10('    $',f10.2))",(a(m),m=1,4)        ! 反復指定は 10
```

と書くと，出力結果は，

```
 $      2.25    $     30.20    $    400.70    $   5000.60    $
```

となります．すなわち，最後に余分な 5 個目の “$”が出力されるのです．これは，前述のように数値と編集記述子が対応しなくなった時点で出力が終了するのですが，5 個目の “$”の時点ではまだ終了するかどうか未定だからです．これを防ぐために使うのがコロン (:) 編集です．上記の format を修正して，

```
print "(10('    $',f10.2:))",(a(m),m=1,4)      ! f10.2 の後にコロンを挿入
```

のように f10.2 の後に “:”を入れておけば，対応しなくなった時点からさかのぼって，“:”の場所で出力が打ち切られます．これならば，

```
 $      2.25    $     30.20    $    400.70    $   5000.60
```

のように，“$”は 4 個しか出力されません．

5.4 データの入力方法

プログラムが実行しているとき，そのプログラム中の指定した変数に外部からデータを代入することを “入力する”といいます．計算条件を設定するための変数にデータを入力できるようにしておけば，プログラムの実行を開始してから条件を設定して，それに応じた計算をさせることができます．これにより，プログラムはその動作だけを利用する “ブラックボックス”になり，コンパイルしたプログラムを “アプリケーション”として他の人に提供することも可能です．

5.4.1 入力文の一般型

データ入力には read 文を用います．

```
read(nd,*) 変数 1，変数 2，  ...
```

nd は装置番号で，*nd* に適当な整数を与えると “fort.*nd*”という名のファイルから入力します[37]．装置番号に関する条件や注意事項は出力の場合と同じで，原則として $nd \geqq 10$ にして下さい．出力ファイルと違うのは fort.*nd* という名のファイルが存在していなければエラーになるので，あらかじめ用意しておかなければならないことです．なお，“*”の位置には，write 文のように format による書式指定ができますが，あまり使うことがないので説明は省略します．

たとえば，fort.30 という名前のファイルに，

```
   5.2    1.5    3
```

と書いて保存しておき，プログラム中に，

```
read(30,*) x,y,z
```

と書けば，この read 文実行後，x = 5.2，y = 1.5，z = 3.0 となって実行が継続します．入力ファイルに改行が入っていても，read 文の変数入力が完了するまで読み込みを続けるので，fort.30 の入力数値は次のように 3 行に分けて書くこともできます．

```
5.2
1.5
3
```

[37]出力ファイルと同様，“fort”の部分はコンパイラに依存します．ファイル名は 5.6 節で説明する open 文を使えば変更することができます．また，コンパイラによっては，open 文を使わなくても装置番号に対応する環境変数の指定で，任意の名前を持つファイルに変更することが可能です．

read 文の実行時に，ファイルが存在しなかったり，データが足らない場合には，エラーになってプログラムが強制終了します．これに対し，read 文が要求する数値よりもファイルに書かれている数値の方が多い場合は，read 文に記述されている全ての変数に数値が代入された時点で入力が終了します．この後，別の read 文で再び入力を実行すると，最後に読み込んだ行の次の行から入力を再開します．たとえば，fort.20 という名前のファイルに，

```
    5.2    1.5
    10     20
```

と書いて保存しておき，プログラム中に，

```
read(20,*) x,y
read(20,*) m,n
```

と書けば，この 2 回の read 文実行後，x = 5.2，y = 1.5，m = 10，n = 20 となります．read 文の処理は行単位で行われるので，最後に読み込んだ行に余分な数値が書かれている場合は無視されます．たとえば，上記の fort.20 を書き換えて，

```
    5.2    1.5    30    40
    10     20
```

のように 1 行目に数字を余分に書いても，2 回目の read 文の結果は変わりません．

read 文の変数の位置には，配列名や，do 型出力と同型の do 型入力を書くこともできます．これらは，出力と入力という方向が異なりますが，入力要素数やくり返しの意味は同じです．

ファイルではなく，キーボード (正確には標準入力) を使って入力したいときには，以下のように書きます[38]．

```
read *, 変数1, 変数2,  ...
```

この文を実行すると，プログラムの実行が一時停止し，キーボードからの数値入力を待つ状態になります．そこで，適切な数値をキーボードから入力すると，その数値を所定の変数に代入した後，実行が再開します．このため，read 文のタイミングを考慮して数値を入力しなければ，いつまでたっても停止したままです．そこで，read 文の前に入力を促すような文字列を出力する print 文を入れることをお勧めします．たとえば，

```
print *,'Input X and Y :'
read *,x,y
```

と書いておけば，入力のタイミングもわかるし，どの変数へ入力するための数値を要求されているかもわかります．

5.4.2 入力時のエラー処理

ファイルからデータを入力するとき，要求したファイルが存在しなかったり，書き込まれたデータ数より多くのデータを入力しようとすれば，実行時エラーになってプログラムは強制終了します．これを防ぐため，read 文中にエラー処理指定を入れることができます．

```
read(nd,*,err=num) 変数1, 変数2,  ...
```

ここで，*num* には文番号を与えます．この read 文を実行したとき，入力エラーが起こると *num* で指定した文番号の行へジャンプします．たとえば，

[38] 標準入力の装置番号は 5 です．よって，“read *,”と書く代わりに，“read(5,*)”と書くこともできます．また装置番号を “*”にして，“read(*,*)”と書くこともできます．

```
    do k = 1, 100
      read(10,*,err=999) x,y,z
       .......
    enddo
999  x = 100
```

と書けば，エラーが起こると文番号 999 の行にジャンプして，その行から実行を継続します．

もし“ファイルの終了”，すなわち，データを入力するときに，それ以上入っていなかった，という場合を検知するだけなら，err=*num* の代わりに end=*num* と書くこともできます．両方入れてもかまいません．入力データ数が不明のときには，err や end 指定を入れておき，データ終了時点で次の処理に進むようなプログラムにしておくと良いでしょう．

5.4.3 ネームリストを用いた入力

便利なデータ入力手段として，“ネームリスト”を用いる方法があります．たとえば，

```
read(10,*) x,y,n
```

という入力文では，入力ファイル fort.10 を，

```
  10.0   1.e10   100
```

のように作成しますが，作成するためには入力変数の対応を常に覚えておかなければなりません．必要なデータを全部書き込まなければならないし，順番を間違えることもできません．

これに対し，ネームリスト入力では，入力データを“**変数＝データ値**”という代入形で記述するので，どの変数に代入するかを入力ファイルの中で明示することができます．

ネームリスト入力を使うときは，まず入力する可能性のある変数や配列名を namelist 文で登録します．namelist 文は次のような形式です．

```
namelist /ネームリスト名/ 変数1，変数2，  ...
```

namelist 文は非実行文なので，全ての実行文より前に書かなければなりません．また，変数や配列名の登録だけなので，型宣言は別途必要です．たとえば，

```
real x,y,a(10)
integer n
namelist /option/ x,y,n,a
```

と書きます．ローカル変数だけではなく，use 文で指定されたモジュール中で宣言されているグローバル変数も登録可能です．

namelist 文を使ってネームリストに登録された変数に対し，入力文は，

```
read(nd, ネームリスト名)
```

だけです．これをネームリスト入力文といいます．ネームリスト入力文は変数を指定しません．必要ならば，

```
read(nd, ネームリスト名,err=num)
```

のように，文番号 *num* を使った err=*num* を追加してエラー発生時の処理をすることも可能です．たとえば，ネームリスト名が option ならば，

```
read(15,option,err=999)
```

などのように書きます．

ネームリスト入力文に対する入力ファイルは次の形式で用意します．変数の順番は namelist 文の登録順とは無関係なので，自由に並べることができます．

```
&ネームリスト名
  変数1 = データ1，変数2 = データ2，...
/
```

“&ネームリスト名”から“/”までがネームリスト入力文1回で入力されるデータです．たとえば上例のようにネームリスト名が option のときには，

```
&option
  x=10.0, y=1.e10, n=100
/
```

のようにファイルに書いておきます．入力を開始すると，ネームリスト入力終了の記号“/”を読み込むまで入力を続けるので，次のように1行ずつ書くこともできます．

```
&option
  x=10.0
  y=1.e10
  n=100
/
```

ネームリスト入力にはもう一つ利点があります．それは，必ずしも登録された変数全部を入力ファイルに記述する必要がないことです[39]．記述しなかった変数には，ネームリスト入力文の実行前までに代入されていた値がそのまま残ります．このため，あらかじめ全ての登録変数にデフォルト値を代入しておけば，変更したい変数だけ入力ファイルに記述することができます．たとえば，

```
real x,y,a(10)
integer n,i
namelist /option/ x,y,n,a
x = 100.0
y = 100.e10
n = 0
do i = 1, 10
  a(i) = i
enddo
read(10,option)
```

のようにプログラムを書いたとします．入力ファイルとして fort.10 という名のファイルに，

```
&option  x=10.0, a(3)=5.0 /
```

と書き込んでおけば，read 文実行後，x は変更されますが，y や n はそのままです．配列 a の場合には，代入された要素 a(3) のみが変更されます．

なお，ネームリストに登録された変数の内容は，次のネームリスト出力文で出力することもできます．

```
write(nd, ネームリスト名)
```

この場合，全登録変数が“**変数＝データ値**”という形で出力されます．もっとも，ネームリスト出力文を実行すると，登録された全変数のデータが標準形式で出力されるので，変数が多いと煩雑です．取りあえず値を確認したいとき以外は，あまり使わない方が良いと思います．

[39]逆に，入力ファイルに同じ変数を複数回記述することもできます．この場合は最後に代入した値が有効になります．

5.5 書式なし入出力文によるバイナリ形式の利用

これまで，print 文・write 文による出力や read 文による入力には，文字を使って表現した数値を用いていました．これは我々人間の都合によるものです．計算機内部の数値は 2 進数で表現されていて，“10”とか，“1.000e20”とかの文字ではありません．しかし，2 進数の並びでは我々が理解できないため，出力するときは“2 進数→ 10 進数文字表現”というデータ変換を行って画面に表示したりファイルに保存し，入力するときは，“10 進数文字表現→ 2 進数”というデータ変換を行って所定の変数に数値を代入するのです．この文字で表現した形式を“テキスト形式”といいます．

しかし，2 進→ 10 進変換には時間がかかる上に，文字に変換することでデータ量が増えます．これは，文字データが 1 文字あたり 1byte 必要だからです．たとえば，倍精度実数の 2 進数表現は 8byte ですが，実数を有効数字 15 桁で出力するには 15 文字 (15byte) 以上必要です．そこで，大量のデータを精度を落とさずに保存するときは，内部表現のままで保存するのが有効です．この内部表現を“バイナリ形式”といいます．Fortran でバイナリ形式の入出力を行うときは，write 文や read 文において *form* を省略した書式を用います．これを“書式なし入出力文”といいます．これに対し，これまで説明してきた *form* を指定する write 文や read 文は“書式付き入出力文”です．

書式なし write 文は，次のように装置番号だけをかっこ内に指定した write 文です．

```
write(nd) データ 1，データ 2，  ...
```

また，書式なし read 文は，装置番号だけをかっこ内に指定した read 文です．

```
read(nd) 変数 1，変数 2，  ...
```

書式なし read 文は，書式付き read 文と同様に，文番号 *num* を使って，

```
read(nd,err=num) 変数 1，変数 2，  ...
```

のように，err=*num* や end=*num* を追加したエラー発生や終了時の処理をすることもできます．

書式なし write 文で作成したバイナリ形式ファイルから，書式なし read 文を使ってデータを入力するときは，保存した形式に合わせて読み込まなければならないので，いくつか注意が必要です．

まず，write 文で出力したときのデータと同じ数値型の変数を同じ順番で read 文に並べる必要があります．たとえば，

```
real x,y
integer n
x = 10.0
y = 100.0
n = 10
write(20) x,n,y
```

というプログラムで作成された fort.20 というファイルから数値を入力するには，

```
real x,y
integer n
read(20) x,n,y
```

のように書かなければなりません．この場合，x も n も同じ 10 だからと思って，

```
read(20) n,x,y
```

と入力すると，出力と数値型の異なる n と x には，正しい値が代入されません．

また，テキスト出力のような“改行”はありませんが，write 文 1 回ごとに印 (ヘッダ) が付くので，

write 文 1 回の出力に対し，read 文 1 回で入力しなければなりません[40]．ただし，write 文 1 回で書き込まれたデータ数よりも read 文 1 回で入力するデータが少ないのは問題ありません．このとき入力しなかったデータは読み飛ばしたことに相当します．たとえば，上記の fort.20 のデータを，

```
read(20) x,n
read(20) y
```

のように 2 行の read 文に分けて入力しようとすると，1 行目の read 文実行時における x と n には正常な値が代入されますが，2 行目の read 文実行時に「入力データがない」というエラーで強制終了します．逆に，上記の write 文を修正して，

```
write(20) x,n
write(20) y
```

という 2 行で出力したファイルから，

```
read(20) x,n,y
```

のように，read 文 1 回で読み込むこともできません．

書式なし write 文は，1 回で出力できるデータ数に特に制限はありません．このため，大きな配列を 1 回で出力することも可能です．たとえば，

```
real p(10000),q(10000),r(10000)
write(30) p,q,r
```

のように，1 回の write 文で複数の配列の全要素を出力することができます．

もっとも，出力用のプログラムと入力用のプログラムのメンテナンスを考えれば，出力する配列要素数を次のようにあらかじめ出力しておいた方が良いと思います．

```
real p(10000),q(10000),r(10000)
write(30) 10000
write(30) p,q,r
```

これを入力するときは，出力数を考慮して

```
real p(10000),q(10000),r(10000)
integer i,imax
read(30) imax
read(30) (p(i),i=1,imax),(q(i),i=1,imax),(r(i),i=1,imax)
```

のように書きます．こうしておけば，出力用のプログラムを修正して要素数を減らしても，入力用のプログラムの変更は不要です．

5.6 ファイルのオープンとクローズ

write 文や read 文を使ってファイルから入出力をする場合，何も指定がなければ，装置番号 *nd* を付加した “fort.*nd*” という名のファイルを使用します．これに対し，任意の名前を持つファイルを使いたいときには，open 文を使って入出力文の実行前にファイル名を指定しておきます．これを “ファイルをオープンする ” といいます．open 文は以下の形式です．

[40] このように，書式なし write 文を使って保存したバイナリ形式ファイルには，データだけでなく Fortran 独自のヘッダが付加されています．このため，C 言語など，他のプログラミング言語で作成したプログラムから読み込む場合や，データ解析ソフトを使って解析するときにはうまく入力できない場合があります．そのような利用が目的でデータを保存するときは，テキスト形式で保存した方が良いでしょう．

```
open(nd,file=name [, form=format] [, status=stat] [, err=num] )
```

[ ] は，その内容が省略可能という意味です．それぞれの記述 (制御指定子) の意味を表 5.3 に示します．

表 5.3　open 文の制御指定子の意味

指定子	指定情報	指定子の意味と注意
nd	装置番号	整数を与える (整数型変数や整数式を与えることも可能) オープンした後，read 文や write 文の装置番号として使う
name	ファイル名	文字列で指定する (文字変数も可能) ファイル名は大文字・小文字を正しく指定する必要がある
format	ファイル形式	文字列で指定する 省略するとテキスト形式の入出力が仮定される バイナリ形式の入出力を使うときは「'unformatted'」を指定する
stat	ファイル情報	文字列で指定する 既存のファイルを使うときは「'old'」を，存在しないファイルを使うときは「'new'」を指定する 条件に合わないとエラーが発生する
num	文番号	エラーのときにジャンプする行の文番号を指定する 省略すると，エラーが起きたときにはプログラムが強制終了する

たとえば，装置番号 10 のテキスト形式ファイルを “text.out” という名にするときは，

```
open(10,file='text.out')
```

と書きます．また，装置番号 30 のバイナリ形式ファイルを “binary.dat” という名にするときは，

```
open(30,file='binary.dat',form='unformatted')
```

と書きます．ただし，既存のファイルを指定して write 文で書き込むと，それまで書き込まれていたデータが上書きされて消えるので注意が必要です．上書きを防ぎたいときには，

```
open(30,file='binary.dat',form='unformatted',status='new',err=999)
```

のように，status='new' と err を指定します．status に'new' を指定すると，ファイルが存在しなければ新しく作成し，存在すればエラーになります．そこで，エラーの処理を err で指定した文番号 999 の行に用意しておけば上書きを防ぐことができます．

status の指定を省略して open 文を実行したとき，指定したファイルが存在しなければ，その名前のファイルが新たに作成されます．このとき，そのファイルが read 文の入力用であれば，データが入っていないので read 文実行時にエラーになりますが，プログラム終了後も作成された空のファイルが残ってしまいます．そこで，ファイルが入力用のときは，status='old' を指定して，その存在をチェックした方が良いでしょう．たとえば，バイナリ形式ファイル “binary.inp” を入力ファイルとして装置番号 20 に指定するには，

```
open(20,file='binary.inp',form='unformatted',status='old',err=999)
```

のように書きます．この open 文では，オープンした時点でファイルが存在しなければ，err=999 で指定した文番号 999 の行へジャンプします．

ファイルへ出力する場合，write 文実行時の出力命令のタイミングと実際にディスクに書き込まれるタイミングは必ずしも一致していません．これは入出力ハードウェアを効率よく運用するために，一時記憶領域への読み書きが介在するためです．このため，確実に書き込みを完了させたいときにはファイルをクローズします．クローズは，次の close 文で行います．

```
close(nd)
```

nd はクローズするファイルの装置番号です．たとえば，装置番号 30 のファイルをクローズするときは，

```
close(30)
```

のように書きます．close 文を実行すると，その時点までに装置番号 nd に出力した全てのデータがディスクに書き込まれます．もっとも，プログラムが正常に終了すれば，全てのファイルが自動的にクローズされるので，通常は close 文を書かなくても問題ありません．

ファイルをクローズすると，装置番号 nd と open 文で指定したファイル名の関係は切れます．このため，同じ装置番号に新たに別のファイルを指定してオープンすることも可能です．また，同じ名前のファイルを再度オープンすることもできます．ただし，再オープンしてもそれ以前の読み書きの継続にはならず，ファイルの先頭に戻って読み書きをします．すなわち，“巻き戻し”をすることになります[41]．ファイルを巻き戻すと，read 文による入力は一からやり直すことになり，write 文による出力はファイルの先頭から書き直します．このため，巻き戻してから write 文で出力すると，それ以前に書き込んだデータは消去されます．

なお，巻き戻すのが目的であれば，rewind 文を使うことで，クローズせずともファイルを巻き戻すことができます．rewind 文とは，次のような文です．

```
rewind(nd)
```

nd は再度最初から読み書きするファイルの装置番号です．巻き戻しを利用すれば，まず write 文を使ってファイルにデータを出力しておき，次にそれを巻き戻して，read 文を使ってそのデータを入力して使うことも可能です．

[41]「巻き戻す」という言葉を使うのは，磁気テープが昔の計算機における大容量記録媒体の一つだったからです．しかし，最近は音楽用カセットテープでさえ使わなくなったのですから，あまり適切な言葉とは言えないかもしれませんね．

演 習 問 題 5

(5–1)　連立常微分方程式の解法：2 次の Runge-Kutta 法

$N+1$ 個の質量 m のおもりが一直線に並んでいて，隣り合ったおもりはバネ定数 k のバネでつながれているとする．このおもりの位置を左側から，z_0，z_1，$\cdots$，z_N とすると，i 番目のおもりの運動方程式は，その速度を v_i として，

$$\frac{dz_i}{dt} = v_i$$
$$m\frac{dv_i}{dt} = -k(z_i - z_{i-1}) - k(z_i - z_{i+1})$$

である．このおもりの運動を 2 次の Runge-Kutta 法で解析せよ．

ここで，2 次の Runge-Kutta 法とは変数ベクトル $\boldsymbol{x}$ に対する連立常微分方程式 $\dfrac{d\boldsymbol{x}}{dt} = \boldsymbol{f}(t, \boldsymbol{x})$ に対し，時刻 t_n の値 $\boldsymbol{x}_n$ から出発して時刻 $t_{n+1} = t_n + \Delta t$ の値 $\boldsymbol{x}_{n+1}$ を

$$\boldsymbol{k}_1 = \boldsymbol{f}(t_n, \boldsymbol{x}_n)$$
$$\boldsymbol{x}'_n = \boldsymbol{x}_n + \Delta t\, \boldsymbol{k}_1$$
$$\boldsymbol{k}_2 = \boldsymbol{f}(t_n + \Delta t, \boldsymbol{x}'_n)$$
$$\boldsymbol{x}_{n+1} = \boldsymbol{x}_n + \frac{\Delta t}{2}(\boldsymbol{k}_1 + \boldsymbol{k}_2)$$

という手順で計算する方法である．Δt は適当に決める．プログラムは

(1) 時間を進めるメインプログラム
(2) Runge-Kutta 法の 1 ステップを計算するサブルーチン
(3) 運動方程式の右辺を計算するサブルーチン

の 3 個から構成するものとする．さらに，z_i や v_i は引数で与え，m や k はグローバル変数としてモジュールと `use` 文を使って共有するようにせよ．なお，初期条件は以下の通りとする．

$$z_i = hi + g\sin(k_p i) \qquad (i = 0 \cdots N)$$
$$v_i = 0$$

ここで，k_p は適当な整数 p を使って，$k_p = \dfrac{2\pi p}{N}$ で与える．また，端にあるおもり（z_0 と z_N）は動かないとする．（つまり，計算しないで初期条件のままとする）

(5–2)　連立常微分方程式の解法：4 次の Runge-Kutta 法

問題 (5–1) を修正して 4 次の Runge-Kutta 法で解析せよ．ここで，4 次の Runge-Kutta 法とは微分方程式 $\dfrac{d\boldsymbol{x}}{dt} = \boldsymbol{f}(t, \boldsymbol{x})$ に対し，時刻 t_n の値 $\boldsymbol{x}_n$ から出発して時刻 $t_{n+1} = t_n + \Delta t$ の値 $\boldsymbol{x}_{n+1}$ を

$$
\begin{aligned}
\boldsymbol{k}_1 &= \boldsymbol{f}(t_n, \boldsymbol{x}_n) \\
\boldsymbol{x}'_n &= \boldsymbol{x}_n + \frac{\Delta t}{2}\boldsymbol{k}_1 \\
\boldsymbol{k}_2 &= \boldsymbol{f}(t_n + \frac{\Delta t}{2}, \boldsymbol{x}'_n) \\
\boldsymbol{x}'_n &= \boldsymbol{x}_n + \frac{\Delta t}{2}\boldsymbol{k}_2 \\
\boldsymbol{k}_3 &= \boldsymbol{f}(t_n + \frac{\Delta t}{2}, \boldsymbol{x}'_n) \\
\boldsymbol{x}'_n &= \boldsymbol{x}_n + \Delta t\, \boldsymbol{k}_3 \\
\boldsymbol{k}_4 &= \boldsymbol{f}(t_n + \Delta t, \boldsymbol{x}'_n) \\
\boldsymbol{x}_{n+1} &= \boldsymbol{x}_n + \frac{\Delta t}{6}(\boldsymbol{k}_1 + 2(\boldsymbol{k}_2 + \boldsymbol{k}_3) + \boldsymbol{k}_4)
\end{aligned}
$$

という手順で計算する方法である．初期条件や，境界条件は問題 (5–1) と同じにし，問題 (5–1) の結果と比較せよ．

(5–3)　放物型偏微分方程式の解法：陽解差分法

次の温度 $T(x,t)$ に関する熱伝導方程式を，陽解差分法 (Explicit 法) を用いて解析せよ．

$$\frac{\partial T}{\partial t} = \kappa \frac{\partial^2 T}{\partial x^2}$$

このような時間発展の偏微分方程式を解くには，まず等間隔 h ごとに並んだ x 座標点 $x_i = hi, (i = 0 \sim L)$ に対して初期の温度分布を与え，時間間隔 Δt ごとに時間を進めた温度分布を計算する．陽解差分法とは，n ステップ目の温度分布を $T_i^n = T(x_i, \Delta t\, n)$ と表したときに，熱伝導方程式を，

$$\frac{T_i^{n+1} - T_i^n}{\Delta t} = \kappa \frac{T_{i+1}^n - 2T_i^n + T_{i-1}^n}{h^2}$$

と近似する方法である．この方程式を T_i^{n+1} に関して解けば，

$$T_i^{n+1} = T_i^n + \frac{\kappa \Delta t}{h^2}\left(T_{i+1}^n - 2T_i^n + T_{i-1}^n\right) \qquad (i = 1, 2, \ldots, L-1)$$

となるので，これを使って T_i^n から T_i^{n+1} をくり返し計算していけばよい．

ただし，境界条件として，$T_0^n = 0$，$T_L^n = 0$ とする．また，初期条件は，

$$
\begin{cases}
T_i^0 = i & (i < L/2) \\
T_i^0 = L - i & (i > L/2)
\end{cases}
$$

とする．プログラムは

(1) 時間を進めるメインプログラム
(2) 初期条件を計算するサブルーチン
(3) 陽解法で 1 ステップ進めるサブルーチン

の 3 個から構成するものとする．温度 T_i^n と，h，Δt，κ などのパラメータはグローバル変数にしてモジュールと use 文で共有する．また，メモリの節約のため，T_i^n を配列 T1(i) で表し，T_i^{n+1} を配列 T2(i) で表して，T1 と T2 のデータを適宜交換しながら計算を進めるようにせよ．

完成したプログラムを使って，$\dfrac{\kappa\Delta t}{h^2} = 0.2$ にすると安定に計算ができるのに，$\dfrac{\kappa\Delta t}{h^2} > 0.5$ にすると，不安定になって解が発散することも確かめよ．

(5–4)　放物型偏微分方程式の解法：陰解差分法

温度 $T(x,t)$ に関する熱伝導方程式を，陰解差分法 (Implicit 法) を用いて解析せよ．

ただし，陰解差分法とは，問題 (5–3) の偏微分方程式を

$$\frac{T_i^{n+1} - T_i^n}{\Delta t} = \frac{1}{2}\left[\kappa\frac{T_{i+1}^{n+1} - 2T_i^{n+1} + T_{i-1}^{n+1}}{h^2} + \kappa\frac{T_{i+1}^n - 2T_i^n + T_{i-1}^n}{h^2}\right]$$

と近似する方法である．この方程式を変形すれば，

$$a_i T_{i-1}^{n+1} + b_i T_i^{n+1} + c_i T_{i+1}^{n+1} = d_i \qquad (i = 1, 2, \ldots, L-1)$$

の形になるので，問題 (2–7) で示した連立 1 次方程式の解法を使って T_i^{n+1} が計算できる．

ここで，境界条件や初期条件は問題 (5–3) と同じでよい．プログラムは

(1) 時間を進めるメインプログラム
(2) 初期条件を計算するサブルーチン
(3) 陰解法で 1 ステップ進めるサブルーチン

の 3 個から構成するものとする．温度 T_i^n と，h，Δt，κ などのパラメータはグローバル変数にしてモジュールと use 文で共有する．また，メモリの節約のため，T_i^n を配列 T1(i) で表し，T_i^{n+1} を配列 T2(i) で表して，T1 と T2 のデータを適宜交換しながら計算を進めるようにせよ．

完成したプログラムを用いて計算すると，$\dfrac{\kappa\Delta t}{h^2} > 0.5$ にしても不安定にならないことも確かめよ．

(5–5)　楕円型偏微分方程式の解法

電荷密度 $\rho(x,y)$ が与えられているときに電位の満足する 2 次元ポワソン方程式

$$\frac{\partial^2 V(x,y)}{\partial x^2} + \frac{\partial^2 V(x,y)}{\partial y^2} = -\rho(x,y)$$

を解くには次のようにする．

等間隔 h ごとに並んだ x 座標点 $x_i = hi, (i = 0\sim M)$ と y 座標点 $y_j = hj, (j = 0\sim N)$ に対して，$V_{i,j} = V(x_i, y_j)$，$\rho_{i,j} = \rho(x_i, y_j)$ として，上の偏微分方程式を差分化すれば次式になる．

$$\frac{V_{i-1,j} - 2V_{i,j} + V_{i+1,j}}{h^2} + \frac{V_{i,j-1} - 2V_{i,j} + V_{i,j+1}}{h^2} = -\rho_{i,j}$$

これは $V_{i,j}$ に関する連立 1 次方程式である．これを解くには，以下のような漸化式に変形して反復法を用いる．

$$V_{i,j}^{(k+1)'} = \frac{1}{4}\left(V_{i-1,j}^{(k+1)} + V_{i,j-1}^{(k+1)} + V_{i+1,j}^{(k)} + V_{i,j+1}^{(k)} + \rho_{i,j}h^2\right)$$
$$V_{i,j}^{(k+1)} = V_{i,j}^{(k)} + \omega(V_{i,j}^{(k+1)'} - V_{i,j}^{(k)})$$

これを適当な初期条件 $V_{i,j}^{(0)}$ から出発して，k 回目の反復値 $V_{i,j}^{(k)}$ と $k+1$ 回目の反復値 $V_{i,j}^{(k+1)}$ の差が十分小さくなれば，元の連立方程式を近似的に満足する解になると考えられる．これが反復法である．

なお，第 1 式の $V_{i,j}^{(k+1)'}$ をそのまま次の反復値 $V_{i,j}^{(k+1)}$ として用いる手法を，ガウス・ザイデル法という．しかし，ガウス・ザイデル法は収束が遅いので，第 2 式を使って解を改良する．これを，SOR(Successive Over Relaxation，逐次過緩和) という．ω は加速係数であり，$\omega = 1$ のときがガウス・ザイデル法，$\omega > 1$ が SOR である．ただし，$\omega \geqq 2$ では発散するので，$1 < \omega < 2$ の範囲で最も収束が速くなる数値を選ぶ．

この反復法を使って，次の電荷密度における電位を計算するプログラムを作成せよ．

$$\rho(x,y) = \sin\left(\frac{2\pi x}{Mh}\right)\sin\left(\frac{4\pi y}{Nh}\right)$$

ここで，初期条件は $V_{i,j}^{(0)} = 0$，境界条件は $V_{0,j} = V_{i,0} = V_{M,j} = V_{i,N} = 0 \quad (i = 0\sim M,\ j = 0\sim N)$ でよい．また，収束の判定は

$$\max_{i,j} |V_{i,j}^{(k+1)} - V_{i,j}^{(k)}| < \varepsilon$$

で行うこと．ただし，ε は適当な小さい値とする (たとえば，10^{-12})．

プログラムは，

(1) 反復を行い，収束条件を確認するメインプログラム
(2) 初期値を代入するサブルーチン
(3) ガウス・ザイデル法 +SOR を 1 ステップ行うサブルーチン

の 3 個から構成するものとする．

(5–6)　分子動力学を用いた分子運動の解析

距離 r 離れた 2 個の分子間の位置エネルギーを表す近似式の一つに次式のようなレナード・ジョーンズポテンシャルがある [3].

$$U(r) = 4\varepsilon\left\{\left(\frac{\sigma}{r}\right)^{12} - \left(\frac{\sigma}{r}\right)^{6}\right\}$$

この式は，分子が近づくと r^{-12} の項が優勢になって斥力（反発力）が働き，遠ざかると r^{-6} の項が優勢になって引力が働くことを示している．レナード・ジョーンズポテンシャルの勾配を計算することで分子間に働く力が計算できる．すなわち，2 個の分子 i と j の位置ベクトルを $\boldsymbol{r}_i$，$\boldsymbol{r}_j$ とすると，分子 j が分子 i に及ぼす力は，

$$\boldsymbol{F}_{ij} = -\nabla_i U(|\boldsymbol{r}_i - \boldsymbol{r}_j|) = \frac{4\varepsilon}{\sigma}\left\{12\left(\frac{\sigma}{r_{ij}}\right)^{13} - 6\left(\frac{\sigma}{r_{ij}}\right)^{7}\right\}\frac{\boldsymbol{r}_i - \boldsymbol{r}_j}{r_{ij}}$$

となる．ここで，$r_{ij} = |\boldsymbol{r}_i - \boldsymbol{r}_j|$ である．

今，分子が N 個あるとすると，分子 i に加わる力は i 以外の分子から受ける力の合力なので，

$$\boldsymbol{F}_i = \sum_{\substack{j=1\\ i\neq j}}^{N}\boldsymbol{F}_{ij} = \sum_{\substack{j=1\\ i\neq j}}^{N}\frac{4\varepsilon}{\sigma}\left\{12\left(\frac{\sigma}{r_{ij}}\right)^{13} - 6\left(\frac{\sigma}{r_{ij}}\right)^{7}\right\}\frac{\boldsymbol{r}_i - \boldsymbol{r}_j}{r_{ij}}$$

となる．これを用いて N 個の分子の運動方程式

$$m_i\frac{d^2\boldsymbol{r}_i}{dt^2} = \boldsymbol{F}_i = \sum_{\substack{j=1\\ i\neq j}}^{N}\frac{4\varepsilon}{\sigma}\left\{12\left(\frac{\sigma}{r_{ij}}\right)^{13} - 6\left(\frac{\sigma}{r_{ij}}\right)^{7}\right\}\frac{\boldsymbol{r}_i - \boldsymbol{r}_j}{r_{ij}} \qquad (i = 1...N)$$

を解くプログラムを作成せよ．ここで，分子数が増えると時間がかかるので，効率よく時間積分をするためにベルレ法を用いる [3]．ベルレ法とは次のような 2 段階の計算で位置ベクトルと速度ベクトルを更新する方法である．

まず，現時刻 (n ステップ後) の位置ベクトル $\boldsymbol{r}_i^n$ を用いて計算した各分子に加わる力 $\boldsymbol{F}_i^n$ と現時刻の速度ベクトル $\boldsymbol{v}_i^n$ を使って次式のように位置ベクトルを更新する．

$$\boldsymbol{r}_i^{n+1} = \boldsymbol{r}_i^n + \Delta t \boldsymbol{v}_i^n + \frac{\Delta t^2}{2m_i}\boldsymbol{F}_i^n$$

ここで，Δt は 1 ステップの時間間隔である．

次に，この更新された位置ベクトル $\boldsymbol{r}_i^{n+1}$ を用いて新しい力 $\boldsymbol{F}_i^{n+1}$ を計算し，次式のように速度ベクトルを更新する．

$$\boldsymbol{v}_i^{n+1} = \boldsymbol{v}_i^n + \frac{\Delta t}{m_i}\frac{\boldsymbol{F}_i^{n+1} + \boldsymbol{F}_i^n}{2}$$

このとき，新しい力 $\boldsymbol{F}_i^{n+1}$ は次ステップの位置ベクトルや速度ベクトルの更新に利用できることに注意すること．

なお，力の計算に r_{ij}^{-2k} が入っているため，2 分子の距離がかなり接近すると，力が大きくなりすぎて，安定に計算することができなくなる．そこで，r_{ij}^{-2} を次式のように修正するとよい．

$$\frac{1}{r_{ij}^2} \to \frac{1}{(x_i - x_j)^2 + (y_i - y_j)^2 + (z_i - z_j)^2 + \varepsilon^2}$$

ここで，ε は最も接近する可能性のある距離で，σ に比べて十分小さい数にする．

プログラムは，

(1) 時間を進めるメインプログラム
(2) 初期条件を計算するサブルーチン
(3) ベルレ法第 1 段階で位置ベクトルを更新するサブルーチン
(4) ベルレ法第 2 段階で速度ベクトルを更新するサブルーチン

の 4 個から構成するものとする．

初期条件としては，まず適当な間隔離れた 2 個の分子でテストする．これにより正常に動作することが確認されたら，正三角形の頂点に配置した 3 個の分子や正四面体の頂点に配置した 4 個の分子などの動きを計算せよ．

(5–7)　1 次元移流方程式の解法

次の関数 $f(x,t)$ に関する 1 次元移流方程式を，CIP 法を用いて解析せよ [4]．

$$\frac{\partial f}{\partial t} + u\frac{\partial f}{\partial x} = 0$$

ここで，CIP 法とは，間隔 h の x 座標点 $x_i = hi, (i = 0 \sim L)$ で定義された，n ステップ目の関数値 $f_i^n = f(x_i, n\Delta t)$ と，その x 方向微分 $g(x,t) = \dfrac{\partial f}{\partial x}$ の値 $g_i^n = g(x_i, n\Delta t)$ が与えられているとき，時間間隔 Δt 後の関数値と微分値を，

$$f_i^{n+1} = a_i X^3 + b_i X^2 + g_i^n X + f_i^n$$
$$g_i^{n+1} = 3a_i X^2 + 2b_i X + g_i^n$$

で近似する方法である．ここで，$X=-u\Delta t$ である．また，a_i，b_i は次式で与えられる．

$$a_i=\frac{g_i^n+g_{i+1}^n}{h^2}+\frac{2(f_i^n-f_{i+1}^n)}{h^3}$$
$$b_i=\frac{3(f_{i+1}^n-f_i^n)}{h^2}-\frac{2g_i^n+g_{i+1}^n}{h}$$

初期条件は，グリッド数 L と，その中間の値，i_1，i_2 を適当に決めて，

$$\begin{cases} f_i^0=1 & (i_1<i<i_2) \\ f_i^0=0 & (\text{それ以外}) \end{cases}$$

とせよ．なお，$h=1$，$u=-1$ とする．

まず，n ステップ目の座標点データ f_i^n と g_i^n を 1 次元配列 `f1(i)` と `g1(i)` で表し，これらと適当な値の u，h，Δt を用いて，$n+1$ ステップ目の座標点データ f_i^{n+1} と g_i^{n+1} を表す 1 次元配列 `f2(i)` と `g2(i)` を計算するサブルーチンを作成する．そして，これらを適宜交換しながらくり返すことで，f_i^n や g_i^n が時間発展するプログラムを作成すればよい．なお，i の最大値 L では a_i，b_i を計算することができないので，常に $f_L^n=0$，$g_L^n=0$ で良い．

プログラムは，

(1) 時間を進めるメインプログラム
(2) 初期条件を計算するサブルーチン
(3) CIP 法で関数値と微分値を更新するサブルーチン

の 3 個から構成するものとする．

(5–8)　粒子の密度分布

2 次元領域，$0\leqq x\leqq X_m$，$0\leqq y\leqq Y_m$，の内部に N 個の粒子があるとき，粒子の密度分布は以下のような手順で計算することができる．

まず，適当な整数 L と M を使って，2 次元領域を $h_x=X_m/L$ ごとに並んだ x 座標点 x_0，x_1，$\cdots$，$x_L(=X_m)$ と，$h_y=Y_m/M$ ごとに並んだ y 座標点 y_0，y_1，$\cdots$，$y_M(=Y_m)$ で指定したグリッドを導入する．次に 2 次元の密度分布配列 `Rho(0:L,0:M)` を用意して，全ての配列要素に初期値 0 を代入しておいてから，以下の `do` ブロックを用いて各グリッドに所属する粒子数をカウントする．

```
do k = 1, n                  ! n  = N
   i = x(k)/hx               ! hx = hx
   j = y(k)/hy               ! hy = hy
   Rho(i,j) = Rho(i,j) + 1
enddo
```

ここで，`(x(k),y(k))` は `k` 番目の粒子の座標である．また，`i` と `j` は整数型変数であり，`x(k)/hx` と `y(k)/hy` の小数点以下を切り捨てて代入することで粒子が所属するグリッド番号を計算するようになっている．この粒子数のカウント後，全ての配列要素を 1 グリッドあたりの平均粒子数 N/LM で割れば空間的な密度分布を得ることができる[42]．

4.8.6 節で紹介した乱数発生用のサブルーチン `random_number` を使って，一様乱数を 2 個ずつ発生させ，これらに X_m と Y_m を掛けることで 2 次元領域中にランダムで一様に分布した粒子の座標点を N 個作成し，それらを用いて粒子の密度分布を計算せよ．また，その密度分布のデータを用いて，全グリッド点における密度の平均と標準偏差を計算せよ．そして，N が大きくなると，標準偏差が小さくなることを確かめよ．

[42]ただし，プログラム中で整数 N を整数 LM で割ると切り捨てが起こる可能性があるので，この平均粒子数の計算は N を実数化して行ってください．

(5–9)　ブラウン運動の解析

ある点から粒子を速度 v で出発させたところ，時刻 Δt ごとに周りの粒子に衝突して方向を変えるとする．このときの粒子の進む様子を計算してみよう．

問題 (5–8) と同じ 2 次元領域で，粒子の初期位置は，全て $(\frac{X_m}{2}, \frac{Y_m}{2})$，すなわち，2 次元領域の中央とする．計算は

(1) 1 ステップごとに乱数 R を発生させ，$\theta = 2\pi R$ で，方向角 θ を計算する
(2) 1 ステップ後の位置を $(x_{s+1}, y_{s+1}) = (x_s + v\Delta t \cos\theta, y_s + v\Delta t \sin\theta)$ で計算する
(3) 粒子が $0 \leqq x \leqq X_m$，$0 \leqq y \leqq Y_m$ の領域から外に出たら計算をストップする

のように進めればよい．

この過程によって N 個の粒子の N_T ステップ後の位置を計算した後に，問題 (5–8) の手法を使ってその密度分布を計算せよ．また，ステップ N_T を増やすと，密度分布がどの様に変化するかを調べよ．

プログラムは，

(1) 時間を進めるメインプログラム
(2) 初期条件を計算するサブルーチン
(3) 粒子の座標を更新するサブルーチン
(4) 粒子の位置から密度分布を計算するサブルーチン

の 4 個から構成するものとする．項目 (4) の密度分布を計算するサブルーチンは，問題 (5–8) で作成したプログラムを利用して作成すればよい．

第 6 章　文字列の活用

これまで print 文や format 文に記述して，数値の意味などを表示するのに使ってきた文字列ですが，この他にも様々な用途があります．また，固定した文字の並びだけではなく，文字列変数を使って条件に応じてその内容を変更することもできるし，文字列と文字列を連結したり，文字列の一部を取り出したりするなどの “文字列演算”もできます．本章では文字列をより積極的に活用する手法を紹介します．

6.1　文字列定数と文字列変数

Fortran における文字列とは 2 個の「’」または「"」で囲んだ文字の並びのことです．たとえば，

```
'abc'  'Taguchi␣T.'  "(123.5678+X^2)"  "漢字も書けます"
```

等が文字列です．文字列にはスペースや記号を入れることもできます[43]．「’」と「"」は同等なので，どちらを使うかは自由です．たとえば，通常は「’」で囲み，文字列の中に「’」を使いたいときは，「"Teacher’s"」のように全体を「"」で囲む，というように決めておけば良いでしょう[44]．

コンピュータの “文字”は文字コードという整数値と対応していて，文字列は文字コードを表す整数の配列で実現されています．このため，整数定数の「1」と文字列の「’1’」は全く異なるものです．文字コードは半角英数字なら 1 文字あたり 1byte の整数，全角の漢字やひらがなら 1 文字あたり 2byte の整数です．半角英数字のコードは，現在ほとんど全てのコンピュータで ASCII コードが使われています[45]．このため，半角英数字だけでプログラムを書けば移植性の問題はありません．しかし，全角文字は OS や利用環境によって文字コードが違う可能性があるので，プログラム中に日本語を記述すると，別のコンピュータに移したときに文字化けすることがあります[46]．Fortran において日本語が使えるのはコメント文とテキスト表示くらいなので，それほど重要ではありません．以下，文字列中の “文字”は 1byte の ASCII コードであると仮定して説明をします．

「’」または「"」で囲んだ文字列は，“文字列定数”ですが，文字列を代入して保存する “文字列変数”を作ることもできます．文字列変数を作るには character 宣言を使います．character 宣言は，以下のような書式です．

```
character 文字列変数 1*文字数 1, 文字列変数 2*文字数 2, ...
```

文字数とは，文字列変数に代入可能な最大の文字数です．指定できる最小の文字数は 1 です．また，拡張宣言文 (4.8.1 節) にして属性などを付加することも可能です．たとえば，

```
character c1*10,c2*20,cc*1
character chr(20)*30,chs(100,200)*50
```

と宣言すると，文字列変数 c1 には 10 個まで，c2 には 20 個までの文字を代入することができます．これに対し cc には 1 文字しか入りません[47]．また chr や chs のように文字列の配列を宣言することも可能です．chr は 30 文字まで入る 1 次元配列で，chs は 50 文字まで入る 2 次元配列です．

同じ文字数の文字列変数を複数個宣言するときは，以下の 2 種類の書式で宣言することもできます．

---

[43] 本章では半角スペース記号を明記するときに “␣”を使います．

[44] ちなみに，「’」で囲んだ文字列の中に「’」を入れたいときは，「’Teacher’’s’」のように「’」を 2 個続けて記述します．「"」の場合も同様です．

[45] ASCII コード表は付録 E にあります．

[46] パソコンで使われているのは Shift JIS コードが多く，Linux などの UNIX 系 OS で使われているのは EUC コードが多いようです．最近は，コード体系をより統一化した Unicode を使う OS が増えてきました．

[47] 文法的には “*文字数”を省略して変数名だけで宣言すると，文字数が 1 の文字列になります．しかし，プログラムがわかりにくくなるので，常に “*1”を付加して文字数 1 を陽に指定した方が良いでしょう．

```
character(文字数)  文字列変数1, 文字列変数2, ...
character(len=文字数)  文字列変数1, 文字列変数2, ...
```

ここで，“文字列変数”には，変数名か，配列名とその要素数を記述します．たとえば，文字数 10 の文字列変数や文字列配列を宣言するときは，

```
character(10) cs1,ds1,cas1(100)
character(len=10) cs2,ds2,cas2(100)
```

などと書きます．2 行目の書式は，入力に手間がかかりますが「文字数」という意味を明示しているところが利点です．文字数は，次のように parameter 変数 (4.8.2 節) で指定することもできます．

```
integer, parameter :: kp = 10
character(kp) ch2
character(len=kp) ch3
```

文字列変数に文字列を代入するには，数値の代入と同じで，イコール「=」を使います．たとえば，上記の 10 文字の文字列変数 c1 に文字列定数を代入するには，

```
c1 = 'abc'
```

のように書きます．このとき，代入される文字 'abc' は 3 文字なので，10 文字の変数 c1 の先頭から順に 1 文字ずつ代入され，残りの領域は半角スペースが代入されます．スペースも文字ですから，c1 の文字数は，代入した文字列の文字数にかかわらず，10 のままです．図に描けば，以下のようなイメージです．

c1	a	b	c	␣	␣	␣	␣	␣	␣	␣

逆に，代入する文字列の文字数の方が多い場合には，その文字列の先頭から代入できる最大文字数までが代入され，残りは切り捨てられます．拡張宣言文を使えば，次のようにあらかじめ文字列定数を代入した文字列変数を用意することも可能です．

```
character :: chr*10='abcde'
```

この場合，文字列変数の文字数が 10 文字なのに代入しているのは 5 文字ですから，後の 5 文字はスペースが代入されています．

6.2 部分文字列と文字列演算

文字列変数に代入された文字列は部分的に取り出すことができます．これを“部分文字列”といいます．部分文字列は，文字列変数の先頭から数えて何番目から何番目という範囲を「:」を使って指定します．たとえば，c1 という文字列変数の n1 番目から n2 番目の文字を取り出すには，

```
c1(n1:n2)
```

と指定します．この文字列は，n2-n1+1 文字の文字列として扱われます．たとえば，c1='abcdefg' ならば，c1(3:5) は，'cde' です．n1 と n2 を等しくすることで 1 文字を取り出すこともできます．たとえば，

```
c1 = 'abcdefg'
do i = 1, 7
  print *,c1(i:i)
enddo
```

とすれば，1 文字ずつ縦に出力されます．

文字列配列の部分文字列を取り出す場合には，要素指定を先に，部分文字列指定を後に書きます．たとえば，1 次元の文字列配列 chr に対し，k 番目の要素の部分文字列は，

```
chr(k)(n1:n2)
```

のように指定します．かっこが連続するので，順番に気を付けて下さい．

文字列は連結することもできます．文字列の連結には演算子「//」を用います．たとえば，

```
c1 = 'abc'//'xyz'
```

と書くと，c1 には'abcxyz' という文字列が代入されます．文字列の連結は，文字列定数と文字列変数，文字列変数と文字列変数という組み合わせでも可能です．ただし，文字列変数に代入された文字列を連結するときには注意が必要です．たとえば，

```
character c1*10,c2*20
c1 = 'abc'
c2 = c1//'xyz'
```

と書いた場合，c2 に'abcxyz' という文字列が代入されると思ったら間違いです．正しくは

```
'abc␣␣␣␣␣␣␣xyz'
```

が代入されます．これは，上記のように 10 文字の文字変数 c1 に 3 文字の'abc' を代入しても，c1 の文字数は変わらないからです．末尾の不要なスペースを削除するには，部分文字列を使う必要があります．

```
character c1*10,c2*20
c1 = 'abc'
c2 = c1(1:3)//'xyz'
```

このプログラムならば，c2 に'abcxyz' が代入されます．

しかし，この方法は変数に代入されている文字数が不明のときには使えません．そこで，関数 trim が用意されています．trim は末尾のスペースを除去した文字列を返す組み込み関数です．たとえば，上記の例は，

```
character c1*10,c2*20
c1 = 'abc'
c2 = trim(c1)//'xyz'
```

と書くことができます．この結果も c2 には'abcxyz' が代入されます．

文字列をサブルーチンの引数にするときは，「(*)」を指定して宣言します．たとえば，

```
subroutine csubr1(chr)
  implicit none
  character(*) chr          ! 文字数は不要
   .......
```

の chr のように宣言します．Fortran の文字列には，文字コードの並びだけではなく，文字数の情報も含まれています．このため，(*) 指定のように文字数が明記されていなくても，コール側で指定した文字数の文字列として使用することができます．たとえば，

```
program ctest1
  implicit none
  call csubr1('abcde')
end program ctest1

subroutine csubr1(chr)
  implicit none
  character(*) chr
  print *,chr,len(chr)     ! 文字列 'abced' と文字数 5 が出力される
end subroutine csubr1
```

のようなプログラムにおいて，サブルーチン csubr1 中の print 文の出力は，文字列 'abcde' と文字数 5 になります．ここで，関数 len は文字列の文字数を取得する関数です．文字列に関する組み込み関数は，6.5 節で説明します．

文字列には大小関係があり，これを利用して if 文で条件分岐をすることもできます．2 個の文字列を比較するときは，以下の手順で行います．

(1) 文字数の長さが異なるときは，短い方の文字列の末尾にスペースを追加して文字数を等しくする
(2) 2 個の文字列を先頭から 1 文字ずつ比較していって，全て同じならば，“等しい (==)”
(3) 異なる文字があれば，最初に異なる文字の ASCII コードを比較して，コード値の大小が文字列の “大 (>)”または “小 (<)”

付録 E の表からわかるように，ASCII コードは，“スペース”＜“数字”＜“英大文字”＜“英小文字”の順で大きくなり，個々の文字は次のような順序になっています．

```
␣ < '0' < '1' <...< '9' < 'A' < 'B' <...< 'Z' < 'a' < 'b' <...< 'z'
```

たとえば，文字列の比較を使って次のようなプログラムを書くことができます．

```
character c1*10,c2*20
c1 = 'abcde'                          ! 3 番目が 'c'
c2 = 'abdce'                          ! 3 番目が 'd'
if (c1 < c2)  print *,'c1 < c2'
if (c1 == c2) print *,'c1 == c2'
if (c1 > c2)  print *,'c1 > c2'
```

この結果は，“c1 < c2”です．なぜなら，3 番目の文字が初めて異なり，c1 は 'c'，c2 は 'd' ですが，'c'<'d' だからです．なお，c1 は 10 文字，c2 は 20 文字ですが，後ろにスペースを補うので，この例の結果には無関係です．

6.3 出力における文字列の利用

最もよく文字列を使う場面は，print 文や write 文の中に入れて表示の補助に使うことでしょう．たとえば，

```
real spd,pres
spd = 10.0
pres = 960.0
print *,' Wind Speed = ',spd,'  Pressure = ',pres
```

と書けば，出力が

```
  Wind Speed =    10.0000000000000        Pressure =    960.000000000000
```

のようになって，どの数値がどの変数の値なのかが一目でわかります．

ここで，文字列はその長さに合わせて出力されています．これは，print 文が文字列に入っている文

字数の情報に合わせて出力するからです．この機能は，format で文字列の出力形式を指定するときにも使うことができます．文字列の出力指定には，A 編集 (表 5.1) を用いますが，A 編集は，「a」だけ書くと，文字列の文字数が出力幅になります．たとえば上記の print 文を，

```
  print 600,' Wind Speed = ',spd,' m/s'
  print 600,' Pressure = ',pres,' hPa'
600 format(a,f8.2,a)
```

と書けば，次のような文字幅に合わせた出力結果が得られます．

```
  Wind Speed =    10.00 m/s
  Pressure =   960.00 hPa
```

5.3 節の format の説明で紹介しましたが，出力形式指定を print 文や write 文の中に埋め込む書式も，文字列の利用方法の一つです．たとえば，

```
  print 600,x
  write(10,600) x
600 format(' x = ',es12.5)
```

というプログラムは次のように書き換えることができます．

```
  print "(' x = ',es12.5)",x
  write(10,"(' x = ',es12.5)") x
```

すなわち，format 文のかっこ以下を，両端のかっこを含めて文字列にして print 文や write 文の書式指定の位置 (*form*) に書き込みます．このとき，文字列変数を使えば，複数の場所で同じ format を使うときに便利です．

```
  character :: form*20="(' x = ',es12.5)"
  print form,x
  write(10,form) x
```

文字列変数を利用すれば，出力内容に応じて実行時に format を変更することも可能です．たとえば，

```
  real x
  character form*20
  if (abs(x) >= 1.e5) then
    form = "('x = ',es12.5)"
   else
    form = "('x = ',f10.5)"
  endif
  print form,x
```

とすれば，x の絶対値が 10^5 以上のときには es12.5 編集で，さもなくば f10.5 編集で出力されます．

また，部分文字列を利用して変更することも考えられます．たとえば，

```
  real x
  character form*20
  form = "('x = ',f10.5)"
  if (x >= 100.0) form(10:13)='12.3'        ! 10.5 から 12.3 に変更
  print form,x
```

のように書けば，x が 100.0 以上のときは f12.3 編集で，さもなくば f10.5 編集で出力されます．

6.4 数値・文字列変換

ここまでは write 文や read 文の書式指定に文字列を使用していましたが，装置番号の位置に文字列を記述することもできます．これは“文字列”から“数値”への変換，またはその逆変換を行うときに使用します．6.1 節で述べたように，「123」という数値と「'123'」という文字列は計算機内部の表現が異なりますが，処理の過程で 123 という数値からそれに相当する文字を作ったり，逆に'123' という文字列を数値として計算に利用したい場合があります．そもそも，書式付き write 文は“計算機内部の 2 進数”を“文字で表現された 10 進数”に変換して出力する動作であり，書式付き read 文はファイルなどから“文字で表現された 10 進数”を入力して“計算機内部の 2 進数”に変換する動作です．装置番号の位置に文字列を利用すると，その変換機能だけを使うことができるのです．

“数値”→“文字列”変換をするには write 文を用います．たとえば，

```
real x
character ch*20
x = 123.5
write(ch,"(f10.5)") x
```

と書けば，文字列変数 ch に' 123.50000' という文字列が代入されます．ただし，文字列に変換するときは，出力文字数以上の長さを持つ文字列変数を用意しておく必要があります．

open 文 (5.6 節) を使ってファイルをオープンするとき，ファイル名は文字列で指定する必要があります．そこで，file01.dat，file02.dat，...，file10.dat という通し番号の付いた 10 個のファイルが存在して，それぞれを装置番号 11〜20 としてオープンするときは，次のように書きます．

```
character name*20
integer m
do m = 1, 10
  write(name,"('file',i2.2,'.dat')") m
  open(10+i,file=name)
enddo
```

ここで，整数 m の出力編集を i2.2 にしているのは，スペースを '0' で埋めて，整数の出力を 01，02，... のようにするためです．

逆に，“文字”→“数値”変換をするには read 文を用います．たとえば，

```
real x
character ch*20
ch = '12345.0'
read(ch,*) x
```

と書けば，実数型変数 x に 12345.0 という“数値”が代入されます．

6.5 文字列に関する組み込み関数

文字列に関する組み込み関数は，trim や len の他にも色々あります．代表的なものを表 6.1 に示します．本節ではこれらの関数の中で比較的良く使うものの使用方法について説明します．

まず，6.2 節にも出てきた関数 len は，文字列の文字数を取得するための関数ですが，使用の際は注意が必要です．たとえば，

```
character c1*10
integer m,n
m  = len('abc')          ! m = 3
c1 = 'abc'
n  = len(c1)             ! n = 10
```

とすると，m の値は 3 ですが，n の値は 10 になります．これは，文字列変数の文字数が宣言時の文字数

表 6.1　文字列に関する組み込み関数

文字列関数	引数の型	関数値の型	関数の意味
trim(c)	文字列	文字列	末尾の空白を削除
len(c)	文字列	整数	文字列の文字数
len_trim(c)	文字列	整数	末尾の空白を削除した文字列の文字数
adjustl(c)	文字列	文字列	文字列を左にそろえた文字列
adjustr(c)	文字列	文字列	文字列を右にそろえた文字列
index(c,cp)	2 個の文字列	整数	文字列 c 中で文字列 cp が含まれていれば, その開始位置を返す　無ければ 0 を返す
repeat(c,n)	文字列と整数	文字列	文字列 c を n 個連結した文字列
char(n)	整数	1 文字の文字列	ASCII コード値を与えると, それに対応する半角英数文字を返す
ichar(c)	1 文字の文字列	整数	半角英数 1 文字を与えると, それに対応する ASCII コード値を返す

だからです．末尾のスペースを除去した文字列の長さを取得するには，関数 trim を使って除去してから関数 len で調べるか，関数 len_trim を使います．たとえば，上のプログラムで，最後の n への代入文を

```
n = len(trim(c1))  または  n = len_trim(c1)
```

で置き換えれば，n = 3 になります．

最後の char と ichar の使用例を示します．この二つの関数を組み合わせれば，read 文や write 文を使わなくても，文字と数値の変換をすることができます．たとえば，2 桁の整数を 3 桁の 8 進数で表示するプログラムは以下のようになります．

```
integer zero,num
character c1*10
num = 12
c1 = '000'
zero = ichar('0')                          ! '0' の文字コードを整数値に変換
c1(3:3) = char(zero+mod(num,8))            ! num の 1 の位の文字
c1(2:2) = char(zero+mod(num/8,8))          ! num の 8 の位の文字
c1(1:1) = char(zero+mod(num/64,8))         ! num の 64 の位の文字
print *,'decimal(',num,') = octal( ',trim(c1),' )'
```

ASCII コードでは，'0','1','2',...,'8','9' の順で値が 1 ずつ増加しています．そこで，先頭の'0' のコード値に 1 桁の数字を加えれば，その数字のコード値が得られます．同様に，大文字の英数字や，小文字の英数字も 1 ずつ増加して並んでいるので，先頭の文字である'A' や'a' のコード値に，その文字の順番の差の数字を加えて逆変換すれば，対応する文字を得ることができます．整数を 16 進数で表示するプログラムも考えてみて下さい．

第 7 章　配列計算式

第 6 章で説明した文字列の処理は，それ以前に説明した数値の計算と少し毛色が違います．数値計算が 1 個の数値と 1 個の数値の間の演算を基本としているのに対し，文字列の処理では，“文字列”という文字コードの集合をひとまとめにして取り扱い，文字列と文字列の連結や文字列変数への代入を一つの式で実行することができます．文字列とは文字コードの配列なのですから，これを数値計算的に処理するならば，本来は 1 文字ずつ処理しなければなりません．文字列の演算が可能なのは，Fortran コンパイラが，“文字列”という数値集合の処理を，内部で 1 文字ずつ処理するプログラムに変換してくれるからです．

Fortran では，この概念を一般の配列に拡張した“配列演算”が可能です．配列演算とは，数値の集合を配列名で代表させ，配列要素と配列要素の演算を，配列名と配列名の演算の形で記述するものです．配列演算を使えば，do 文を使わずに配列を使った計算を記述することができます．これを“配列計算式”と呼びます．配列計算式を使う利点は，単にプログラムが短くなるだけではありません．計算が最適になるように，くり返しの順序やメモリ配置をコンパイラが決めてくれるので，場合によっては，do 文を使って書くより高速に計算させることができます．

本章では，配列計算式の使い方について説明します．また，プログラム実行中に任意の要素数を持つ配列を用意する“動的割り付け”についても説明します．

7.1　基本的な配列計算式

配列演算は，配列の全要素に対して，全て同じパターンの演算をするのが基本です．このため，1 行の配列計算式で用いる配列は，次元と各次元の要素数が全て等しい“同型の配列”でなければなりません．その点に注意すれば，配列の全ての要素に同じ定数を代入するときや，2 個の配列の対応する要素間で全て同じ四則演算をするときに，あたかも配列名が一つの変数であるかのような記述をすることができます．たとえば，次のような記述が可能です．

```
real x(10,10),y(10,10),z(10,10)
x = 1.0
y = 2.0
z = x + 3.5*x*y**2
```

このプログラムを実行すると，配列 x の全ての要素は 1.0 になり，配列 y の全ての要素は 2.0 になり，配列 z の全ての要素は 15.0 $(=1.0+3.5\times1.0\times2.0^2)$ になります．このプログラムは do 文を使った以下のプログラムと同じ結果になります[48]．

```
real x(10,10),y(10,10),z(10,10)
integer i,j
do j = 1, 10
  do i = 1, 10
    x(i,j) = 1.0
    y(i,j) = 2.0
    z(i,j) = x(i,j) + 3.5*x(i,j)*y(i,j)**2
  enddo
enddo
```

配列計算式中の定数は，全要素に対して共通の値として計算します．

次のような，組み込み関数を使った配列計算式も可能です．

[48]ただし，ループの処理はコンパイラが自動的に生成するので，実際にこの通りの順序で計算するかどうかはわかりません．あくまでも，このプログラムと同じ結果になるという意味だと理解してください．

```
real a(10,10),b(10,10),c(10,10)
 .......
a = sqrt(b*sin(c)/(3.2*c + 1.5e-3))
```

この配列計算式を do 文で表すと，次のようになります．

```
do j = 1, 10
  do i = 1, 10
    a(i,j) = sqrt(b(i,j)*sin(c(i,j))/(3.2*c(i,j) + 1.5e-3))
  enddo
enddo
```

配列計算式においては，式中に含まれる全ての配列が同型でなければならないので，異なる次元や要素数の配列が混じっているとエラーになります．ただし，下限は異なってもかまいません．たとえば，

```
real a(3,3),r(-1:1,0:2)
 .......
a = 3.14*r**2
```

のように計算することができます．ただし，この配列計算式を do 文で表せば，

```
do j = 1, 3
  do i = 1, 3
    a(i,j) = 3.14*r(i-2,j-1)**2
  enddo
enddo
```

のように，対応する要素の位置がずれるので注意して下さい．

このように，配列計算式を使うとプログラムがシンプルになります．逆に，単一変数の計算式と配列計算式との区別がなくなるので，両者が入り交じるとプログラムがわかりにくくなり，エラーが見つけにくくなるという欠点もあります．

なお，サブルーチンの引数配列を使って配列計算をするときには注意が必要です．たとえば，

```
subroutine subr(a,b)
  implicit none
  real a(*),b(*)
  a = b**2              ! これはエラーになる
```

のようなプログラムでは，配列計算式はエラーになります．なぜなら，“*”を入れて配列宣言をした場合は，要素数が不定だからです．これに対し，

```
subroutine subr(a,b,n)
  implicit none
  real a(n),b(n)
  integer n
  a = b**2              ! この場合は OK
```

のように，整合配列を使うと，配列計算が可能になります．

7.2 部分配列

配列名だけを使った配列計算式では全ての要素について計算をしますが，常に全要素の計算が必要とは限りません．そこで，配列の一部を取り出した配列，“部分配列”を指定することができます．部分配列は“:”(コロン) を使って要素番号の範囲を指定します．たとえば，1 次元配列 a に対して，

```
a(n1:n2)
```

と書けば，a(n1)〜a(n2) という n2-n1+1 個の要素から構成された 1 次元配列として配列計算式に使うことができます．また，2 次元配列 b に対して，

```
b(n11:n12,n21:n22)
```

と書けば，b(n11,n21)〜b(n12,n21)〜b(n11,n22)〜b(n12,n22) という (n12-n11+1)×(n22-n21+1) の 2 次元配列として配列計算式に使うことができます．また，ある次元の要素番号を固定して，

```
b(n,n21:n22)
```

と書けば，b(n,n21)〜b(n,n22) という n22-n21+1 個の要素から構成された 1 次元配列として配列計算式に使うことができます．

逆に，要素番号に “:”だけを記述すると，「その次元の全要素」という意味になります．たとえば，

```
b(:,n21:n22)
```

と書けば，第 1 次元は全要素で，第 2 次元が n21〜n22 の範囲の要素から構成された 2 次元配列として配列計算式に使うことができます．このため，全ての要素番号を “:”にすると，配列全体を代表することになります．そこで，全配列要素を使った配列計算式を書くときにも “:”を使って配列であることを明示することができます．たとえば，7.1 節の 2 次元配列計算式，

```
a = sqrt(b*sin(c)/(3.2*c + 1.5e-3))
```

は，次のように書くことができます．

```
a(:,:) = sqrt(b(:,:)*sin(c(:,:))/(3.2*c(:,:) + 1.5e-3))
```

この方が，単一変数の計算式と区別できるので良いと思います．本書でも，これ以降はこの書式を使って全要素の配列計算式を記述します．

要素範囲の指定に “:整数値”を追加して，do 文のような増分値を指定することもできます．たとえば，1 次元配列 a に対して，

```
a(n1:n2:n3)
```

と書けば，a(n1)，a(n1+n3)，a(n1+2*n3)，... という要素からなる 1 次元配列として配列計算式に使うことができます．この場合，終了要素は n2 とは限らないので注意して下さい．たとえば，次のように書くことができます．

```
real a(10),b(5)
 .......
b(1:5) = a(1:10:2)
```

この結果は，b(1)=a(1)，b(2)=a(3)，b(3)=a(5)，b(4)=a(7)，b(5)=a(9) という 5 個の代入文と等価になります．増分値は負数を与えることも可能なので，次のように逆向きに代入させることも可能です．

```
real a(10),c(10)
 .......
c(1:10) = a(10:1:-1)
```

この結果は，c(1)=a(10)，c(2)=a(9)，...，c(10)=a(1) という 10 個の代入文と等価になります．

配列計算式で部分配列を使うときは，その部分配列がその他の配列と同型であれば良いので，宣言文での次元や要素数が一致している必要はありません．たとえば 1 次元配列なら，

```
real a(10),b(5)
 .......
a(3:5) = b(1:3)**2
```

のように書くことができます．また，2 次元配列の部分配列を使って，

```
real a(10,10),b(5,4),c(10)
 .......
a(3:5,6:8) = b(1:3,1:3)**2
```

などと書くことができます．この配列計算式を do 文で表せば，

```
do j = 6, 8
  do i = 3, 5
    a(i,j) = b(i-2,j-5)**2
  enddo
enddo
```

となります．ある次元の要素番号を固定した 2 次元配列は 1 次元配列として扱えるので，2 次元配列と 1 次元配列が混在した計算も可能です．たとえば，

```
real a(10,10),c(10)
 .......
c(3:8) = a(3,3:8)**2
```

と書くことができます．この配列計算式を do 文で表せば，

```
do j = 3, 8
  c(j) = a(3,j)**2
enddo
```

となります．

なお，部分配列を使った配列計算式で，左辺と右辺に同じ配列を使う場合には，単純に do 文で置き換えた動作と結果が異なる場合があるので注意が必要です．たとえば，

```
real a(10)
do i = 1, 10
  a(i) = i
enddo
a(2:10) = a(1:9)            ! この配列計算式は単純な do 文で置き換えられない
print *,(a(i),i=2,10)
```

というプログラムを考えます．この中の配列計算式 a(2:10)=a(1:9) を do 文にすると，

```
do i = 2, 10
  a(i) = a(i-1)
enddo
```

と置き換えられそうですが，実際にやってみると print 文の出力結果は異なります．配列計算式の場合は 1 から 9 までの異なる数字が出力されるのに対し，この do ループで置き換えると全て 1 が出力されます．これは，配列計算式が do ループのように 1 要素ずつ計算と代入をくり返すのではなく，まず右辺の配列要素計算を全て行って結果を補助配列に保存しておき，その後で保存した補助配列から左辺の配列への代入を行うためです．

この配列代入機能を利用すれば，次のように配列要素の順番を逆転させるのも簡単です．

```
a(1:10) = a(10:1:-1)
```

これと同じ動作をするプログラムを 1 回の do ループで書くのは難しいです．この代入機能は，配列計算

式を使う利点の一つです[49].

部分配列は要素が等間隔に並んでいますが，整数の 1 次元配列を使って要素指定をすれば，任意の順番で指定した配列を作ることができます．たとえば，次のような指定ができます．

```
real a(10)
integer m(3),i
do i = 1, 10
  a(i) = i
enddo
m(1) = 7;  m(2) = 2;  m(3) = 5
print *,a(m)
```

最後の出力結果は，

```
7.00000000000000  2.00000000000000  5.00000000000000
```

となります．すなわち，a(m) とは，[a(m(1)), a(m(2)) a(m(3))] という要素数 3 の配列のことです．a(m(2:3)) のような部分配列を使った要素指定や，7.4 節の配列構成子を使った要素指定も可能です．整数配列で要素指定をすると，要素指定配列の要素数を持つ配列になります．

なお，サブルーチンや関数副プログラムの引数に部分配列を与えた場合は，その部分配列と同型の配列としてデータをやりとりします．たとえば，b(10,10) という宣言をした 2 次元配列を使って，

```
call sub(b(5,:))
```

と書けば，10 個の要素を持つ 1 次元配列，[b(5,1), b(5,2),..., b(5,10)] をサブルーチンに与えたのと同じ結果になります．また，

```
call sub(b(1:3,1:3))
```

と書けば，要素数が 3×3 の 2 次元配列を与えたことに相当します．

4.5 節で説明したように，サブルーチンの引数に配列を与えた場合，サブルーチン側で受け取るのはその先頭要素アドレスだけで，サブルーチン側では，そのアドレスから順にたどって配列要素を決定します．このため，元の配列における並びが重要です．しかし，部分配列を引数に与えた場合には，その部分配列と同型の補助配列を経由して処理をするので，あたかも元の配列の一部を取り出した配列のように処理をすることができます．

7.3 where 文による条件分岐

配列演算は便利ですが，全ての要素について同じ形の計算をするので，要素の条件に応じて異なる処理をさせることはできません．そこで，配列要素の条件に応じて動作を分岐させるための where 文が用意されています．where 文は，以下のような形式です．

```
where (配列条件) 配列計算式
```

これは，“配列条件”に合った要素に対してのみ，“配列計算式”を実行するという意味です．たとえば，

```
real a(10,10),b(10,10)
 .......
where (a(:,:) > 0) b(:,:) = a(:,:)**2
```

のように書きます．この where 文の部分を do 文で表せば，次のようになります．

[49]ただし，配列要素数が非常に大きくて，コンピュータのメモリ利用可能限界に近くなると，補助配列の生成のおかげでメモリオーバーになる可能性もあります．

```
do j = 1, 10
  do i = 1, 10
    if (a(i,j) > 0) b(i,j) = a(i,j)**2
  enddo
enddo
```

where 文はブロック構文にすることもできます．ブロック where 文では，次のように else where 文を付加して，条件に一致しない場合の記述をすることもできます．

```
where (配列条件 1)
  配列計算式 1
   .......
 else where (配列条件 2)
  配列計算式 2
   .......
 else where
  配列計算式 0
   .......
end where
```

この場合，配列条件 1 を満足する要素は配列計算式 1 のブロックを実行し，配列条件 1 を満足せず，配列条件 2 を満足する要素は配列計算式 2 のブロックを実行し，... という具合に続いて，全ての条件を満足しない要素は配列計算式 0 のブロックを実行するという動作になります．中間の else where に関するブロックは省略可能です．if 文と違って，“then”が不要なこと，“全ての条件以外”を表す文が “else where”であること，ブロックの最後に “end where”文を付加すること，などに注意して下さい．たとえば，

```
real a(10,10),b(10,10)
 .......
where (3.0*b(:,:) > 0)
  a(:,:) = b(:,:)**2
 else where
  a(:,:) = b(:,:)**3
end where
```

のように書くことができます．このブロック where 文を do 文で表せば，次のようになります．

```
do j = 1, 10
  do i = 1, 10
    if (3.0*b(i,j) > 0) then
      a(i,j) = b(i,j)**2
     else
      a(i,j) = b(i,j)**3
    endif
  enddo
enddo
```

where ブロック中で使用する配列は，全て同型でなければなりません．このため，where ブロックの中に，単一変数の計算式や，次元や要素数の異なる配列計算式を入れることはできません．

7.4 配列構成子

配列演算を使えば，do 文を使わなくても配列要素の計算ができますが，これまでの書式では配列要素ごとに異なる定数を代入することはできません．そこで，1 次元配列だけですが，定数や変数などを並べて明示的に配列要素を与える“配列構成子”が用意されています．n 個の要素からなる 1 次元の配列構成子は以下の形式です．

```
(/数値 1, 数値 2, ..., 数値 n/)
```

このとき，n 個の数値は全て同じ型の数値型でなければなりません．たとえば，

```
real a(5)
a(:) = (/1.0,2.0,3.0,4.0,5.0/)
```

と書けば，右辺は実数型の 1 次元配列構成子であり，この配列計算式は，a(1)=1.0, a(2)=2.0, a(3)=3.0, a(4)=4.0，a(5)=5.0 という 5 個の代入文を実行したことに相当します．

数値の位置には変数や計算式を書くこともできます．たとえば，

```
real a(5),b
b    = 2.5
a(:) = (/b,2*b,3*b,4*b,5*b/)
```

と書けば，この配列計算式は，a(1)=b，a(2)=2*b，a(3)=3*b，a(4)=4*b，a(5)=5*b という 5 個の代入文を実行したことに相当します．

部分配列で指定することも可能です．たとえば，

```
real a(5),c(5)
a(:) = (/1.0,2.0,3.0,4.0,5.0/)
c(:) = (/a(3:5),a(1:2)/)
```

と書けば，この配列計算式は，c(1)=a(3)，c(2)=a(4)，c(3)=a(5)，c(4)=a(1)，c(5)=a(2) という 5 個の代入文を実行したことに相当します．

この配列構成子は，数値が全て定数であれば，次のように拡張宣言文の初期値代入に使うこともできます．

```
real :: a(5)=(/1.0,2.0,3.0,4.0,5.0/)
```

しかし，配列要素が多いと，数値を並べて配列構成子を記述するのに手間がかかります．そこで，5.2 節で説明した do 型並びを使って数値の設定をすることができます．do 型並びには，増分値なしと増分値付きの 2 種類があります．増分値を記述しなかったときの増分値は 1 です．

```
(データ 1, データ 2, ..., 整数型変数=初期値, 終了値)          ! 増分値なし
(データ 1, データ 2, ..., 整数型変数=初期値, 終了値, 増分値)   ! 増分値付き
```

この形式を配列構成子中に記述すると，まず整数型変数 (カウンタ変数) を初期値にして，データ 1，データ 2，... と並べ，次に，カウンタ変数に増分値を加えて，再度，データ 1，データ 2，... と並べ，カウンタ変数が終了値より大きくなるまでくり返して得られる数値を並べたことに相当します．

たとえば，

```
(/(n, n=1,5)/)
```

という do 型並びを使った配列構成子は，整数型の配列構成子，

```
(/1,2,3,4,5/)
```

と同じです．また，

```
(/(n, n**2, n=1,5)/)
```

という do 型並びを使った配列構成子は，整数型の配列構成子，

```
(/1,1,2,4,3,9,4,16,5,25/)
```

と同じです．増分値を指定して，

```
(/(n, n**3, n=1,8,2)/)
```

という do 型並びを使った配列構成子は，整数型の配列構成子，

```
(/1,1,3,27,5,125,7,343/)
```

と同じです．

複数の do 型並びを並べたり，通常の数値と混在させることも可能です．たとえば，

```
(/(real(n), n=0,2),1.5,1.8,(real(n), n=3,5)/)
```

という実数型の配列構成子は，

```
(/0.0,1.0,2.0,1.5,1.8,3.0,4.0,5.0/)
```

と同じです．このとき，型変換関数の real(n) を単に n と書くとエラーになるので注意して下さい．これは，n が整数型なので，n だけを書くと一つの配列構成子の中に実数型と整数型が混在するからです．

do 型並びは多重にすることも可能です．たとえば，

```
(/((i+j,i=1,3),j=1,4)/)
```

という do 型並びを使った整数型の配列構成子は，

```
(/2,3,4,3,4,5,4,5,6,5,6,7/)
```

と同じです．多重にした場合，カウンタ変数は内側が先に進みます．また，do 型並びを do ブロックの中に入れるときには，カウンタ変数が do 文のカウンタ変数とも重複しないようにしなければなりません．

本節最初の例は，次のように do 型並びを使って書くことができます．

```
real a(5)
a(:) = (/(n, n=1,5)/)
```

ここで，右辺は整数型の配列で，左辺は実数型の配列ですが，この場合は全要素に対して整数型から実数型への変換が行われて代入されるのでエラーにはなりません．なお，do 型並びは拡張宣言文での初期値代入に使うこともできますが，カウンタ変数はその宣言文より前に宣言されている必要があります．

配列構成子は 1 次元配列しか用意されていません．このため，2 次元以上の配列計算で使うときは，1 次元ごとの部分配列に代入するか，配列の形状を変換する組み込み関数 reshape を使います．reshape については 7.5 節で説明します．

7.5 配列に関する組み込み関数

配列を引数にする組み込み関数も色々用意されています．まず，配列要素を利用した計算に関する関数を表 7.1 に示します．

表 7.1 配列要素の計算に関する組み込み関数

配列関数	引数の型	関数値の型	関数の意味
sum(a)	配列*	配列要素の型	全ての配列要素の和
product(a)	配列*	配列要素の型	全ての配列要素の積
minval(a)	配列*	配列要素の型	全ての配列要素の最小値
maxval(a)	配列*	配列要素の型	全ての配列要素の最大値
dot_product(a,b)	2 個の 1 次元配列[50]	配列要素の型	2 個の 1 次元配列の内積
matmul(a,b)	2 個の 2 次元配列[51]	2 次元配列	2 個の 2 次元配列の行列積

たとえば，

```
real a(5)
integer i
a(:) = (/(i, i=1,5)/)
print *,sum(a)                  ! 配列 a の全要素の合計
```

と書けば，配列 a の全要素の合計 (15.0) が出力されます．部分配列を利用すれば，指定した範囲の配列要素だけを使った計算も可能です．たとえば，

```
real a(10)
integer i,n
a(:) = (/(i, i=1,10)/)
n = 5
print *,product(a(1:n))         ! 部分配列の要素 a(1)～a(n) の積
```

と書けば，a(1) から a(n) までの積 (120.0) が出力されます．

表 7.1 で引数の型に “*”の付いた関数を使う場合，引数配列の後に整数型の引数を追加すると，その整数値が指定する次元に関してのみ計算をすることができます．たとえば，

```
real b(3,4),x,y(4),z(3)
integer i,j
do j = 1, 4
  b(:,j) = (/(sin(0.5*(i+j)), i=1,3)/)
enddo
x = maxval(b)                   !  配列 b の全要素の最大値
y(:) = maxval(b,1)              !  配列 b の第 1 次元方向の要素の最大値
z(:) = maxval(b,2)              !  配列 b の第 2 次元方向の要素の最大値
```

と書けば，x には 2 次元配列 b の全要素の最大値が代入されますが，1 次元配列 y には，b の第 2 次元を固定して第 1 次元方向に要素を比較したときの最大値がそれぞれ代入されます．また，1 次元配列 z には，b の第 1 次元を固定して第 2 次元方向に要素を比較したときの最大値がそれぞれ代入されます．すなわち，次元を指定すると，その次元を抜いた要素数の配列が結果になります．

次に，配列情報に関する関数を表 7.2 に示します．

[50] a の要素数と b の要素数は等しくなければなりません．

[51] 行列積の計算上，a の第 2 要素数と b の第 1 要素数は等しくなければなりません．なお，a か b のどちらかは 1 次元配列でも計算できます．この場合，結果は 1 次元配列になります．

表 7.2　配列情報に関する組み込み関数

配列関数	引数の型	関数値の型	関数の意味
minloc(a)	配列	整数型配列	全ての配列要素の最小値の位置
maxloc(a)	配列	整数型配列	全ての配列要素の最大値の位置
lbound(a)	配列*	整数型配列	配列の各次元の最小要素番号リスト
ubound(a)	配列*	整数型配列	配列の各次元の最大要素番号リスト
shape(a)	配列	整数型配列	配列の各次元の要素数リスト
size(a)	配列*	整数	配列の全要素数
reshape(a,s)	配列と整数型配列	a の型の配列	配列 a を配列 s で指定された型の配列に変換する
allocated(a)	配列	論理	配列 a が割り付けられていれば真，さもなくば偽

表 7.2 で，引数の型に “*”の付いた関数は，引数配列の後に次元を示す整数型の引数を追加して，その次元に関する値を取り出すこともできます．たとえば，

```
real array(3,4)          ! 3 × 4 の配列
integer m,n1,n2
m  = size(array)         ! 全要素数 3 × 4=12
n1 = size(array,1)       ! 第 1 次元要素数 3
n2 = size(array,2)       ! 第 2 次元要素数 4
```

と書けば，m には，array の全要素数である 12 が代入され，n1 には array の第 1 次元の要素数である 3 が，n2 には array の第 2 次元の要素数である 4 が代入されます．

また，表 7.2 で関数値の型が “整数型配列”になっている関数は，引数に与えた配列の次元を要素数に持つ 1 次元整数型配列が戻り値です．その際，その 1 次元配列の各要素には，引数配列の各次元の情報が代入されています．たとえば，下限と上限を取得するための関数，lbound と ubound は以下のように使います．

```
real bb(-10:10,4)          ! 2 次元配列
integer blow(2),bupp(2)    ! 2 次元なので要素数 2 の配列
blow = lbound(bb)
bupp = ubound(bb)
print *,blow,bupp
```

ここで，2 次元配列 bb は，第 1 次元の下限が −10，上限が 10，第 2 次元の下限が 1，上限が 4 ですから，配列構成子で書けば，blow は (/-10,1/)，bupp は (/10,4/) になります．このとき，ubound(bb,2) のように，次元に関する引数を加えると，その結果は 1 個の整数 (この例では 4) になります．

reshape は，配列の形状を変換する関数です．そもそも，どんな次元の配列もメモリ上では 1 次元的に並んでいるので，2 次元配列 a(3,4) と 1 次元配列 b(12) のように，全要素数が等しい配列は同じように取り扱えるはずです．しかし，配列演算は次元および各次元の要素数が等しい同型の配列間でしか許可されていません．そこで，a(3,4) と b(12) の間で演算をするには，形式上，同型になるような変換が必要です．これを実行するのが関数 reshape です．reshape には，最初の引数に変換したい配列を与え，2 番目の引数に変換後の形状を表す 1 次元配列を与えます．ここで形状を表す 1 次元配列とは，変換後の配列が，k1×k2×...×kn の n 次元配列であれば，

```
(/k1,k2,...,kn/)
```

で与えられる要素数 n の 1 次元整数型配列のことです．たとえば，

```
real a(3,4)
integer i,j
a(:,:) = reshape((/((sin(0.5*(i+j)), i=1,3), j=1,4)/), (/3,4/))
```

と書くことができます．reshape の第 1 引数は do 型定数なので 1 次元配列ですが，これを第 2 引数の (/3,4/) で 3×4 の 2 次元配列に変換するよう指定しているわけです．

この配列の形状を表す 1 次元配列は，関数 shape を使って取得することができます．たとえば，上記の a(3,4) という宣言をした配列に対し，shape(a) は，1 次元整数型配列，(/3,4/) になります．そこで，上記の配列計算式は次のように書くことができます．

```
a(:,:) = reshape((/((sin(0.5*(i+j)),i=1,3),j=1,4)/),shape(a))
```

しかし，これではまだ do 型変数における i や j の範囲指定が定数のままです．そこで，ここは先ほど説明した関数 size で書き直すと良いでしょう．

```
a(:,:) = reshape((/((sin(0.5*(i+j)),i=1,size(a,1)),j=1,size(a,2))/),shape(a))
```

これなら，宣言文の要素数を変更するときに，この配列計算式を変更する必要はありません．

3.3.3 節で紹介した論理型データを配列にしておき，以下のような組み込み関数を利用すると，大量の論理演算を一度に行うことができます．

表 7.3　論理型の配列に関する組み込み関数

配列関数	引数の型	関数値の型	関数の意味
all(a)	論理型配列*	論理型	全ての配列要素の AND
any(a)	論理型配列*	論理型	全ての配列要素の OR
count(a)	論理型配列*	整数型	配列中の「真」の数

これらの関数は，引数の型に “*”があることからわかるように，引数配列の後に次元を示す整数型の引数を追加して，その次元に関する論理演算値を取り出すこともできます．たとえば，1 次元配列に代入されている数値に対し，それらが全て 50 より大きいかどうかを一度に判定するには，以下のように書くことができます．

```
real fdata(100)
logical flag(100)
...
flag(:)  = fdata(:) > 50        ! 論理演算も配列計算をすることができる
if (all(flag)) print *,'All Data > 50'
```

このように，論理演算も配列計算によって一度に判定することができ，その判定結果を代入した論理型の配列を使えば，それらを一度に判定することが可能になります．

7.6　配列の動的割り付け

配列を宣言するときには，整数定数を与えて要素数を明示しなければなりません．これは，その情報に基づいて，プログラムの動作開始時に計算で使用するメモリを確保するためです．しかし，最近のコンピュータではプログラムの実行中に必要になったメモリを確保したり，不要になったメモリを解放するという動作が頻繁に行われています．このような，プログラムの実行中に必要に応じてメモリを確保することを “動的割り付け”といいます．Fortran も Fortran90 から動的割り付けが可能になり，プログラムの動作に必要な要素数を持つ配列を実行中に確保できるようになりました．動的割り付けを使うと，メモリ効率の良い汎用性のあるプログラムを作成することができます．本節では，その方法を紹介

します.

まず，サブルーチンの中で必要に応じた要素数の配列を確保する簡単な方法として，サブルーチンの引数を使った配列宣言があります．たとえば，次のサブルーチンで real 宣言されている 2 次元配列 b がこれに相当します.

```
subroutine memory1(a,m,n)
  implicit none
  real a(m,n),b(m,n),x              ! 引数 m と n を使って宣言する
  integer m,n,i,j
  x = 10.0
  b(:,:) = x*reshape((/((i*j,i=1,m),j=1,n)/),(/m,n/))
  a(:,:) = b(:,:)**3
end subroutine memory1
```

配列 b は，このサブルーチンがコールされた時点で，引数 m と n の値を使って確保されます．このとき，b(m,n) のような単純な宣言だけではなく，b(0:2*m,n-1) のような，下限指定や計算式を使った宣言も可能です．ここで，配列 a も引数を使って宣言されていますが，a は引数に含まれているので 4.5 節で説明した整合配列です．整合配列は，引数配列の取り扱いを決めるしくみにすぎず，配列自体はサブルーチン側にありません.

これに対し，引数に含まれていない配列 b はローカル配列であり，宣言されたサブルーチンの一時メモリ領域に所属します．4.8.1 節で説明したように，一時メモリ領域に所属するローカル変数は，サブルーチンを呼び出した時点で生成されます．引数の値はサブルーチン開始時に確定しているので，これを使って配列に必要なメモリを確保することが可能なのです．ただし，これはあくまでもサブルーチンで宣言する配列についてのみ使える機能であり，メインプログラムの配列やモジュール中のグローバル配列としては使えません.

そこで，メモリの動的割り付けを明示的に行うしくみが用意されています．Fortran で動的割り付けを行うには，二つの手続きが必要です．一つは，割り付けたメモリを配列として使用するための配列名の宣言です．これは，割り付けたメモリの先頭アドレスを保持するためのメモリを用意することだと考えられます．もう一つは，その配列名に実際のメモリを割り付ける動作です．前者は宣言文なので非実行文，後者はプログラム実行中に行うので実行文です.

メモリを割り付ける予定の配列名は，allocatable 属性を付けて型宣言をします[52]．このとき，配列の次元情報をコロン “:”を使って示す必要があります．配列の次元はコロンの数で決まり，実行時に変更することはできません．たとえば，

```
real, allocatable :: ab(:),z2(:,:)
integer, allocatable, dimension(:,:) :: km1, km2
```

のように宣言します．1 行目のように，名前の後ろに次元情報を付加しても良いし，2 行目のように，dimension 属性を使うことも可能です．この例の場合，ab は 1 次元実数型配列，z2 は 2 次元実数型配列，km1 と km2 は 2 次元整数型配列として割り付けることができます.

配列の割り付けは allocate 文で行います．allocate 文は，次のように配列の要素指定をサブルーチンのようにかっこで囲んで記述します.

```
allocate ( ab(100), z2(0:m,-m:m), km1(k-1,2*k) )
```

かっこ内は，配列宣言と同じ形式です．ab や km1 のように，要素数に整数だけを与えると，その次元の下限は 1 になるし，z2 のように “:”を使って下限指定をすることも可能です．配列の数値型は型宣言の際に確定しているので，この例のように，一つの allocate 文で異なる数値型の配列を同時に割り付けることも可能です．allocate 文は実行文なので，z2 や km1 のように整数型変数や整数式の計算結果を

[52]配列に属性を付加する方法と dimension 属性については 4.8.1 節参照.

使って割り付けることも可能です．一旦割り付けられた配列は，通常の宣言文で宣言した配列と同様に使用することができます．

ただし，むやみに割り付けをくり返すと，コンピュータで利用できる容量を超える“メモリオーバー”になる可能性があります．そこで，不用になった配列のメモリ領域は deallocate 文で解放することができます．deallocate 文は，次のように配列名だけをかっこで囲んで指定します．

```
deallocate ( ab, km1 )
```

deallocate 文も異なる数値型の配列を同時に解放することが可能です．たとえば，先ほどのサブルーチン memory1 は，以下のように書き換えることができます．

```
subroutine memory1(a,m,n)
  implicit none
  real a(m,n),x
  real, allocatable :: b(:,:)          ! 割り付け用 2 次元実数型配列
  integer m,n,i,j
  allocate (b(m,n))                    ! m×n の 2 次元配列を割り付け
  x = 10.0
  b(:,:) = x*reshape((/((i*j,i=1,m),j=1,n)/),(/m,n/))
  a(:,:) = b(:,:)**3
  deallocate (b)                       ! メモリの解放
end subroutine memory1
```

なお，引数変数を使って宣言したローカル配列は一時メモリ領域に所属しますが，allocate 文で割り付けた配列は固定メモリ領域に所属します[53]．サブルーチンにおける一時メモリ領域は固定メモリ領域より使用可能容量が少ない可能性があるので，要素数の大きな配列を割り付けるときは allocate 文を使う方が良いでしょう．

allocate 文を使った動的割り付けは，メインプログラムでも利用できるし，モジュールの中で配列名宣言をして，グローバル配列として使用することも可能です．ただし，allocate 文で割り付けられた配列を，解放しないで再度 allocate 文で割り付けようとすると実行時エラーになります．また，割り付ける前にその配列を使用しようとしても実行時エラーになります．そこで，配列が割り付けられているか否かを確認する関数が用意されています．これが表 7.2 の最後にある関数 allocated です．allocated は論理型の値を返す関数で，allocatable 属性を持つ配列名を引数に与えると，その配列がすでに割り付けられていれば“真”，割り付けられていなければ“偽”を返します．3.3.3 節で説明したように，論理型データは if 文の条件として使うことができるので，たとえば次のように利用することができます．

```
real, allocatable :: a(:)
integer i,n
 .......
if (allocated(a)) then
  a(:) = (/(100*i,i=1,n)/)     ! 代入 1
 else
  allocate ( a(n) )
  a(:) = (/(200*i,i=1,n)/)     ! 代入 2
endif
```

この場合, 配列 a がすでに割り付けられていれば“代入 1”を実行し, 割り付けられていなければ, allocate 文で割り付けてから“代入 2”を実行します．

メモリの動的割り付けや解放は，それほど時間がかかる処理ではありません．計算の途中で一時的に利用する配列は，必要なときに割り付けて，不要になったら解放するようにしておけば，メモリの利用効率が高くて汎用性のあるプログラムになります．

[53]ただし，サブルーチン終了後も数値を保持するために配列を固定メモリ領域に所属させる場合には，配列名の宣言に save 属性 (4.8.1 節) を加えて，配列名も固定メモリ領域に所属させなければなりません．

第 8 章　モジュールによるプログラムのパッケージ化

4.7 節でモジュールを使ったグローバル変数の利用方法について述べましたが，モジュールの内部にはサブルーチンや関数副プログラムを含めることもできます．色々なサブルーチンや関数を集めたモジュールを作っておけば，必要に応じて use 文で呼び出せる便利なプログラムパッケージになります．モジュール中に記述したサブルーチンや関数は，コンパイル時に引数の型や形状などをチェックするので，エラーを見つけやすいという利点もあります．本章では，モジュールを使ったパッケージの作り方について説明します．

8.1　モジュール内部でのサブルーチンや関数副プログラムの記述

モジュール中にサブルーチンや関数を書くときは，contains 文を記述し，その後にサブルーチンや関数を記述します[54]．たとえば，次のように書くことができます．

```
module modsub1
   implicit none              ! implicit none は，ここに書いておけば全体で有効
contains
 subroutine dataset1(s,x,n)
   real s(n),x
   integer n,i
   do i = 1, n
      s(i) = i*x
   enddo
 end subroutine dataset1
 function cube(x)
   real cube,x
   cube = x**3
 end function cube
end module modsub1
```

このように contains 文の後には複数のサブルーチンや関数を書くことができます．また，implicit none は，module 文の直下で一度宣言しておけば，そのモジュール内部全体で有効になるので，以下のサブルーチンや関数では省略することができます．

モジュールに記述したサブルーチンや関数は，use 宣言をしたルーチン内でのみ有効です．たとえば，このモジュール modsub1 の中にあるサブルーチン dataset1 や関数 cube を使用するには，以下のように記述します．

```
use modsub1                   ! モジュールの use 宣言がなければリンクエラーになる
implicit none
real a(10)
integer i
call dataset1(a,10.0,10)
do i = 1, 10
   print *,i,cube(a(i))       ! 関数 cube は型宣言をする必要がない
enddo
```

このプログラムのように，モジュールに記述した関数を使うときには，関数値の型宣言は不要です．これは，モジュールの中に関数値の情報が含まれているためです．そればかりではなく，モジュールの中に記述したサブルーチンや関数は引数のチェックもしてくれます．このため，上記のプログラム中の call 文を以下のように修正するとコンパイルエラーになります．

[54] contains 文は内部副プログラムを記述するときにも使えます (10.2 節参照)．なお，本章では記述を簡単にするために “関数副プログラム” を単に “関数” と書きます．

```
call dataset1(a,10,10.0)     ! ① 変数の型が一致していない
call dataset1(a,10.0)        ! ② 引数が足らない
```

①の call 文がエラーになるのは，第 2 引数が整数型で第 3 引数が実数型になっていますが，モジュールで宣言しているサブルーチンの引数は，第 2 引数が実数型，第 3 引数が整数型なので型が一致していないからです．また，②の call 文がエラーになるのは引数が 2 個しかないからです．モジュールを使えば，より安全に使用できるサブルーチンや関数のパッケージができます．

ただし，型宣言のチェックは少し厳密であることにも注意が必要です．たとえば，次のように利用することはできません．

```
use modsub1
implicit none
real x
call dataset1(x,10.0,1)     ! ③ 引数が配列ではない
print *,x
```

これは，モジュールの subroutine 文に書かれている第 1 引数 s の宣言が配列 s(n) であるのに対し，③の call 文では単一変数を与えているためです．4.5 節で説明したように，通常のサブルーチンであればこれはエラーにはならず，正常に動作します．

ただし，dataset1 のように整合配列を使った場合や，s(*) のような不定形の配列宣言をした場合には，次元の異なる配列を与えることはできます．たとえば，

```
use modsub1
implicit none
real b(10,10)
call dataset1(b,10.0,100)     ! これはエラーではない
print *,b
```

というプログラムはエラーにはなりません．

次元も含めたチェックを行うときは，モジュール中の引数配列を “:”を使って宣言します．“:”を使って宣言すると，そのサブルーチンをコールしたときに，配列の型だけでなく次元の情報も渡されます．加えて配列要素数の情報も渡されるので，整合配列のように引数に配列要素数の情報を入れておく必要はありません．引数配列名を与えるだけで，配列要素数はサブルーチン側の配列関数で取り出すことができます．

たとえば，次のようなサブルーチンを含んだモジュールを作ることができます．

```
module modsub2
   implicit none
contains
 subroutine dataset2(d,x)
   real d(:,:),x
   integer i,j
   do j = 1, size(d,2)
      do i = 1, size(d,1)
         d(i,j) = i*j*x
      enddo
   enddo
 end subroutine dataset2
end module modsub2
```

このサブルーチン dataset2 の引数にはどこにも配列要素数の情報が書かれていませんが，それでも size 関数は call 文で与えた引数配列の宣言要素数を返します．たとえば，次のようにこのサブルーチンを利用することができます．

```
use modsub2
implicit none
real b(10,5)
integer i,j
call dataset2(b,10.0)
do j = 1, 5
   do i = 1, 10
      print *,i,j,b(i,j)
   enddo
enddo
```

“:”を使って宣言すると，そのサブルーチンをコールしたときに，引数に与えた 2 次元配列 b の先頭アドレスだけでなく，宣言した形状 (b(10,5)) の情報もサブルーチン側に伝わるところがポイントです．なお，下限を指定したいときには，“real d(0:,0:)”のように下限のみ指定して宣言します．

また，配列の次元チェックもするので，

```
use modsub2
implicit none
real a(10)
call dataset2(a,10.0)     !  第 1 引数が 2 次元配列でない
```

のように，1 次元配列を与えてコールするとエラーになります[55]．

モジュール内部のサブルーチンや関数では，その引数変数名も公開しています．このため，それぞれの引数変数名に値を代入した形でサブルーチンをコールすることもできます．この場合，引数を記述する順番は自由です．たとえば，最初の dataset1 の call 文は以下のように書き換えることができます．

```
call dataset1(a,10.0,10)
        ↓
call dataset1(x=10.0,n=10,s=a)
```

このように，モジュールの中にサブルーチンや関数を記述して使用すると色々便利な機能が付いてくるので，ある程度汎用性のあるプログラムができたら，モジュールにまとめてパッケージ化することをお勧めします．

8.2 変数宣言との共存

4.7 節でモジュールを導入したのはグローバル変数を利用するためでした．モジュール中で宣言した変数は，そのモジュールを use 宣言したルーチンにおける共有変数になるからです．この共有変数は，サブルーチンや関数を含めても同様に利用できますが，宣言文は contains 文より前に記述します．たとえば，

```
module modsub3
   implicit none
   real :: yb = 2
contains
 subroutine dataset3(d,x)
   real d(:,:),x
   integer i,j
```

[55]配列の形状を無視して間接アドレスに頼ってプログラムを書くのは少し危険な利用方法です．筆者のような昔の人間には「自由度が下がる」という気持ちがあるのですが，これから Fortran を利用する皆さんはこのようなチェック機能を積極的に利用して，安全なプログラムを書くことを心がけた方が良いと思います．

```
   do j = 1, size(d,2)
      do i = 1, size(d,1)
         d(i,j) = i*j*x*yb        ! yb は共有変数
      enddo
   enddo
 end subroutine dataset3
end module modsub3
```

のように，変数 yb の宣言文を contains 文より前に書けば，yb はグローバル変数として利用することができます．それと同時に，yb はモジュール内部のサブルーチンや関数にとっても共有変数になるので，サブルーチン dataset3 の中で使っている yb は，サブルーチン内で宣言しなくてもエラーにはならず，contains 文より上で宣言した yb です．逆に言えば，不用意に変数を使ってしまうこともあるので，4.7 節と同様に名前の付け方には注意しましょう．

なお，同じ名前に対して，contains 文より上での変数宣言とサブルーチンや関数内部でのローカルな宣言を同時に行ってもエラーにはなりません．この場合にはローカルな宣言が優先します．たとえば，上記のサブルーチン dataset3 の中で yb を宣言してもエラーではなく，外部宣言の yb とは別の変数として利用することができます．

8.3 private と public

以上のように，モジュールを使えば，サブルーチンや関数，およびそれに付随したグローバル変数をひとまとめにして，パッケージ化することができます．このようなパッケージを作っておけば，ユーザー側は，use 宣言をするだけでそのモジュールの様々な機能が使えます．

しかし，全ての変数やサブルーチンをユーザーに公開する必要はありません．モジュール内部のサブルーチンが使うだけの関数や共有変数を含めることも考えられます．しかし，use 文を使って利用宣言をすれば，そのモジュール中で宣言したものは全て公開されるので，余計なものも見えてしまいます．use 文に only 句を加えて使用を制限することもできますが，最初から使用しないことがわかっているものを利用者側が隠すのでは，あまり使いやすいパッケージとは言えません．

そこで，use 宣言をしたルーチンでの公開・非公開をモジュール側で制御する方法が用意されています．まず，変数に関しては，拡張宣言文に private 属性を付けて宣言すれば非公開になります．

たとえば，

```
module modsub4
   implicit none
   real yb
   real, private :: zc=0           ! zc は非公開
contains
 subroutine dataset4(d,x)
   real d(:,:),x
   integer i,j
   zc = zc + 1
   do j = 1, size(d,2)
      do i = 1, size(d,1)
         d(i,j) = i*j*x*(yb+zc)
      enddo
   enddo
 end subroutine dataset4
end module modsub4
```

のように宣言すれば，yb は通常のグローバル変数として use 宣言をしたルーチン中で利用できますが，zc はできません．しかし，モジュール内部のサブルーチンでは zc を共有変数として使うことができます．

しかし，モジュール中のサブルーチンはこの方法で非公開にすることができないし，プログラムが複雑になってくると，非公開の変数の方が公開したい変数より多くなることもあります．そこで，モジュー

ルに private 文を記述することで，デフォルトの状態を非公開にすることができます．全てを非公開にしておいて，公開したいものだけ，public 文を使って公開を宣言します．public 宣言は，サブルーチン名，関数名，変数名などが混在しても問題はなく，全て同時に宣言することができます．public 文は複数書いても良いし，公開宣言の順番は動作とは無関係です．

たとえば，

```
   module modsub5
      implicit none
      public yb, dataset5            ! 公開宣言
   private                           ! これ以降は原則非公開
      real :: yb, zc=0
   contains
    subroutine dataset5(d,x)
      real d(:,:),x
      integer i,j
      zc = zc + 1
      do j = 1, size(d,2)
         do i = 1, size(d,1)
            d(i,j) = cube(i*j*x*(yb+zc))
         enddo
      enddo
    end subroutine dataset5
    function cube(x)
      real cube,x
      cube = x**3
    end function cube
   end module modsub5
```

などと書けば，変数 yb とサブルーチン dataset5 は use 文の宣言をしたルーチン中で利用できますが，変数 zc や関数 cube は利用できません．これらはあくまでもモジュール内でのみ共有する変数や関数として働きます[56]．なお，private 文は，implicit 宣言と contains 文との間であればどこに書いても良いのですが，本書では公開・非公開の境界を示すため，public 文の後に記述しています．

8.4 value 属性

4.4 節で説明したように，Fortran におけるサブルーチンや関数の引数は，“call by reference”，すなわち間接アドレスによってデータを受け渡すのが基本で，引数を利用するときはこのことを前提にしてプログラムを書く必要があります．しかし，モジュール内部のサブルーチンや関数を利用すると，宣言側とコール側の引数対応をチェックしてからコンパイルするので，引数の属性を “call by value”の直接アドレスに変更して，コール側からデータを渡すだけの利用に特化した引数にすることも可能です．

“call by value”の引数を使うときは，そのデータを受け取るサブルーチンの引数の変数宣言に value 属性を付加します．たとえば，次のようなモジュール内部のサブルーチン value1 を書くと，value 属性を指定した引数 x は，直接アドレスを使ったデータ受け取り専用の変数になります．

```
   module modsub6
      implicit none
   contains
    subroutine value1(x, y)
      real, value :: x
      real :: y
      y = x**2                       ! y は通常の引数なので代入すれば戻り値になる
!     x = 1                          ! これは引数のテスト用
    end subroutine value1
   end module modsub6
```

[56]変数に関しては public 属性を付加することで公開することも可能です．

これに対し，引数 y は real 宣言をしているだけなので，通常の “call by reference”の間接アドレス引数で，例のように値を代入すると戻り値になります．すなわち，サブルーチン value1 は，x に与えた実数値の 2 乗を y に代入して戻り値にするという動作になります．

4.4 節で説明したように，直接アドレス指定でのデータ渡しとは，サブルーチン側のメモリ領域に直接データを書き込むことなので，ローカル変数と同じレベルのメモリアクセスが可能になります．すなわちメモリアクセスが速くなるのが利点です．逆にいえば，サブルーチン側でそのデータを頻繁に使うことがなければ，value 属性を付加するメリットはあまり無いとも言えます．

なお，“call by value”の引数にサブルーチン内部で値を代入することは原理的には可能なはずなのですが，コンパイラによってはエラーになることがあるようです，たとえば，上記のサブルーチンの中にあるコメントを外して，x=1 を実行するように修正したプログラムを手元の 2 種類のコンパイラでやってみると，一つはエラーにならずにコンパイルが終了しましたが，もう一つはコンパイルエラーになって動きませんでした．前者のエラーにならなかったプログラムを実行してみると，コール側で与えた引数変数の値が戻り値のように変化しなかったので，“call by value”になっていることは確認できました．とはいえ，エラーが出る可能性があるのですから，基本的には value 属性を与えた変数は値を受け取るだけに限定し，他の値を代入して使わないようにした方がいいと思います[57]．

value 属性を持つ引数は，モジュール内部のサブルーチンや関数以外でも利用することが可能です．ただし，デフォルトの変数受け渡しが “call by reference”なので，コール側で引数の確認ができるしくみを併用する必要があります．このしくみには，10.2 節の内部副プログラムや 10.3 節の interface 文の利用があります．

8.5 総称名機能

Fortran の組み込み関数には “総称名機能”があります．たとえば，関数 abs(x) は，引数 x が整数型のときには整数型の絶対値を計算して整数型の値を返し，x が実数型のときには実数型の絶対値を計算して実数型の値を返します．すなわち，引数の型に応じて使用する関数を選択して計算してくれるのが総称名機能です．モジュール内部のサブルーチンや関数は，この総称名機能を実現することができます．総称名機能を使えば，引数の型に応じて計算をさせる関数を選択することができるだけではなく，単一変数か配列かで別の関数を選択することもできるし，引数の数が異なるサブルーチンを同じ名称で呼び出すことも可能です．

8.1 節で説明したように，モジュール内部のサブルーチンや関数を使用すると，コンパイル時に各引数の型や形状などをチェックします．総称名機能とは，異なる引数で指定する複数のサブルーチンまたは関数をグループ化しておいて，そのグループの中から引数の型などが一致するものを引数チェックによって選択する機能です．このグループに付けられた名前が総称名になります．

このグループ化と総称名の指定は，interface 文を使って行います．すなわち，次のような総称名を指定した interface 文と end interface 文ではさんだブロックの中に，module procedure 文を使って，それぞれの引数に応じたサブルーチンや関数の名前をリストアップしておきます．

```
interface 総称名
   module procedure サブルーチン名または関数名 1
   module procedure サブルーチン名または関数名 2
    .......
end interface
```

interface 文で指定したブロックは宣言文の一種なので contains 文より前に書かねばなりません．

[57] 筆者は Fortran と C 言語の両方を利用しているので，念のために C 言語のような “call by value”の引数利用が可能かどうか確認してみたのですが，よく考えると直接アドレスの引数に値を代入して別の用途に使うのは，変数を減らしてメモリをちょっとだけ節約する程度のメリットしかありません．よって，Fortran における value 属性変数は，別の用途に利用できない変数であるとの前提で用いる方がいいと思われます．

総称名の記述例を以下に示します.

```
module modsum
   implicit none
   public sumarray                          ! sumarray は公開する
private
   interface sumarray                       ! sumarray が総称名になる
      module procedure sumarray_int
      module procedure sumarray_real
      module procedure sumarray_realm
   end interface
contains
 function sumarray_int(a)                   ! 引数が整数配列のとき
   integer sumarray_int,a(:)
   integer i,n
   n = 0
   do i = 1, size(a)
      n = n + a(i)
   enddo
   sumarray_int = n
 end function sumarray_int
 function sumarray_real(a)                  ! 引数が実数配列のみのとき
   real sumarray_real,a(:),s
   integer i
   s = 0
   do i = 1, size(a)
      s = s + a(i)
   enddo
   sumarray_real = s
 end function sumarray_real
 function sumarray_realm(a,c)               ! 引数が実数配列と実数のとき
   real sumarray_realm,a(:),c,s
   integer i
   s = 0
   do i = 1, size(a)
      s = s + a(i)*c
   enddo
   sumarray_realm = s
 end function sumarray_realm
end module modsum
```

モジュール modsum には 3 個の関数 sumarray_int，sumarray_real，sumarray_realm があり，それぞれ引数の型や数が異なりますが，これらを interface 文で sumarray という総称名のグループにしています．なお，private 文を入れているので，総称名 sumarray には public 宣言が必要ですが，逆に sumarray_int などの内部関数が外部から利用できないようになっているというメリットがあります．

たとえば，このモジュールは以下のように利用することができます.

```
use modsum
implicit none
real a(10)
integer b(10),i
do i = 1, 10
   a(i) = i
   b(i) = -i
enddo
print *,sumarray(b),sumarray(a),sumarray(a,-3.0)
```

この場合，配列 b は整数型配列なので，sumarray(b) は sumarray_int を利用して計算します．これに対し，配列 a は実数型配列なので，sumarray(a) は 1 引数の sumarray_real を利用して計算し，sumarray(a,-3.0) は 2 引数の sumarray_realm を利用して計算します．もし引数がどの関数とも一

致しない場合にはエラーになります．よって，汎用性を持たせるには，可能性がある引数並びを持つ関数を全て用意しておく必要があります．

なお，sumarray_real と sumarray_realm のように，引数が 1 個程度の増減の場合には，総称名機能を使わなくても，引数の宣言に optional 属性を加えることで処理することができます．たとえば，以下のようなモジュール関数を書くことができます．

```
module modrsum
   implicit none
contains
 function rsumarray(a,c)
   real rsumarray,a(:),s,c0
   real, optional :: c          ! optional 属性を付けて宣言する
   integer i
   c0 = 1
   if (present(c)) c0 = c       ! present 関数で c の存在を確認する
   s = 0
   do i = 1, size(a)
      s = s + a(i)*c0
   enddo
   rsumarray = s
 end function rsumarray
end module modrsum
```

この関数のポイントは，引数 c の宣言に optional 属性が付いていることです．optional 属性が付いた引数は，関数 present(x) を使って，存在の有無を確認することができます．present(x) は，引数 x が存在すれば「真」，存在しなければ「偽」を返す論理型関数です．関数 rsumarray では引数 c の存在を確認して，存在する場合には変数 c0 にその値を代入し，存在しなければ，最初に代入した c0=1 のままで計算を行うようになっています．

このモジュールを使って，

```
 print *,rsumarray(a),rsumarray(a,3.0)
```

と書けば，引数の有無に応じて，異なる動作をした結果を出力します．

総称名機能を使えば，ユーザーから見てより使いやすいパッケージになります．積極的に利用してみてはいかがでしょう．

第 9 章　構造体の利用

これまでの説明で取り扱ってきたデータは，整数型や実数型といった“数値”，文字の並びである“文字列”，および論理型の“真偽”でした．これらは必要に応じてコンパイラが解釈を変えていますが，本質的にはコンピュータが内蔵しているデータ型を直接利用しています．それぞれのデータを複数並べた“配列”を利用することもできますが，これもある特定のデータ型を並べたものなので，コンピュータのハードウェアで直接実現することができます．

これに対し，複数のデータの集合体を新しいデータ型として定義し，定義されたデータ型を持つ変数を宣言してプログラムの中で取り扱うことができます．このデータ型を“構造型”といい，構造型を持つ変数を“構造体”といいます．たとえば，学生の情報である，“学籍番号”，“名前”，“成績”，“成績の平均”をまとめた構造型を定義し，それらを一つの変数で代表させて処理をすることができます．このとき，内部に含まれるデータの型が全て同じである必要はありません．学籍番号や成績は整数型で，名前は文字列，成績の平均は実数型といったそれぞれの特性に応じた型を使って，それらを 1 個のデータとして取り扱うことができます．また，構造型の定義の中に別途定義した構造体を含めることもできるし，構造体間の演算を定義することもできます．本章では，この構造体の利用方法について説明します．

9.1　構造型と構造体の宣言

構造体を利用するには，まず構造型の宣言が必要です．構造型は type 文と end type 文で構成要素 (成分) の宣言文をはさんだブロックで宣言します．すなわち，以下のような形式です．

```
type 構造型名
   宣言文 1
   宣言文 2
    .......
end type 構造型名
```

たとえば，学生の情報である，“学籍番号 id (整数型)”，“名前 name (20 文字の文字列)”，“5 講義の成績配列 score(5) (整数型)”，“成績の平均値 average (実数型)”という情報をまとめた構造型の場合は，以下のように宣言します．

```
type student
   integer id, score(5)
   character name*20
   real average
end type student
```

ここで，id と score は同じ整数型なので，簡単のために一度に宣言していますが，分けて宣言してもかまいません．この宣言の順番は 9.3 節で説明する構造体構成子の並びに影響します．

定義した構造型を持つ構造体の宣言には type(構造型) 文を使います．

```
type(構造型) 変数 1，変数 2，...
```

“変数”のところは配列にすることも可能です．たとえば，上記の student 構造型を持つ構造体を利用するときは次のように宣言します．

```
type(student) s1,st(100)
```

ここで，s1 は 1 変数であり，st は要素 100 個の 1 次元構造体配列です．なお，type(構造型) による変数宣言文は，type 文による構造型の定義より後に書かなければなりません．

type(構造型) による宣言文は，拡張宣言文にして属性を付加したり初期値を代入したりすることもできます．

```
type(構造型)［, 属性 1，属性 2，...］:: 変数 1［=初期値 1］, 変数 2［=初期値 2］，...
```

ここで，初期値が必要なときには構造体構成子を与えます．構造体構成子は 9.3 節で説明します．

たとえば，save 属性を付加した構造体を宣言するときは，

```
type(student), save :: sa,sb
```

のように書きます．

9.2 構造体の成分

宣言した構造体の成分に値を代入したり，成分の値を使って計算をするときは，“構造体%成分”のように“%”でその成分を指定します．指定した各成分はそれぞれの型を持つ変数や配列として取り扱うことができるので，たとえば上記のように宣言した student 構造体 s1 の成分に値を代入するには，

```
s1%id       = 1000
s1%name     = 'Taguchi'
s1%score(:) = (/51, 60, 72, 33, 45/)
s1%average  = 52.2
```

のように書くことができます．また，成分配列 score に代入されている成績の平均を計算して成分 average に代入するには，

```
s1%average = 0
do i = 1, 5
   s1%average = s1%average + s1%score(i)
enddo
s1%average = s1%average/5
```

のように書くことができます．

同じ構造型を持つ構造体間では代入文を使って全ての成分を一度に代入することができます．たとえば，上記のように宣言した構造体 s1 と構造体配列 st がある場合には，

```
st(25)  = s1
st(1:5) = st(6:10)
```

のように代入することができます．

構造型の中には，別途定義した構造型を持つ構造体を入れることもできます．このとき，その構造型に含まれている成分の構造体の成分を利用するときは“構造体%成分の構造体%成分の構造体の成分”のように続けて指定します．たとえば，上記の student 構造型を利用した別の構造型を

```
type department
   integer n
   type(student) st(1000)
end type department
```

のように定義して，

```
type(department) d1
```

のように構造体宣言をした場合には，

```
d1%st(25)%id = d1%st(1)%score(2) + 133000
```

のように“成分の成分”を指定して利用します．

構造体は print 文や write 文で出力することができますが，その際は，宣言された成分ごとに並んで出力されます．たとえば，本節の最初のように student 構造体の成分に値を代入しておいて，

```
print *,s1
```

のように出力した場合には，

```
100     51     60     72     33     45 Taguchi              52.2000000000000
```

のような並びで出力されます．format を使って出力形式の指定をする場合には，成分の順番で変数が並んでいることを前提にして対応する編集記述子を並べる必要があります．

以上の説明で示した例で構造体を使ったプログラムを書くと以下のようになります．

```
program type_test1
   implicit none
   integer i
   type student
      integer id, score(5)
      character name*20
      real average
   end type student
   type(student) s1

   s1%id = 100
   s1%name = 'taguchi'
   s1%score(:) = (/51, 60, 72, 33, 45/)
   s1%average = 0
   do i = 1, 5
      s1%average = s1%average + s1%score(i)
   enddo
   s1%average = s1%average/5
   print *,s1
end program type_test1
```

この例のように type 文による構造型の定義は，type(構造型) 文による構造体の宣言より前に書かねばならないことに注意してください．

9.3 構造体構成子

構造体にデータを代入するには，個々の成分に代入する方法以外に，構造体構成子を使う方法があります．構造体構成子は，次のように構造型名を関数名のように使って，各成分を宣言した順番に並べて与える形式です．

構造型名（**成分** 1，**成分** 2，...）

たとえば，9.2 節のプログラム例の中にある成分ごとの代入文は，以下のように書くことができます．

```
s1 = student(1000,(/51, 60, 72, 33, 45/),'Taguchi',52.2)
```

構造体構成子の引数には変数や計算式を使うことも可能です．また，成分の数値型と異なる数値を与えた場合には成分の数値型に変換して代入します．

配列成分に全て同じ値を代入するのであれば，配列構成子を使わなくてもかまいません．たとえば，次のように書くこともできます．

```
s1 = student(1000,10,'Taguchi',10.0)
```

このとき，配列 s1%score の要素には全て 10 が代入されます．

type(構造型) 文は拡張宣言も可能なので，初期値を代入することができますが，このときも構造体構成子を使います．たとえば，次のように書くことができます．

```
type(student) :: s1 = student(1000,(/51, 60, 72, 33, 45/),'Taguchi',52.2)
```

ただし，初期値の代入に使う場合には，全ての成分が定数でなければなりません．

9.4 モジュールを使った構造体パッケージ

構造型は，9.2 節のプログラム例のようにメインプログラムや通常のサブルーチンの中で定義することも可能ですが，定義したルーチン内でのみ有効なので，複数のルーチンで同じ構造型を使用する場合にはそれぞれのルーチンに同じ type 文を記述しなければなりません．そこで，プログラム全体で利用する構造型を定義する場合には，モジュールに入れて共有するのが良いでしょう．モジュールで定義すれば，そのモジュール内部のサブルーチンや関数副プログラムでも使えます．構造型は，データ集合の取り扱いを容易にすることを目的として定義するのですから，それを操作するサブルーチンや関数をモジュールにまとめた構造体パッケージを作っておくと便利です．

たとえば，9.1 節の student 構造型を定義するモジュールとして，以下のようなものが考えられます．

```
module modstudent
   implicit none
   type student
      integer id, score(5)
      character name*20
      real average
   end type student
contains
 subroutine setaverage(s)
   type(student) s
   integer i
   s%average = 0
   do i = 1, 5
      s%average = s%average + s%score(i)
   enddo
   s%average = s%average/5
 end subroutine setaverage
end module modstudent
```

このモジュール内部にあるサブルーチン setaverage は，9.2 節のプログラム例の中にある成分配列 score の平均を計算して成分 average に代入する部分を取り出したものです．なお，private 文を書いてデフォルトを非公開にするときには，構造型の student も public 文で公開宣言する必要があります[58]．

たとえば，このモジュールとその内部サブルーチンを利用すれば，9.2 節のプログラム例を以下のように書き換えることができます．

```
program type_test2
   use modstudent
   implicit none
   type(student) s1
   s1 = student(100,(/51, 60, 72, 33, 45/),'taguchi',0.0)
   call setaverage(s1)
   print *,s1
end program type_test2
```

このように，構造型の定義が入ったモジュールの use 宣言をするだけで，構造体の宣言や内部サブルーチンの利用が可能になります．

[58]拡張宣言文のように type 文に public 属性と“::”を付けて，“type, public :: student”のように公開を指定することも可能です．

9.5 構造体演算

Fortran には複素数型が用意されていて，複素数を使った計算式を実数型と同様に書くことができます．複素数型はコンピュータ内蔵のデータ型ではなく，2 個の実数型を 1 個のデータ型として取り扱う一種の構造体です．同様に，複数の数値から構成される数学的要素を構造体で表して，その要素間の演算や，要素と実数型との演算などを定義しておけば，その要素に関する計算を直感的に理解しやすい形式で記述することができます．本節では，2/3 とか 12/7 のような「整数／整数」で表される “分数”を例として，構造体で数学的要素の演算を処理する手法について説明します．

まず，分数を次のような 2 個の整数 num と den を成分とする fraction 構造型で表現するとします．

```
type fraction
   integer num,den
end type fraction
```

ここで，num を分子の整数，den を分母の整数と見なして計算に利用することを考えます．この構造型の定義と，単純な足し算と掛け算の関数を用意したモジュールの作成例を以下に示します．

```
module modfraction
   implicit none
   public fraction,fsum,fmul
private
   type fraction
      integer num,den
   end type fraction
contains
 function fsum(f1,f2)                  ! 加算関数 (fsum = f1 + f2)
   type(fraction) f1,f2
   type(fraction) fsum
   fsum%den = f1%den*f2%den
   fsum%num = f2%den*f1%num + f1%den*f2%num
 end function fsum
 function fmul(f1,f2)                  ! 乗算関数 (fmul = f1 * f2)
   type(fraction) f1,f2
   type(fraction) fmul
   fmul%den = f1%den*f2%den
   fmul%num = f1%num*f2%num
 end function fmul
end module modfraction
```

加算関数 fsum と乗算関数 fmul は，2 個の fraction 構造体を引数に与えれば，それらの加算や乗算を計算した結果の fraction 構造体を関数値として返す関数副プログラムです．どちらも内容は簡単なので説明は省略します．このモジュールでは，後の便宜のためにデフォルトを非公開にし，fraction，fsum，fmul を public 宣言しています．

これらの関数を使えば，分数の足し算と掛け算を組み合わせた計算も記述できます．たとえば，

```
use modfraction
implicit none
type(fraction) f1,f2,f3,f4,f5
f1 = fraction(1,2)              ! f1 = 1/2
f2 = fraction(1,3)              ! f2 = 1/3
f3 = fraction(1,4)              ! f3 = 1/4
f4 = fsum(fsum(f1,f2),f3)       ! f4 = f1 + f2 + f3
f5 = fsum(f3,fmul(f1,f2))       ! f5 = f3 + f1*f2
```

というように書くことができます．

構造体を使えば，このように “分数”という数学的要素の計算を，その成分を意識せずに記述することができます．ただ，加算や乗算というなじみのある演算を関数の形で書くのはあまりスマートではあり

ません．そこで，“+”や“*”のような記号 (演算子) を使った演算を定義することができます．演算子を使った演算を定義するには 8.5 節で総称名機能を持たせるために使った interface 文において，総称名の代わりに operator(演算子) を指定し，その演算を処理する関数副プログラムを module procedure 文を使って指定します．たとえば，上記の fsum の代わりに“+”を使い，fmul の代わりに“*”を使いたいときは，contains 文の前に

```
interface operator (+)
   module procedure fsum
end interface
interface operator (*)
   module procedure fmul
end interface
```

を入れておきます．総称名機能と同様に，引数の組み合わせに応じた複数の関数を module procedure 文で指定することもできます．なお，デフォルトを非公開にしている場合には，public 文で operator(演算子) の公開を宣言しておく必要があります．

ただし，上記の関数 fsum と fmul のままでは演算の定義に使えません．サブルーチンや関数の引数には値を代入して戻り値にすることができますが，演算に使う関数の引数は戻り値の代入を禁止する必要があるからです．この禁止のため，関数の引数には intent(in) 属性を付加しておきます．すなわち，引数宣言の部分を以下のように書き換えます．

```
type(fraction) f1,f2
        ↓
type(fraction), intent(in) :: f1,f2
```

intent(in) は「入力のみである」ということを明示する属性で，これを付けて宣言された変数に値を代入するような実行文を書くとコンパイルエラーになります．

以上により，モジュール全体を書き直すと次のようになります．

```
module modfraction
   implicit none
   public fraction, operator(+), operator(*)
private
   type fraction
      integer num,den
   end type fraction
   interface operator (+)
      module procedure fsum
   end interface
   interface operator (*)
      module procedure fmul
   end interface
contains
 function fsum(f1,f2)
   type(fraction), intent(in) :: f1,f2
   type(fraction) fsum
   fsum%den = f1%den*f2%den
   fsum%num = f2%den*f1%num + f1%den*f2%num
 end function fsum
 function fmul(f1,f2)
   type(fraction), intent(in) :: f1,f2
   type(fraction) fmul
   fmul%den = f1%den*f2%den
   fmul%num = f1%num*f2%num
 end function fmul
end module modfraction
```

このモジュールでは，fsum や fmul の public 宣言をしていないので，ユーザーが直接使うことはでき

ません．構造体の演算は，演算子だけで記述する必要があります．

このように演算子を定義しておけば，上記の実行プログラム中の f4 や f5 の計算は次のように書き直すことができます．

```
f4 = f1 + f2 + f3
f5 = f3 + f1*f2
```

演算の優先順位は数値計算と同じになるので，f5 の計算では f1*f2 の計算を最初に行い，その結果を f3 に加えるという順番になります．

ただし，以下の計算はエラーになります．

```
f4 = f1 + 1
```

なぜなら，用意している加算演算関数は「分数＋分数」のみであり，「分数＋整数」の関数は用意していないからです．よってこの計算を実行するには以下のように書かねばなりません．

```
f4 = f1 + fraction(1,1)
```

しかし，これではわかりにくいというのであれば，「分数＋整数」の関数も作成して入れておきます．たとえば，上記のモジュールに以下のような関数を追加します．

```
function fsumi(f1,d)
  type(fraction), intent(in) :: f1
  integer, intent(in) :: d
  type(fraction) fsumi
  fsumi%den = f1%den
  fsumi%num = f1%num + f1%den*d
end function fsumi
```

その上で，interface 文を修正して，

```
interface operator (+)
   module procedure fsum
   module procedure fsumi
end interface
```

のように fsumi を加えれば「分数＋整数」の計算も処理してくれます．ただし，交換則を定義することはできないので，

```
f4 = 1 + f1
```

のような「整数＋分数」の計算も処理したい場合には，「整数＋分数」の関数も別途用意しておく必要があります．

なお，operator(演算子) の “演算子”は，四則演算のような既存の記号に加えて，独自の演算子を定義することもできます．新規に定義する演算子は，“ .add. ”や“ .mult. ”のように，“ . 演算子名. ”という 2 個のドットではさんだ文字列の形をしている必要があります．たとえば，operator(+) の代わりに operator(.add.) を定義し，operator(*) の代わりに operator(.mul.) を定義してやれば，

```
f4 = f1 .add. f2 .add. f3
f5 = f3 .add. f1 .mul. f2
```

のように記述することができます．ただし，f5 の計算結果は (f3+f1)*f2 に等しくなります．なぜなら，新しく定義した演算子の場合には，優先順位が決められていないので，左側から順に処理するためです．“+”のような演算子記号と混在させた場合には，演算子記号による演算が優先されます．

代入文の動作を拡張することも可能です．標準では同型の構造体を使った「構造体＝構造体」という代入文のみ可能で，異なる型を持つ構造体間の代入文や，「構造体＝整数」のような既存のデータ型との

間の代入文はエラーになります．代入文の動作を拡張すれば，そのような代入文も可能になります．

代入文の動作を拡張するには，まず引数を 2 個持ち，その第 1 引数に第 2 引数の値から導かれるデータを代入する動作を持つサブルーチンを作成します．次に，interface 文で assignment(=) をそのサブルーチンに指定すれば，「第 1 引数＝第 2 引数」という形式の代入文を処理することができます．

たとえば，上記のモジュールに以下のような文を追加すれば，「fraction 構造体＝整数」という文を処理することができます．

```
    public assignment(=)                    ! assignment(=) を公開する
     .......
    interface assignment (=)                ! assignment(=) のサブルーチン指定を追加
       module procedure fsubi
    end interface
contains
     .......
 subroutine fsubi(f,d)                      ! assignment(=) を処理するサブルーチン
    type (fraction), intent(out) :: f       ! 第 1 引数は intent(out) 属性
    integer, intent(in) :: d                ! 第 2 引数は intent(in) 属性
    f%num = d
    f%den = 1
 end subroutine fsubi
```

ここで，サブルーチン fsubi の第 1 引数には intent(out) 属性を持たせ，第 2 引数には intent(in) 属性を持たせなければなりません．intent(out) 属性は，intent(in) の逆で，必ず戻り値に使用するという属性です．このため，この属性を持つ変数に値を代入する文が内部になければエラーになります[59]．

必要ならば，「構造体＝構造体」という既存の代入文の動作を変更することもできます．「構造体＝構造体」の動作を変更するときは，右辺の構造体の成分を使って必要な計算をし，結果を左辺の構造体の成分に代入するようなサブルーチンを作成して，assignment(=) に追加指定します．

たとえば，ここまでの単純な分数計算では，分子の整数 num と分母の整数 den に制約がないので，結果が既約分数になるとは限りません．そこで，既約分数にする処理を加えた代入用のサブルーチンを作成して assignment(=) に追加指定すれば，代入するだけで常に既約分数が保証されることになります．たとえば，次のようなプログラムにします．

```
    interface assignment (=)
       module procedure fsub                ! この文を追加
       module procedure fsubi
    end interface
     .......
 subroutine fsub(f,f1)                      ! このサブルーチンを追加
    type (fraction), intent(out) :: f
    type (fraction), intent(in) :: f1
    integer n
    n = GCD(f1%num,f1%den)                  ! 分子と分母の最大公約数を計算する[60]
    f%num = f1%num/n                        ! 分子を n で割る
    f%den = f1%den/n                        ! 分母を n で割る
 end subroutine fsub
```

ただし，代入文の動作を変更するときは注意が必要です．変更した代入文はモジュール中の代入文でも有効なので，場合によっては実行時エラーになることがあります．このため，代入文を処理するサブルーチンでは，この例のように成分ごとに代入するような記述にして下さい．

構造体を利用すれば，プログラムがよりスマートに書けますが，異なるデータ型が混在する場合には，メモリアクセスが遅くなる可能性もあるため，Fortran 本来の目的である高速数値計算には向かないこともあります．プログラムの見やすさか，高速処理か，そのあたりを考慮して構造体を活用して下さい．

[59] ちなみに，どちらでも使える変数を宣言するときは，intent(inout) 属性を使います．

[60] 関数 GCD(n1,n2) は整数 n1 と n2 の最大公約数を計算して関数値とする関数副プログラムですが，内容は省略します．

第 10 章　Fortran プログラミング補足集

ここまで Fortran を用いた計算機プログラムの書き方について色々説明してきましたが，これで全てではありません．本書の目的は，プログラミングの初心者が，基本的な Fortran プログラムの書き方から学び始めて，最終的に数値計算やコンピュータシミュレーションのような応用プログラムを書けるようにすることです．このため，筆者の経験を元にして必要な項目を選択し，重要だと思われるものから優先的に書いてきました．それでも新しい文法も含めてかなりの内容を網羅したつもりです．しかし，Fortran にはまだ色々と便利な書式や関数があるので，本章では，その中のいくつかを選んで補足的に説明したいと思います．

10.1　数値型の精度指定

1.5 節で，基本的な数値型には整数型と実数型があり，実数型には単精度実数型と倍精度実数型があるという話をしました．Fortran のデフォルト実数型は単精度ですが，コンパイラの自動倍精度化機能 (付録 C) を使えばデフォルトを倍精度実数型に変更できるので，本書では 2 種類の実数型を区別せず，単に“実数型”と表現しています．しかし，大量の実数型データをバイナリ形式で保存するときには，保存容量を圧縮するために精度を落とすことが考えられますし，“4 倍精度実数型”という倍精度よりも有効桁数の多い実数型を使って，より高精度の計算をすることも可能です[61]．本節では，定数や変数の精度を変更する書式について説明します．

変数の精度は宣言文で指定しますが，以下の 3 種類の記述方法があります．ここでは，基本的な宣言文だけを紹介しますが，4.8.1 節で述べた属性や数値代入を含む拡張宣言も可能です．

```
型指定*精度数　変数 1，変数 2，...
型指定（精度数）　変数 1，変数 2，...
型指定（kind=精度数）　変数 1，変数 2，...
```

ここで，“精度数”のところには，数値型の byte 数を整定数で指定します．実数型の場合，単精度なら 4，倍精度なら 8，4 倍精度なら 16 です．たとえば，単精度実数型変数を宣言するには，

```
real*4 xs1,ys1,as1(100)
real(4) xs2,ys2,as2(100)
real(kind=4) xs3,ys3,ds3(100,100)
```

などと書きます．また，通常の整数型は 4byte ですが，最近は“倍精度整数”という 8byte の整数型 (使用可能な範囲は -2^{63}〜$2^{63}-1$) を使うことができるので，これを利用するときには，

```
integer*8 n81,m81,k81(100)
integer(8) n82,m82,k82(100)
integer(kind=8) n83,m83,ka83(100,100)
```

などと書きます．

3 種類の書式の中でどれが良いかは一概には言えません．一番目の“*”を使った書式は，古い Fortran から使われているので，コンパイラが古い場合や，昔のプログラムを継承する可能性がある場合には覚えておいた方が良いと思います．しかし，これからプログラムを書き始める場合には，二番目か，精度数の意味を明記した三番目の方が良いと思います．これらは，次のように `parameter` 変数 (4.8.2 節) を使って精度数を指定することができます．

[61] ただし，4 倍精度実数型は，必ず実装しなければならない規格ではないようで，コンパイラによっては使えない場合があります．gfortran でも，バージョン 4.8 では使えますが，4.2 では使えませんでした．

```
integer, parameter :: kp=16
real(kp) xq2,yq2,aq2(100)
real(kind=kp) xq3,yq3,dq3(100,100)
```

これに対し，“*”を使った書式では必ず数字を使って指定しなければなりません．parameter 変数で精度指定をしておけば，計算機環境に応じて精度を変更する必要があるときに便利です．

ただし，高精度の計算をする場合は，定数も高精度にしなければなりません．たとえば，“1.23”と書けばデフォルトの実数型になるので，本書の暗黙指定では倍精度実数型です．よって，このまま 4 倍精度の計算に使うと精度が落ちてしまいます．$A\times10^B$ で表される実数を 4 倍精度で表現するときは “AqB” と書きます．このため，“1.23”は “1.23q0”と書かなければなりません[62]．

しかし，parameter 変数で変数の精度を指定しているときに，e や q のような文字を使って定数の精度を指定していると，精度変更のときに定数の変更が別途必要になります．そこで，数値の後にアンダースコア “_”と精度数を付けて定数の精度を指定することができます．たとえば，倍精度の “1.23”は “1.23_8”と書くことができ，4 倍精度の “1.23q0”は，“1.23_16”と書くことができます．この精度指定には parameter 変数を使うことができるので，次のように書くことができます．

```
integer, parameter :: kp=16
real(kp) xx,yy
xx = 1.23_kp
yy = 1.2345e-15_kp*xx**3
```

この例の yy に代入している実数の指数指定は e を使っていますが，精度数が 16 なので，4 倍精度定数になります．

10.2 内部副プログラム

4.4 節で説明したように，各ルーチンの内部で宣言された変数 (ローカル変数) はルーチン内部でのみ有効で，引数以外の変数は他のルーチンから見えません．これに対し，サブルーチンや関数副プログラム自体は常にグローバルであり，特に指定することなくどのルーチンからでも使うことができます．逆に言えば，ルーチンを隠すことができないので，プログラム全体で重複しないようにサブルーチン名や関数名を決める必要があります．大規模なプログラムを作成するときには，簡単な関数副プログラムを加えるだけでも名前の選択に注意しなければなりません．

この問題を避ける一つの方法に “内部副プログラム”の利用があります．内部副プログラムというのは，メインプログラムやサブルーチンなどの内部にサブルーチンや関数副プログラムを記述するものです．内部副プログラムはそれを含むルーチン内部でのみ有効で，外部からは見えません．よって，異なるルーチンの中に名前が同じで異なる内部副プログラムを記述することもできます．いわば「ローカル副プログラム」です．

内部副プログラムは，利用するルーチンに記述した実行文の一番最後の次に contains 文を書いて，そこから end 文までの間に，通常のサブルーチンや関数副プログラムと同様の形式で記述します．すなわち，8.1 節で説明したモジュールの内部ルーチンと同じ形式です．たとえば，次のように書くことができます．

[62] ちなみに，デフォルト実数が単精度のとき，倍精度実定数は “AdB” と書きます．たとえば，“1.2”は “1.2d0”です．また，実数化の組み込み関数 real の結果が単精度実数型になるので，倍精度実数化が必要なときには関数 dble を使います．

```
program internal_test
   implicit none                ! この implicit none は内部副プログラムでも有効
   real x,y,z
   integer mb                   ! この mb は共有変数
   x  = 10
   y  = 2
   mb = 5
   call isub(x,y)
   z = cube(x)
   print *,x,y,z
contains                        ! これ以降が内部副プログラム
   subroutine isub(x,y)
      real x,y
      integer mb                ! この mb はローカル変数
      mb = 2
      y  = mb*x**2
   end subroutine isub
   function cube(x)
      real cube,x
      cube = mb*x**3            ! この mb は共有変数
   end function cube
end program internal_test
```

このプログラムでは，contains 文の直前の print 文までがメインプログラムの本体で，contains 文以降のサブルーチン isub と関数 cube が内部副プログラムです．isub と cube には implicit none の宣言がありませんが，メインプログラム本体で宣言していれば内部副プログラムでも有効なので，省略しても問題ありません．また，外部副プログラムに同じ名前を付けたものがある場合には内部副プログラムを優先的に使用します．

内部副プログラムは，以下のような点で外部副プログラムと異なります．

(1) 内部副プログラムを含んでいるルーチンの本体で宣言された変数は内部副プログラム中での共有変数として使える
(2) ただし，内部副プログラム内でその変数と同じ名前の変数を宣言した場合には，その宣言が優先してローカル変数となる
(3) 内部副プログラムの引数は，モジュール内部のサブルーチンや関数と同様の引数チェックを行う
(4) 内部副プログラムの引数は，引数変数に代入する形式で指定することができる

すなわち，第 8 章で説明したモジュール内部のサブルーチンや関数副プログラムと良く似た特徴を持っています．8.4 節で紹介した value 属性を与えて “call by value” の引数にすることも可能です．

たとえば，上記のプログラムにおいて，変数 mb はメインプログラムと内部サブルーチン isub の両方で宣言しています．このため，isub にとってはローカル変数になって，mb=2 ですが，内部関数 cube にとっては共有変数になるので，mb=5 として計算されます．引数チェックの詳細は，モジュール内部のサブルーチンの項 (8.1 節) を参照して下さい．

10.3 interface 文を用いた引数チェック

Fortran は，色々改良されてきたとはいえ，古いプログラムの継承も考慮されているため，文法的にゆるい記述がいくつか残っています[63]．その中に，サブルーチンを呼び出す call 文の記述において引数のチェックをしない，という問題があります．このため，call 文で引数の数を間違えたり引数の数値型

[63] implicit none を付けなければ暗黙の変数宣言と見なされ，宣言しなくても変数が自由に使えるというのもその一つです．キーボードやモニターがなかった時代，プログラムを書くのは時間がかかる作業でした．このため，記述量を減らすために暗黙の宣言などがあったのだと思います．

を間違えたりしてもコンパイルエラーにはならず，実行時に正常な動作をしないという形で不具合が現れます．

これに対し，モジュール内部のサブルーチンや 10.2 節で説明した内部副プログラムには，コンパイル時の引数チェック機能があって，プログラムのエラーが見つけやすくなっています．これは引数の情報を利用者側に公開するしくみを持っているためですが，外部副プログラムに対しても，その引数情報などを公開するしくみが用意されています．これが，interface 文です．interface 文を使ってプログラム中に外部副プログラムの引数の型や形状を宣言しておけば，コンパイラがその情報を使って引数のチェックをしてくれます．

interface 文は，interface と end interface ではさんだブロックで，その中に subroutine 文または function 文，その引数の宣言 (関数副プログラムの場合は関数値の宣言も)，および最後の end 文だけを取り出して入れておきます．たとえば，4.8.4 節のサブルーチン subrout や関数 fun を使用する場合，以下のように書くことができます．

```
   program test_interface
      implicit none
      external sub
      interface
         subroutine subrout(subr,xmin,xmax,n)
            external subr                ! 外部副プログラムは external 文で宣言する
            real xmin,xmax
            integer n
         end subroutine subrout
         function fun(x)
            real fun,x                   ! 関数の場合は関数値も宣言する
         end function fun
      end interface
      call subrout(sub,0.0,3.0,10)
!      call subrout(0.0,3.0,10)          ! これを入れるとエラーになる
      print *,fun(10.0)
!      print *,fun(10)                   ! これを入れるとエラーになる
   end program test_interface
```

この例のように，interface 文中には複数のサブルーチンや関数副プログラムの宣言を書くことができます．また，interface 文中に関数副プログラムの宣言を入れておくと，関数名の型宣言情報が入っているので，その関数名の型宣言を本文中でする必要はなくなります．なお，外部副プログラムを引数にするサブルーチン subrout のような場合には，この例の subr のように external 文での宣言が必要です．

interface 文で引数の型を宣言しておくと，引数変数に値を代入した形でサブルーチンをコールすることもできます．この場合，引数を記述する順番は自由です．たとえば，

```
   program test_interface
      implicit none
      external sub
      integer m
      interface
         subroutine subrout(subr,xmin,xmax,n)
            external subr
            real xmin,xmax
            integer n
         end subroutine subrout
      end interface
      m = 5
      call subrout(n=m**2,subr=sub,xmax=3.0,xmin=0.0)
   end program test_interface
```

と書くことができます．

このような引数のチェック機能や引数変数に代入して指定する機能は，モジュール内部のサブルーチ

ンや関数副プログラムのチェック機能などと同じなので，詳細は 8.1 節を参照して下さい．

interface 文による引数チェック機能を使えば，より安全なプログラムにすることができますが，宣言したルーチン内でしか有効ではないので，使用するルーチンごとに同じ interface 文を書かねばなりません．大規模なプログラムで引数チェック機能を利用したいときには，サブルーチンや関数をまとめたモジュールを作成する方が便利だと思います．

10.4 ビット操作関数

1.10.4 節で説明したように，コンピュータで取り扱っている数値は 2 進数です．たとえば，整数型の数値は 4byte，すなわち 32bit ですから，整数 1 個で，32 個の ON/OFF 判定情報を含めることができます．そこで，この 2 進数情報をダイレクトに利用するための組み込み関数がいくつか用意されています．これを “ビット操作関数”といいます．代表的なビット操作関数を表 10.1 に示します．

表 10.1 ビット操作関数

ビット操作関数	引数の型	関数の型	関数の意味
not(i)	整数型	整数型	否定 (ビット演算)
iand(i,j)	整数型	整数型	論理積 (ビット演算)
ior(i,j)	整数型	整数型	論理和 (ビット演算)
ieor(i,j)	整数型	整数型	排他的論理和 (ビット演算)
ishft(i,n)	整数型	整数型	シフト
ishftc(i,n)	整数型	整数型	循環シフト

否定 (not) は “ビット反転”ともいいます．表 10.1 の上から 4 個の “ビット演算”というのは，整数型の数値を 2 進数で表したときに，各桁のビットごとに表 10.2 の演算を施した結果で表される整数を関数値として返す関数です．

表 10.2 ビット演算の一覧表

i	j	not(i)	iand(i,j)	ior(i,j)	ieor(i,j)
0	0	1	0	0	0
0	1	1	0	1	1
1	0	0	0	1	1
1	1	0	1	1	0

たとえば，

```
i = 13
j = 11
print *,iand(i,j),ior(i,j),ieor(i,j),not(i)
```

というプログラムの出力結果は，

```
 9  15  6  -14
```

となります．11 と 13 を 2 進数で表すと “1011”と “1101”なので，各桁の論理積は “1001”となり，10 進数では 9．論理和は “1111”なので 10 進数では 15．排他的論理和は “0110”なので，10 進数では 6 になるというわけです．

最後の 13 の否定が −14 になるのは少し説明が必要です．ほとんどの計算機では負の整数を 2 の補数

で表現します．2 の補数というのは，正の整数 n に対して，n に加えると 2^M になる整数です[64]．ここで，M は整数を表現する最大ビット数で，整数型数では 32 です．さて，整数 n と n の否定を加えれば，全てのビットが 1 の整数，2^M-1 になります．よって，n の否定に 1 を加えた数が 2 の補数，$-n$ になります．すなわち，n の否定は，$-n$ から 1 を引いた数になるわけです．

これに対し，表 10.1 のシフト関数は整数 i をビットシフトする関数です．数学的に言えば，i に 2^n を掛けることに相当します[65]．n が正のときは上位シフト，負のときは下位シフトになります．ただし，関数 ishft はあふれた桁を切り捨てるのに対し，ishftc は循環させます．すなわち，ishftc を使うと，上位桁にあふれたビットは下位桁に回し，下位桁からあふれたビットは上位桁に回します．

たとえば，

```
i = 13
print *,ishft(i,2),ishft(i,-2),ishftc(i,3),ishftc(i,-3)
```

というプログラムの出力結果は，

```
52  3  104 -1610612735
```

となります．ishft(i,2) は 13 の 2^2 倍なので 52，ishft(i,-2) は 13 を 2^2 で割って切り捨てるので 3 になります．これに対し，ishftc(i,3) は 13 の 2^3 倍なので，104 になりますが，ishftc(i,-3) は下位の 3 桁を上位に回すので，−1610612735 になります．上記の 2 の補数を考慮して，なぜだか考えてみて下さい．

念のために，5.3 節の表 5.1 の注に書いてある B 編集 (2 進数表示) を使って確認してみましょう．まず，関数 ishft を使った

```
i = 13
print "(2b35.32)",ishft(i,2),ishft(i,-2)
```

の出力結果は，

```
00000000000000000000000000110100    00000000000000000000000000000011
```

となります．13 の 2 進数表示 “1101”が，指定された数だけ左や右にずれているのがわかります．また，右にずれて外に出たビットは切り捨てられています．

これに対し，関数 ishftc を使った

```
i = 13
print "(2b35.32)",ishftc(i,3),ishftc(i,-3)
```

の出力結果は，

```
00000000000000000000000001101000    10100000000000000000000000000001
```

となります．この場合は，右にずれて外に出たビットが左の方から入っていることがわかります．

ビット操作は，コンピュータの基本的動作なので高速に処理できるという利点もあります．たとえば，2^n で掛けたり割ったりするときにはシフトを使うと高速化されるし，2^n で割ったときの余りは，2^n-1 との論理積を使えば高速化されます．

---

[64] n と，n の 2 の補数を加えた 2^M は，M ビットが全て 0 で，$M+1$ ビット目だけが 1 です．よって，M ビットだけ取り出せば 0 になることから，計算機的には 2 の補数が $-n$ になるというわけです．

[65] 厳密に言えば，2^n を掛けたときにオーバーフローした場合には結果が異なります．

付録 A　gfortran を用いたコンパイルから実行までの手順

gfortran は Fortran95 の文法で書いたプログラムをコンパイルできる代表的なフリーの Fortran です．多くの Linux ディストリビューションに付属しているし，Windows 版や Mac OS 版も配布されているので，お手持ちのパソコンにインストールすれば，手軽に Fortran を利用することができます．ここでは gfortran を使用したコンパイルから実行までの手順について述べます．

まず，Fortran プログラムを作るときには，ファイル名の最後に “.f90”を付けます．つまり，

```
文字列.f90
```

という名前にします．このファイル名の最後に付加した “ドット (.)+文字”の部分を拡張子と呼びます．作成したプログラムファイルを計算機で実行させるには “コンパイル”と “リンク”という二つの過程が必要です．

コンパイルというのは，Fortran や C 言語など，人間が理解できる言語で書かれたプログラムをコンピュータが理解できる機械語に翻訳することです．コンパイルするアプリケーションを “コンパイラ”といいます．フリー Fortran コンパイラの代表が gfortran です．

gfortran でプログラムをコンパイルをするときは，gfortran コマンドを使って，

```
gfortran プログラムファイル名
```

と入力します．たとえば，test1.f90 というファイル名のプログラムをコンパイルするときは

```
gfortran test1.f90
```

と入力します．コンパイル時に文法エラーなどが見つかると，エラーメッセージを出力して終了します．

gfortran コマンドは，コンパイルが成功すると，引き続きリンクを行います．プログラムはコンパイルしただけでは実行できません．コンパイラはプログラミング言語を機械語に翻訳するだけであり，複数のルーチンを結合して OS 上で起動可能な形式にする作業までは行わないからです．この結合処理がリンクです．単にプログラム中のルーチンのリンクだけではありません．入出力文の `read` や `write`，組み込み関数の `sin` や `log` 等は，標準ライブラリルーチンとして用意されているので，必要に応じてこれらのリンクも行います[66]．

リンクにも成功すると，最後に “a.out”という名前のファイルが作成されます．これが OS 上で直接実行させることができる機械語で書かれたファイルで，これを実行形式ファイルといいます．もしリンクに失敗した場合は，エラーメッセージを出力して終了します．このとき a.out は作成されません．

実行形式ファイルは，一般のコマンドのようにファイル名を入力することでプログラムを実行することができます．ただし，次のように頭に “./”を付ける必要があります[67]．

```
./a.out
```

プログラムが正常に動作すれば，それに書かれた一連の計算を行って終了します．計算の途中に `print` 文の記述があれば，結果を画面に出力するし，`write` 文の記述があれば，指定したファイルに書き出します．また，標準入力の `read` 文の記述があれば，その時点でプログラムが一時停止して入力待ちの状態になります．この状態で必要な数値をキーボードから入力すれば計算は再開します．

以上が，gfortran を用いた最も単純なコンパイルからプログラム実行までの手順ですが，gfortran コマンドにファイル名を与えただけの命令では，どんなプログラムをコンパイルしても a.out という同じ

[66] コンパイラである gfortran 自体はリンクする機能を持っていませんが，gfortran 内部から OS に付属しているリンク用アプリケーション (リンカ) を呼び出すことができるので，リンクも実行することが可能なのです．

[67] “./”は「現在のディレクトリにあるファイル」という意味です．OS の設定によっては不要な場合もありますので，詳細は Linux などの解説書で確認して下さい．

名前の実行形式ファイルになってしまうので不便です．また，Fortran プログラムは計算速度が重要なので，最適化をしてできるだけ高速に処理する方が良いでしょう．さらに 1.5 節で述べたように数値計算をするときは倍精度実数を使うべきなので，自動倍精度化オプションも必要です．

このため，gfortran でプログラムをコンパイル・リンクするときは，以下のようなオプションを付けることをお勧めします．

```
gfortran -O -fdefault-real-8 test1.f90 -o xtest1
```

-O が最適化のオプション，-fdefault-real-8 が自動倍精度化のオプションです[68]．また，-o のオプションの次の文字列 (ここでは xtest1) は実行形式ファイル名指定です．大文字の-O と小文字の-o を間違わないようにして下さい．

この場合，コンパイルとリンクが正常に終了すると，-o オプションで指定した “xtest1 ”という名の実行形式ファイルが作成されます．このため，プログラムを実行するには，

```
./xtest1
```

と入力することになります．

また，付録 F で紹介している数値計算を補助する外部ライブラリを使用したプログラムをコンパイル・リンクする場合は，それらのライブラリファイルをオプションとして付加する必要があります．たとえば，FFTW を含んだプログラムの場合は，

```
gfortran -O -fdefault-real-8 test1.f90 -o xtest1 -L/usr/local/lib -lfftw3
```

のように入力します．ここで，-L オプションが指定している “/usr/local/lib ”は，FFTW ライブラリ (libfftw3.a) の入ったディレクトリを示すオプションなので，使っている環境に応じて書き換える必要があります．

なお，4.5 節の脚注 19 で注意したように，最近の gfortran では，サブルーチンの引数が配列かスカラーかをチェックして，規定に合わない場合にはエラーを出すことがあるようです．このエラーの出し方は，プログラムの予期せぬ実行時エラーを防ぐ可能性があるので必ずしも悪いことではないのですが，正しく動作することがわかっている場合には，余計なお世話だと感じることもあるので難しいところです．

プログラムを書き始めた段階でこのようなエラーが出る場合には, エラーメッセージに従って配列用とスカラー用のプログラムを別途用意する方が良いと思います．しかし, 既存のプログラムをそのまま使いたい場合や, 書き換えなくても正しく動作することが明白な場合には, オプション-fallow-argument-mismatch を付けることでエラーを回避することができます．このオプションは，引数チェックにおける問題点があっても，エラーではなく警告を出すだけにするように指示するもので，これを付ければ，エラーでコンパイルが止まることはなく，他のエラーがなければ正常に終了します．

ただし，このオプションの使用には注意が必要です．このオプションは，その名称からわかるように，引数のミスマッチ，すなわち call 文側と subroutine 文側で引数の型などが異なっている場合に，そのままではエラーになるところを許可してコンパイルを継続させるという指定です．このため，このオプションを付けると，配列かスカラーかの違いだけではなく，実数の引数に整数を与えるという明らかな間違いをした場合でも，エラーではなく警告になります．もっとも，従来の Fortran コンパイラでは，このようなミスマッチがあっても警告さえ出さないことが多かったのですから，まだましだとも言えますが，そのあたりを承知した上でこのオプションを利用するようにしてください．

[68] 10.1 節の脚注で述べたように，デフォルト実数が単精度の場合，倍精度実数型定数は 1.5d3 のように d で指数指定をし，倍精度化関数には dble を使います．gfortran の最近のバージョンでは，-fdefault-real-8 オプションを付けるとこれらが 4 倍精度になるようです．4 倍精度計算は時間がかかるので，これを抑制するときにはオプション-fdefault-double-8 を追加して下さい．

付録B　エラー・バグへの対処法

プログラムを開発する際の最大の問題は“エラー (Error)”の発生です．コンパイル・リンク・実行という一連の過程の中でそれぞれエラーが発生する可能性があります．エラーはプログラム開発にとって付き物であり，避けては通れません．エラーの出ないように慎重にプログラムを書くことはもちろんですが，出たら出たでそれらに如何に素早く対処できるかが腕の見せ所です．ここではエラーをまとめて，それぞれの対処法を示します．なお，具体的なエラーメッセージは gfortran の出力を利用していますが，他のコンパイラでも形式が違うだけで出力情報の内容はあまり変わらないと思います．

(1) コンパイルエラー

コンパイルとは，作成したプログラムを文法に従って機械語に翻訳することです．このため，コンパイルエラーとは文法的な間違いのことです．プログラムに文法的間違いがあれば，コンパイラがエラーメッセージを出力して終了します[69]．コンパイラのエラーメッセージ形式はコンパイラによって違いますが，gfortran では次のような形式で出力されます．この例は，test1.f90 というプログラムファイルをコンパイルしたときのエラー出力です．

```
test1.f90:15.7:
   do i = 1, 100
      1
Error: Symbol 'i' at (1) has no IMPLICIT type
```

以下に，この出力の読み方を示します．

(1) 1 行目はエラーのあるプログラムファイル名 (この例では，test1.f90) とエラーの位置
　この例では，エラーの位置が 15.7 となっていますが，15 はプログラムの行番号，7 は左から数えた文字の位置を示す数値です．
(2) 2 行目はエラーのあるプログラム文のコピー
(3) 3 行目は“1”を上向き矢印のように使って，2 行目の文のエラーが起きている位置を指定
(4) 4 行目は 3 行目が示した部分の具体的なエラーの説明
　このメッセージを頼りにエラーを見つけてプログラムを修正します．

この例の 4 行目のメッセージは「3 行目の 1 が示している変数 i は，暗黙の型がない」という意味です．これは，変数 i が宣言されていないことが原因です．

コンパイルエラーは，エラーメッセージに従って修正すればいいので，それほどの手間ではありません．ただし，ある文に文法エラーがあると，その文が存在しない状態で引き続きコンパイルを行うため，その波及効果で下方のプログラムがエラーになることがあります．たとえば宣言文に誤りがあると，その宣言文で宣言している変数がコンパイラにとっては宣言されていないことになり，その変数を使っている文が全てエラーになります．エラーメッセージが多いときは，一度に全部チェックするのではなく，最初の方のメッセージに従ってある程度修正したら再度コンパイルして，まだエラーが出るかどうかを確認した方が良いと思います．

[69] コンパイラが出すメッセージには“Warning”(警告) と“Error”(エラー) があります．Warning は「このままでも実行できるけど，ひょっとすると予期しない結果が起こる可能性があります」という意味なので，必ずしも修正する必要はありません．このため，メッセージが Warning だけのときは強制終了せずにリンクの作業に移ります．しかし，文法的には間違いがなくても，作成者の意図とは違ったプログラムになっている可能性があるので，念のためにプログラムを確認した方が良いでしょう．

(2) リンクエラー

リンクとは，別々に翻訳されたルーチンを結合して実行形式ファイルにする作業のことです．よって，リンクエラーは用意されていないサブルーチンや関数を使ったときに起こります．たとえば，

```
program test1
  implicit none
  real x,y
  x = 5
  y = 100
  call subr(x,y)          ! subr が用意されていないとリンクエラーになる
  print *,x,y
end program test1
```

のようなプログラムを作って，これだけを単独でコンパイル・リンクすると

```
/tmp/ccy3lcdo.o: In function ‘MAIN__’:
test1.f90:(.text+0x6d): undefined reference to ‘subr_’
collect2: ld returned 1 exit status
```

のようなメッセージが出力されます．以下に，この出力の読み方を示します．

(1) 1 行目はエラーが起こっているルーチンの表示
　この例では “MAIN”，すなわちメインプログラム中で起こっていることが示されています．
(2) 2 行目はエラーの原因
　このメッセージを頼りにエラーを見つけてプログラムを修正します．

この例の 2 行目のメッセージには「subr という名への参照が定義されていない」とあります．これは，メインプログラム中でコールしている subr という名のサブルーチンが用意されていないのが原因です[70]．

標準の組み込み関数が用意されていないことはまずないので，リンクエラーのほとんどはプログラム中で呼び出したサブルーチン名や関数名の書き間違いが原因です．リンクエラーのメッセージにはコンパイルエラーのようなエラー行の表示はありませんが，エラーメッセージの中に見つからなかったサブルーチンや関数の名前が表示されているので，エディタの検索機能を使えば，見つけるのはさほど困難ではありません．

(3) 実行時エラー

一番厄介なのが実行時のエラーです (こういうプログラムは “バグ (虫)”があるといいます)．なぜなら，コンパイルエラーやリンクエラーのようなエラーの場所を特定する情報が出力されないので，どこでエラーが起こっているのかを探す作業が大変だからです．実行時エラーを探すには，プログラムを詳細にチェックするしかありません (この作業を「バグを取る」という意味で “デバッグ”といいます)．

それでも，答えが合っているか否かは別として，プログラムがなんとか終了すれば良いのですが，バグによっては実行した後でうんともすんともいわなくなってしまう場合があります．これは無限ループに入ってしまったか，暴走したかのどちらかだと考えられます．通常，このような状態で実行を強制的に終了させるには，コントロールキーを押しながらCキーを押します．これでまず間違いなく止まります．

もう一つ，以下のようなメッセージを出して強制的に終了する場合があります．

```
Segmentation fault
```

これはプログラムではなく OS が出力しているメッセージで，主として実行中のプログラムがそれ以外の実行中のプログラムのメモリ領域を破壊しようとしたときに起こります．たとえば，次のように配

[70] メッセージ中では MAIN__とか，subr_のように名前の後ろにアンダースコアが付いていますが，これはコンパイラが自動的に付加するものなので無視して下さい．リンクは gfortran がするのではなく，ld というリンカがするのですが，リンカは OS によって異なるので，gfortran を使っていてもこの形式でエラーメッセージが出るとは限りません．

列宣言の範囲外の要素に値を代入しようとすると起こることがあります[71].

```
real a1(10)
integer i
do i = 1, 100000
   a1(i) = i
enddo
```

この例では，10 個しか用意されていない配列に，100000 番目まで値を代入しようとしているのですから，どこのメモリに値を代入しようとしているのか不明なのが問題なのです．

また，サブルーチンの引数に型の合っていない数値を与えたり，戻り値を与える引数に定数を与える，などの不注意なサブルーチンコールでも Segmentation fault が起こる可能性があります．Segmentation fault は，実行時に問題が出た段階で起こるので，適当に `print` 文を入れて，どこまでが出力されて，どこからが出力されないかを確認することでエラーの起こっている場所を特定することができます．

しかし，このような実行時エラーはコンピュータの動作が正常でないという形で表面化するので，まだましです．原因を特定するのに時間がかかるかもしれませんが，修正しなければ動かないのだから実害はありません．実を言うと，一番やっかいなエラーとは，ちゃんと実行してはいるのだけど，計算結果が何となくおかしい，というものです．

筆者は何年も Fortran のプログラムを書いてきましたが，実行時のエラーほど見つけにくいものはありません．それでも見つかれば良い方で，場合によってはバグの存在を知らずに結果を出し，学会直前に気がついた，なんてこともありました．大規模なシミュレーションプログラムを作るとき，長いプログラムを書くのに時間がかかるのはもちろんですが，取りあえず動いたプログラムの計算結果が正しいかどうかを確認する作業にもかなりの時間が必要です．本書で紹介した「エラーの出にくいプログラムの書き方」は，この確認作業にかかる時間を少しでも減らすためのものです．しかし，最後に必要なのは「地道な努力と忍耐」なのです．

[71] 実際にはこの程度の簡単なプログラムでは起こるとは限らないのですが，だからといってこんなプログラムを書いてはいけません．

付録 C　自動倍精度化オプション

コンピュータシミュレーションでは大量の数値を使って複雑な計算をくり返し，必要に応じてそれらを集積する作業を行います．このとき問題となるのが，1.10.2 節で述べた“桁落ち”です．桁落ちを減らすための工夫をいくつか紹介しましたが，究極的には演算精度を上げるしかありません．

Fortran で取り扱える実数には，単精度実数型と倍精度実数型がありますが，数値計算をするには倍精度実数型を使うべきです．倍精度実数計算は単精度実数計算と比べてそれほど実行時間はかかりません．これは最近の CPU が倍精度実数計算用のハードウェアを装備しているからです．

Fortran の仕様では，デフォルトの実数型が単精度なので，`real` 宣言をした実数は単精度実数型です．このため，基本仕様のままで倍精度実数計算をするには，変数宣言をするときに `real*8` や `real(8)` などの 8byte 指定が必要です (10.1 節参照)．

たとえば，デフォルト仕様のままでも，

```
real*8 x,a1(10)
```

と宣言すれば，変数 `x` や配列 `a1` は倍精度になります．しかし，宣言文を変更するだけでは不完全です．`1.0` や `1.23e5` などの実定数も基本仕様では単精度なので，倍精度実定数に書き換える必要があります．以下に，単精度実定数を倍精度実定数にする変更例を示します．

```
1.23e5      →   1.23d5        ! e を d で置き換える
4.56e-15    →   4.56d-15      ! e を d で置き換える
3.14        →   3.14d0        ! 末尾に d0 をつける
1.0         →   1.0d0         ! 末尾に d0 をつける
```

`1.23e5` のような指数部を付加した単精度実定数は，`e` の代わりに `d` を使って書けば倍精度実定数になるのですから，慣れればそれほど問題ではありません．しかし `1.0` や `3.14` のような指数のない定数は，自然な書き方なので間違いに気付かない可能性があります．しかし，基本仕様ではこれらは単精度実定数なので，倍精度にするには `d0` を付けなければなりません．

そもそも倍精度で計算するのを基本にするのに，常に倍精度を意識した書式を利用しなければならないのは，面倒なだけでなく間違いに気がつかないまま精度の悪い計算をしている可能性もあります．そこで最近の Fortran コンパイラのほとんどが自動倍精度化オプションを装備していることを幸いに，本書では単精度実数型と倍精度実数型を使い分けることはせず，単に“実数型”として説明しました．自動倍精度化オプションを付けてコンパイルすれば，`real` だけの宣言文で倍精度実数変数を宣言できるし，`1.0` や `1.23e5` のような定数はそのままで倍精度実定数を意味します．

表 C.1 に筆者が使っている Fortran コンパイラの自動倍精度化オプションを示します．この他のコンパイラについてはそれぞれのマニュアルをチェックして下さい．大抵は用意されていると思います．

表 C.1　コンパイラに応じた自動倍精度化オプション

コンピュータ	コンパイラ	ベンダー	自動倍精度化オプション
パソコン	`gfortran`	GNU	`-fdefault-real-8`
パソコン	`ifx`	Intel	`-r8` または `-real-size 64`
パソコン	`f95`	Absoft	`-N113`
スパコン	`sxf90`	NEC	`-Wf,-A dbl4`

自動倍精度化は，コンパイルするときにオプションを加えれば良いだけなので，プログラムを変更することから考えれば楽なものです．ぜひ利用して下さい．

付録D　Big Endian と Little Endian

大量のデータを精度を落とさずにファイルに保存するときは，5.5 節で説明した書式なし write 文を使ってバイナリ形式で保存する方が良いと思います．これは計算機のメモリに保存された内部形式そのままでファイルに保存するため，書式付き write 文を使ったテキスト形式出力よりも出力データ量が小さくなるからです．

しかし，バイナリ形式で保存するときは注意が必要です．テキスト形式で保存したファイルは，我々でも直接読むことができるのですから，どんなコンピュータでも同じように読むことができます．このため，大型計算機で計算した結果をテキスト形式で出力しておけば，そのままパソコンのプログラムから読み込んで処理をすることができます．しかしバイナリ形式で出力すると必ずしもうまくいきません．バイナリ形式は計算機が直接計算を実行するためのものですから，計算機のアーキテクチャによって違っている可能性があるからです．しかし，現在はバイナリ形式も標準化されていて，ほとんどの計算機で整数型も実数型も統一した形式が採用されています (1.10.4 節参照)．このため，バイナリ形式を使って大型計算機が出力したデータをパソコンのプログラムで処理することができます．

ただやはりいくつかの制約は残っています．その一つが Endian 問題です．コンピュータが扱う数値の内部形式は整数型で 4byte，倍精度実数型で 8byte ですが，この byte の並びが上位の桁からのコンピュータと，下位の桁からのコンピュータがあるのです．前者を Big Endian，後者を Little Endian といいます．イメージ的に表現すれば，1234 という数字を保存すると，Big Endian では 01234 の順番で保存するのに対し，Little Endian では 43210 の順番で保存するようなものです．これらの違いは CPU のアーキテクチャによるものなのでユーザーが変更することはできません．

最近のほとんどのパソコンは，CPU に Intel かその互換チップを使っていますが，これらは Little Endian です．これに対し，以前の Macintosh で使われていた Motorola の CPU や NEC のスパコンは Big Endian です．このため，スパコンの Fortran で計算して書式なし write 文で出力すると，そのままではパソコンの Fortran で読むことができません．

しかし，要はデータ入出力時の形式の問題だけなので，最近の Fortran コンパイラの多くはコンパイラオプションや環境変数などで Big から Little へ，あるいはその逆への変換をしてから入力または出力をすることができるようになっています．たとえば，NEC SX の場合にはプログラム実行前に環境変数 F_UFMTENDIAN を指定することで変換できます．

```
setenv F_UFMTENDIAN 10,20        #  C シェル系
```

環境変数で指定するのは書式なし write 文の装置番号です．この例の場合，write(10) と write(20) が Little Endian に変換して出力されます．この変換を使うと，メモリ上で Big な形式を Little にして出力するのですから若干計算時間にロスが出ます．しかし，byte 並びを逆にするだけですからテキスト変換よりは楽です．

これに対し, 入力側で変換するやり方もあります．たとえば, gfortran の場合には -fconvert=big-endian というオプションを加えれば，書式なし read 文で入力するときに Big Endian から Little Endian に変換してくれます．最近の Fortran コンパイラならば大抵変換機能が付いているようなので，マニュアルで確認すれば良いでしょう．

筆者はどちらかといえばスパコン側で変換する方が良いと思います．スパコンで行う計算は元々時間がかかるものであり，それに比べれば出力時のデータ変換による時間増加はわずかです．これに対してパソコン側で変換するのは時間もかかるし，変換するかどうかを問題に応じて使い分けるのも面倒です．

ただし，スパコンでの計算には時間制限があるので，データ変換のために CPU 時間を削るのはもったいないとも考えられます．スパコンはデータ変換に時間がかかるという話もありますので，どちらが良いかはスパコン管理者に確認した方が良いでしょう．

付録 E　ASCII コード

文字列の比較などに使われる半角英数字の文字コード (ASCII コード) を表 E.1 に示します.

表 E.1　ASCII コード表

	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E	F
2	␣	!	"	#	$	%	&	’	(	)	*	+	,	-	.	/
3	0	1	2	3	4	5	6	7	8	9	:	;	<	=	>	?
4	@	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
5	P	Q	R	S	T	U	V	W	X	Y	Z	[	\	]	^	_
6	‘	a	b	c	d	e	f	g	h	i	j	k	l	m	n	o
7	p	q	r	s	t	u	v	w	x	y	z	{	|	}	~	

半角英文字は 1byte(8bit) 文字であり，左端の数字が上位 4bit，上端の数字 (16 進数) が下位 4bit です. たとえば，“T”の文字は 16 進数の 54，“<”は 3C，スペース “␣”は 20 です．1F 以下と 7F は制御コードに割り当てられていて文字としては定義されていません.

なお，5C のバックスラッシュ “\”は，日本語フォントでは “¥”に割り当てられています.

付録 F　数値計算を補助するフリーのサブルーチンライブラリ

複雑な計算アルゴリズムを使ったプログラムを一から全て書くのは結構大変です．汎用性のあるアルゴリズムを使ったプログラムは，インターネット上にサブルーチンライブラリとして公開されているものがあるので，これらを利用してプログラムを作るのもスマートな方法です．ここでは筆者が使ったことのあるライブラリの中からお勧めをいくつか紹介します．

F.1　高速フーリエ変換

偏微分方程式の解法やデータ解析で良く使う高速フーリエ変換 (Fast Fourier Transform, FFT) は，アルゴリズムが複雑で，効率の良いプログラムを自作するのは大変です．筆者は最も簡単なレベルのプログラムを作ったことはありますが，最近はネットで公開されているものを使っています．お勧めなのは，FFTW(Fastest Fourier Transform in the West) です．

```
http://www.fftw.org/
```

このライブラリには，Fortran から使用できるサブルーチンが入っています．

F.2　多倍長計算

Fortran での倍精度実数は 8byte で，有効数字は 15 桁程度です．4 倍精度実数は 16byte で，有効数字が 33 桁程度に増加しますが，それ以上は無理だし，コンパイラによっては 4 倍精度実数が使えないこともあります．筆者は多項式の解を計算するときに高精度計算が必要になったことがあります．数十次の多項式を計算するときは，最大次数の項と最小次数の項との数値の差が非常に大きくて，桁落ちが深刻な問題になるからです．

このときに便利なのが多倍長計算ライブラリです．これは，任意の精度で実数計算をすることができるものです．通常の計算より処理に時間がかかりますが，精度優先のときにはやむを得ません．ネット上には何種類か公開されていますが，筆者が使って便利だったのは，FMLIB です．

```
http://myweb.lmu.edu/dmsmith/FMLIB.html
```

これはモジュールになっていて，四則演算は「`+ - * /`」の記号を使った数式で書けるし，sin や cos などの関数も充実しています．第 9 章で説明している構造体の知識が少しだけ必要ですが，変数宣言程度なのでそれほど難しくはありません．筆者は，このライブラリを利用して，倍精度実数計算を使った多項式の解を求めるプログラムを，それほど大きな変更をせずに多倍長で計算させることができました．

F.3　長周期擬似乱数発生

計算機シミュレーションには，時間発展の方程式を解くことで未来を予測する決定論的手法と，サイコロを振って確率的に将来の可能性を探る確率論的手法があります．後者の確率論的手法で，サイコロに相当する働きをするのが乱数です．4.8.6 節で紹介したように，Fortran には乱数 (擬似乱数) 発生用の組み込みサブルーチン `random_number` が用意されているので，

```
call random_number(x)
```

のように書けば，擬似乱数 `x` を得ることができます．しかし，擬似乱数は所定のアルゴリズムを用いた計算で発生させるため，一定の周期後に初期値に戻ります．この周期長が問題で，周期の短い擬似乱数を使うと確率分布に偏りが生じて良いシミュレーション結果が得られません．`random_number` では周期長が不足するときにお勧めなのは，周期が非常に長い擬似乱数として有名な，Mersenne Twister です．これを考案した松本眞さんが以下のサイトでプログラムを公開されています．

http://www.math.sci.hiroshima-u.ac.jp/m-mat/MT/mt.html

松本さんが公開されているのはC言語で書かれたものですが，このサイトの中にある「リンク集」にFortran用の公開先が入っています．

F.4　数値計算ライブラリ

連立方程式の解法や数値積分など，汎用性のある数値計算ライブラリがあれば，数値計算のプログラムを作るのが楽になります．ただ，残念ながら筆者はフリーの良いライブラリを知りません．ネットで公開されていて使えそうなものに，GSL（GNU Scientific Library）があります．

https://www.gnu.org/software/gsl/

ただし，基本的にC言語用なので，Fortranで使うには若干テクニックがいるようです．また，ちょっと使ってみましたが，必ずしも速くありません．ベッセル関数などの特殊関数は，その高速バージョンを公開しているサイトからダウンロードした方が良いと思います．筆者は，数値計算の教科書を読んで原理から自作するか，参考文献 [5,6] などに掲載されているプログラムをコピーして使うことが多いです．

なお，これから色々な数値計算アルゴリズムを勉強しようという場合には，筆者が書いた「Fortranハンドブック」もお勧めです [7]．

F.5　可視化ソフトウェア

Fortranを使った数値計算は，あくまでも「数値の計算」なので，入力も数値，出力も数値です．複雑な計算をすると，大量の数値が出力されるので，出力された数値だけで結果を読み取るのは容易ではありません．そこで必要不可欠なのが可視化ソフトウェアです．フリーの可視化ソフトウェアとして有名なものにgnuplotがあります．

http://www.gnuplot.info/

プログラムの結果をファイルに保存しておけば，gnuplotからそれを読み込んで，様々なグラフに加工して表示することができます．

また，コンピュータシミュレーションのような物理的に起こっている現象を模擬する計算の結果などを可視化する場合には，ParaViewもあります．

https://www.paraview.org/

このような可視化ソフトは数値計算やコンピュータシミュレーションにとって必要不可欠なものだと思います．

もっとも，筆者はgnuplotやParaViewを使ったことがありませんのでその使い勝手に関する評価はできません．最近のパソコンは基本操作が全てビジュアルであり，グラフィック出力機能はOSに組み込まれています．Fortranにはグラフィック出力に関する文法はありませんが，他の言語で書かれたグラフィックライブラリとリンクすれば，Fortranプログラムからサブルーチンをコールする形式で図を描くことが可能です．そこで，筆者は自前のFortran用グラフィックライブラリを作成して使っています．このライブラリは，以前は摂南大学のウェブサイトで公開していたのですが，残念ながらそのサイトは閉じたので現在は公開していません．いつかまた別の形で公開できればと思っています．

あとがき

本書は，筆者が摂南大学で教鞭をとっていた頃に卒業研究生向けに書いた Fortran のテキストがベースになっています．これを，大阪大学レーザーエネルギー学研究センター (現：大阪大学レーザー科学研究所) におられた福田優子氏が Web で公開できる Fortran テキストを探しているというので提供したところ，結構気に入ってくれて，学生に配ったり，他の計算機センターに紹介してくれました．その紹介先の一つである東北大学サイバーサイエンスセンターから広報誌 SENAC へテキストの内容を投稿してくれないかとの依頼があったので，加筆・修正をして 2012 年度に 3 回の連載として掲載されました．

その後，阪大の Web で公開していたテキストを見られた技術評論社の方から，「Fortran の解説とそれを利用した数値計算の本として出版しないか」という提案をいただきました．これまで勉強してきたことを活かす良い機会だと思って引き受けてから 2 年近くかかりましたが，なんとか 2015 年 7 月に “Fortran ハンドブック” という題で出版されました [7].

本書のルーツである摂南大学の卒業研究生向けテキストのレベルでは Fortran の基本的な使い方さえ書いてあれば十分でした．それが，このように色々なところで使っていただける機会を得たために，もう少し高度な内容まで含めるように加筆・修正をくり返してきました．また，2017 年に東北大学サイバーサイエンスセンターから「Fortran 入門」というテーマで講習会をしてもらえないかというお話しをいただいたので，講習会用テキストとして少し構成を変え，導入部として簡単な計算機のしくみの説明も加えました．

同じ頃，それまで筆者は使っていたけれど初心者向きではないなと思って省略した文法を追加したバージョンも作りかけていたのですが，卒研生に教えるほどの内容ではなかったのでそのまま放置していました．最近，久しぶりに福田優子氏と Fortran についてお話しする機会があり，その放置していたバージョンを送ったところ，これで阪大の公開版を更新したいという要望をいただきました．そこで全体を見直して必要に応じて修正を加えたのがこの 2024 年度版です．ところが，改訂中に別件で昔のプログラムを gfortran で動かす機会があったのですが，そのとき脚注 19 で示したエラー出力に気づきました．しかしエラーになる理由がわからなかったので，福田氏から高性能 Fortran 推進協議会幹事である NEC の林 康晴氏に聞いていただいたところ，Fortran90 以降の規格の問題であることがわかりました．そこで，この林氏に教えていただいた規格の内容とその対応策もこの改訂版に入れさせてもらった次第です．

最後に，筆者が適当に書いたテキストをこのようなまともな文書にまとめ直すきっかけを与えてくれたばかりか，継続して色々なサポートをしていただいている福田優子氏に深い感謝の意を表します．また，広報誌 SENAC で連載するに当たって，つたない文章を熱心に読んで多数の有益なご意見を下さった東北大学サイバーサイエンスセンターの方々にも深く感謝いたします．加えて，この 2024 年度版を作成するに当たって新しい規格に関する詳しい情報や廃止予定の文法などについて色々と御教授いただきました林 康晴氏にも深く感謝いたします．

参考文献

[1] “入門 Fortran90 実践プログラミング”，東田幸樹・山本芳人・熊沢友信，ソフトバンク，1994.

[2] “数値計算の常識”，伊理正夫・藤野和建，共立出版, 1985.

[3] “コンピュータシミュレーションの基礎”，岡崎　進，化学同人, 2000.

[4] “CIP 法”，矢部　孝・内海隆行・尾形陽一，森北出版, 2003.

[5] “Numerical Recipes in Fortran 77, Second Edition”, W. H. Press, S. A. Teukolsky, W. T. Vetterling, B. P. Flannery, Cambridge University Press, 1996.

[6] “Numerical Recipes in Fortran 90”, W. H. Press, S. A. Teukolsky, W. T. Vetterling, B. P. Flannery, Cambridge University Press, 1996.

[7] “Fortran ハンドブック”，田口俊弘，技術評論社，2015.

索 引

R

S

T

U

V

W

あ

か

さ

た

な

は

ま

ら